

KOWAL



CPC

464, 664 & 6128

EIN DATA BECKER BUCH

KOWAL



CPC

464, 664 & 6128

EIN DATA BECKER BUCH

ISBN 3-89011-101-7*


Copyright © 1985 DATA BECKER GmbH
Merowingerstraße 30
4000 Düsseldorf

Alle Rechte vorbehalten. Kein Teil dieses Buches darf in irgendeiner Form (Druck, Fotokopie oder einem anderen Verfahren) ohne schriftliche Genehmigung der DATA BECKER GmbH reproduziert oder unter Verwendung elektronischer Systeme verarbeitet, vervielfältigt oder verbreitet werden.

Wichtiger Hinweis:

Die in diesem Buch wiedergegebenen Schaltungen, Verfahren und Programme werden ohne Rücksicht auf die Patentlage mitgeteilt. Sie sind ausschließlich für Amateur- und Lehrzwecke bestimmt und dürfen nicht gewerblich genutzt werden.

Alle Schaltungen, technischen Angaben und Programme in diesem Buch wurden von dem Autoren mit größter Sorgfalt erarbeitet bzw. zusammengestellt und unter Einschaltung wirksamer Kontrollmaßnahmen reproduziert. Trotzdem sind Fehler nicht ganz auszuschließen. DATA BECKER sieht sich deshalb gezwungen, darauf hinzuweisen, daß weder eine Garantie noch die juristische Verantwortung oder irgendeine Haftung für Folgen, die auf fehlerhafte Angaben zurückgehen, übernommen werden kann. Für die Mitteilung eventueller Fehler ist der Autor jederzeit dankbar.



Inhaltsverzeichnis

0. Vorwort	1
1. Der CPC und was man mit ihm machen kann . . .	5
2. Nützliche Kleinigkeiten - einfach programmiert . .	9
2.1 Vorbemerkung	9
2.2 Die Prozentrechnung	10
2.3 Der kleinste, der größte und der Durchschnitts- wert	17
2.4 Die alphabetische Ordnung	22
2.5 Spielen Sie Lotto ?	27
2.6 Schaufensterwerbung	34
2.7 Ideenfundgrube	42
3. Auto und Computer	45
3.1 Vorbemerkung	45
3.2 Benzinverbrauch - langfristig kontrolliert	46
3.3 Fahrtstreckenoptimierung	52
3.4 Autokosten - fest im Griff	61
3.5 Ideenfundgrube	63
4. Geld und Kredit	65
4.1 Vorbemerkung	65
4.2 Die Zinsrechnung	66

4.3	Ihr Geld auf Ihrem Sparkonto (Zinseszinsrechnung)	71
4.4	Der effektive Jahreszins	75
4.5	Schulden und ihre Tilgung (Tilgungsplan)	83
4.6	Ideenfundgrube	91
5.	Texten und Drucken	93
5.1	Vorbemerkung	93
5.2	Deutsche Sonderzeichen	95
5.3	Ein individueller Geschäftsbrief	99
5.4	Textverarbeitung	107
5.5	Ideenfundgrube	124
6.	Der Staat und das Geld	125
6.1	Vorbemerkung	125
6.2	Die Rentenberechnung	128
6.3	Der Lohnsteuerjahresausgleich	137
6.4	Ideenfundgrube	161
7.	Haushalt und Gesundheit	162
7.1	Vorbemerkung	162
7.2	Der rationale Haushalt	164
7.3	Ein elektronischer Kalorienzähler	179
7.4	Ideenfundgrube	186
8.	Ihr Rechner und Ihre Hobbys	187
8.1	Vorbemerkung	187
8.2	Malen, zeichnen und die Erstellung von Graphiken	189

8.3	Wir machen Musik	205
8.4	Die Fußball-Bundesliga	220
8.5	Ideenfundgrube	232

9.	Die lieben Kleinen	237
----	------------------------------	-----

9.1	Vorbemerkung	237
9.2	Das Einmaleins	238
9.3	Primzahlenprüfung	242
9.4	Die Ermittlung einer Nullstelle	246
9.5	Der Vokabeltrainer	251
9.6	Ideenfundgrube	258

10.	Spielen mit dem CPC	267
-----	-------------------------------	-----

10.1	Vorbemerkung	267
10.2	Bildschirmtennis in Abwandlung	268
10.3	Vier gewinnt	276
10.4	Ideenfundgrube	279


11.	Schlußbemerkung	287
-----	---------------------------	-----

Anhang A

	Wichtige BASIC-Anweisungen	289
--	--------------------------------------	-----

Anhang B

	Wichtige Fehlermeldungen	293
--	------------------------------------	-----



0. Vorwort

Der extreme Preisverfall auf dem Computermarkt hat dazu geführt, daß man heute Microcomputer für relativ wenig Geld kaufen kann, die noch vor kaum mehr als zehn Jahren fast unbezahlbar waren.

Während damals nur große Organisationen wie Ministerien, Universitäten und Großunternehmen in der Lage waren Computer zur Lösung bestimmter Aufgaben zu nutzen, finden wir heute schon in vielen privaten Haushalten elektronische Geräte, die von ihrer Leistungsfähigkeit her gesehen in der damaligen Zeit ein kleines Rechenzentrum darstellten.

Der eine oder andere Käufer eines Microcomputers (mit oder ohne umfangreicher Peripherie) wird sich - nachdem die ersten Computerspiele gespielt sind - fragen, welche Aufgaben denn nun der Rechner übernehmen kann.

Neben der Behandlung sämtlicher Fähigkeiten und Besonderheiten des Schneider CPC, steht die Beantwortung dieser Frage im Mittelpunkt des vorliegenden Buches. Dabei soll der Leser jedoch nicht nur fertige Konzepte übernehmen (dann könnte er auch seinen Computerhändler direkt nach fertiger Software fragen), sondern er sollte anhand der Programmbeschreibungen erkennen, wie man Ideen problematisiert und programmtechnisch umsetzt.

Die einzelnen Programme sind in der Programmiersprache BASIC geschrieben, die ja heute im Microcomputerbereich eine dominierende Rolle spielt. Zwar wird vorausgesetzt, daß der Leser einige BASIC-Grundkenntnisse besitzt, er


sollte aber auch nach intensiver Durcharbeitung des Bedienungs-handbuches zum Schneider CPC in der Lage sein, die verschiedenen Programme zu verstehen, auch wenn dies zugegebenermaßen nicht immer ganz einfach ist. Die Programme sind daher so aufgebaut, daß erkennbar bleibt, was geschieht. Dies ging natürlich gelegentlich zu Lasten der Programmeleganz und der Rechengeschwindigkeit. Der Nutzen dieses Buches erschließt sich für den Leser somit insbesondere dann, wenn er die vorgestellten Programme verändert oder auf seine eigenen Bedürfnisse hin bearbeitet. Sollten dabei scheinbar unüberwindbare Schwierigkeiten auf Sie zu kommen, dann bedenken Sie:

Es ist noch kein Meister vom Himmel gefallen

Die einzelnen Kapitel in diesem Buch werden in der Regel nach folgendem Muster unterteilt:

- a) Vorstellung und Problematisierung einer Idee
- b) Programm
- c) Liste der im Programm verwendeten Variablen
- d) Programmbeschreibung
und ggf.
- e) Programmergebnisse

Die aufgegriffenen Ideen haben programmiertechnische Lösungen unterschiedlichen Schwierigkeitsgrades zur Folge. Es ist jedoch darauf geachtet worden, daß zu Beginn eines neuen Kapitels zuerst die Ideen bzw. Vorschläge mit den einfacheren Programmlösungen vorgestellt werden, um den in BASIC noch nicht ganz "sattelfesten" Leser langsam an den Aufbau komplizierterer Programme heranzuführen. Prinzipiell ist weiterhin zu beachten, daß nicht immer unsinnige oder von der Logik des Programmablaufs her unmögliche Eingaben abgefangen werden. Darauf wird im einzelnen aber noch hingewiesen.



1. Der CPC und was man mit ihm machen kann

Schaut man sich den Markt für Microcomputer an, so stellt man fest, daß mittlerweile eine ganze Reihe vergleichbarer Rechner der gleichen Preisklasse angeboten werden. Nur gelegentlich fällt ein elektronisches Gerät dieser Art wirklich aus dem Rahmen. Der Schneider (egal in welcher Ausführung) gehört sicherlich zu den Hechten (oder ist er der Hecht ?) unter so vielen Karpfen im Teich. Er weist einige bemerkenswerte Eigenschaften auf, die ihresgleichen suchen.

Neben den außerordentlichen Graphik- und Musikeigenschaften ist insbesondere die Definition von Fenstern (windows) zur Festlegung eines Bildschirmteilbereichs als eigen ansprechbares Ein/Ausgabegerät (stream), sowie die Programmierung zeitabhängiger Unterbrechungen (interrupts) zur gleichzeitigen bzw. zeitabhängigen Durchführung unterschiedlicher Handlungen hervorzuheben. Auf diese Besonderheiten wird bei ihrer Anwendung in den folgenden Kapiteln noch näher eingegangen.

Schon in der Grundausstattung ist das Schneider CPC - System eine voll funktionfähige Rechenanlage. Mit ihr lassen sich die grundlegenden Datenverarbeitungsschritte

- Daten- bzw. Informationsaufnahme (über die Tastatur)

- Informations- bzw. Datenspeicherung (intern oder extern)
- Auswerten bzw. Verarbeiten dieser Daten
- Ergebnisausgabe (über den Bildschirm)

ausführen.

Ein Nachteil bei der Benutzung eines Cassettengerätes als externes Speichermedium ist, daß Cassetten nur sequentiell beschrieben bzw. gelesen werden können. Soll zum Beispiel eine bestimmte Datei von einer Cassette gelesen werden, dann muß diese Cassette erst an die betreffende Stelle gespult werden. Dies kann eine unter Umständen recht unangenehme Sucharbeit zur Folge haben. Es ist auf jeden Fall sinnvoll, nicht zuviele Informationen auf einer Cassettenseite zu speichern und sich mit Hilfe des Bandzählwerks (counter) die einzelnen Positionen zu merken.

Bequemer ist da schon der Einsatz eines Diskettenlaufwerks. Neben der größeren Speicherkapazität einer Diskette, ist ein sehr schneller Zugriff auf bestimmte Positionen möglich (die Diskette dreht sich ständig und sehr rasch im Laufwerk). Man spricht daher auch von einem Speicher mit direkter Zugriffsmöglichkeit (direct-access-Speicher). Bezüglich der zu verwendenden BASIC-Anweisungen unterscheidet sich der Einsatz eines Diskettenlaufwerks prinzipiell nicht von der Benutzung eines Cassettenrecorders. Die Programme sind somit voll kompatibel.

Der Nutzen eines Computers erschließt sich für den Benutzer erst dann, wenn er über die geeignete Software verfügt. Die Frage, wie man an gute und für die individuellen Aufgaben geeigneten Programme kommt, läßt sich auf 3 Arten beantworten:

- 1) Fertig-Software kaufen
- 2) Programmlistings abtippen
- 3) Programme selbst entwickeln

Wenn Ihnen die Zeit oder die Erfahrung zur selbständigen Programmierung fehlt, sind Sie auf den Software-Markt angewiesen. Dies gilt insbesondere dann, wenn Sie Ihren neu gekauften Computer direkt mehr oder weniger professionell einsetzen wollen (man denke da z. B. an Textverarbeitung, Finanzbuchhaltung, Kalkulation usw.). Die Auswahl bei Fertigprogrammen ist jedoch sehr groß. Gerade für den CPC sind in der letzten Zeit sehr viele Programme auf den Markt gekommen, die von der Qualität und vom Preis sehr unterschiedlich sind. Vor dem Kauf lohnt es sich auf jeden Fall, intensive Vergleiche anzustellen. Folgende Gesichtspunkte sollten Sie dabei unbedingt beachten:

- Qualität der Benutzeranleitung
- Service der Anbieter
- Berücksichtigung des individuellen Aufgabenbereichs
- Software-Tests in Fachzeitschriften

Das Abtippen von Programmlistings ist eine nicht unproblematische Angelegenheit. Je nachdem, wie unseriös die Herausgeber sind, muß nämlich damit gerechnet werden, daß die Listings mit Fehlern gespickt sind. Da es, wenn die notwendigen Programmierläuterungen fehlen, sehr schwierig ist, sich in die Logik eines unbekanntes Programms hineinzudenken, kann die Fehlersuche und -bereinigung mehr Zeit in Anspruch nehmen als die selbständige Programmentwicklung. Vor dem Abtippen umfangreicher Programmlistings sollten Sie also die jeweilige Quelle prüfen.

Viele Kritiker der privaten Computernutzung bezeichnen das "Computern" als nicht kreative Tätigkeit. In der Tat kann es sehr stupide sein, Tag für Tag mit dem Rechner "nur" zu spielen. Das "Computern" wird aber ohne Zweifel zu einem kreativen Hobby, wenn Programme selbst geschrieben oder angepasst werden. Dem einen oder anderen Computerbesitzer fehlen aber u. U. die Anregungen und Ideen um kreativ mit

seinem Rechner arbeiten zu können. Für diesen Personenkreis dürfte das vorliegende Buch geradezu ideal sein, da doch die verschiedensten Themen und Vorschläge aufgegriffen und programmtechnisch umgesetzt werden. Zu diesen "fertigen" (aber natürlich noch änderbaren) Programmen kommen zu jedem "Ideenblock" noch weitere Vorschläge (Kapitel "Fundgrube"), die in diesem Buch nicht näher behandelt werden konnten, aber vielleicht den Leser zur Programmentwicklung anregen. Aufgrund der Themenvielfalt müßte auf jeden Fall für alle Interessenten etwas "passendes" dabei sein.

2. Nützliche Kleinigkeiten – einfach programmiert

2.1. Vorbemerkung

In diesem ersten eigentlichen Ideenkapitel werden vergleichsweise einfache und überschaubare Programmlösungen vorgestellt. Hingegen sind die Programmerläuterungen – setzt man sie zu den Schwierigkeitsgraden der Programme in Beziehung – recht umfangreich. Dadurch soll erreicht werden, daß der ungeübte Leser langsam an die etwas komplizierteren Programme der folgenden Kapitel herangeführt wird.

Inhaltlich handelt es sich hier – insbesondere bei den ersten drei Programmen – um Problemlösungen, die in größeren und umfangreicheren Programmen als Unterprogramme benutzt werden können. So werden Sie zum Beispiel im weiteren Verlauf dieses Buches häufig auf die Prozentrechnung stoßen, welche zur Lösung kleinerer Probleme nützliche Dienste erweisen kann.

2.2. Die Prozentrechnung

Häufig ist es notwendig, zwei oder mehrere Werte miteinander zu vergleichen. Dabei dient uns die Zahl 100 als Vergleichsbasis.

Folgende Grundbegriffe der Prozentrechnung sind vorab zu klären:

1. Grundwert:

Der Grundwert entspricht dem Ganzen, also 100 Hundertsteln oder 100%. Er ist stets eine benannte Zahl (DM, kg, m oder ähnliches).

2. Prozentwert:

Der Prozentwert ist ein Teil des Grundwertes. Er ist ebenfalls eine benannte Zahl.

3. Prozentsatz:

Der Prozentsatz ist auch ein Teil des Grundwertes. Er wird jedoch in bezug auf 100 angegeben. Prozentsatz und Prozentwert entsprechen einander.

Ein kleines Beispiel soll Ihnen den Umgang mit diesen Begriffen erleichtern:

Sagen wir einmal, Ihr CPC 464 kostet im Computerladen xyz 898,- DM (Grundwert). Nun gehen Sie zu einem anderen Händler, der 4% (100 - Prozentsatz) unter diesem Preis bleiben will. Entweder fragen Sie diesen Händler nach dem genauen Preis (Prozentwert), oder Sie berechnen ihn mit folgender Formel:

$$\begin{aligned} \text{Prozentwert} &= \frac{\text{Grundwert} * \text{Prozentsatz}}{100} \\ &= \frac{898 \text{ DM} * 96}{100} = 862.08 \text{ DM} \end{aligned}$$

Vielleicht sind Sie nun der Meinung, daß der Einsatz eines Computers für so einfache Rechnungen nicht notwendig ist. Aber, Hand aufs Herz, rechnen Sie so etwas im Kopf? Wahrscheinlich werden Sie einen Taschenrechner oder zumindest Papier und Bleistift benötigen. Und dann muß der eine oder andere vielleicht doch noch einmal kurz überlegen, wie so ein Prozentwert berechnet wird. Dazu kommt noch, daß in dem folgenden Programm auch - falls dies gewünscht wird - der unbekannte Grundwert oder der unbekannte Prozentsatz berechnet werden kann. Und da haben sicherlich viele schon nicht mehr die entsprechenden Berechnungsformeln parat, obwohl sie ganz einfach sind:

$$\text{Prozentsatz} = \frac{\text{Prozentwert} * 100}{\text{Grundwert}}$$

$$\text{Grundwert} = \frac{\text{Prozentwert} * 100}{\text{Prozentsatz}}$$

Pogramm:

```
10 REM I1
20 CLS
30 PRINT "I1 - Programm zur Berechnung von"
40 PRINT TAB(6) "Prozentwerten, Grundwerten"
50 PRINT TAB(6) "oder Prozentsaetzen"
60 PRINT:PRINT:PRINT
70 PRINT TAB(11) "Bernd Kowal, 1985"
80 GOSUB 380
90 PRINT TAB(16) "Auswahl":PRINT:PRINT:PRINT:PRINT
100 PRINT "Prozentwert      =          1":PRINT
110 PRINT "Grundwert       =          2":PRINT
120 PRINT "Prozentsatz     =          3":PRINT
130 PRINT:PRINT:PRINT
140 PRINT "Bitte geben Sie fuer den unbekanntem"
150 PRINT "Wert die entsprechende Zahl ein."
```

```

160 PRINT:INPUT"Ihre Wahl ";a
170 IF a<1 OR a>3 THEN 220
180 CLS
190 PRINT"Bitte geben Sie die Werte fuer die"
200 PRINT"bekannten Groessen an.":PRINT:PRINT
210 ON a GOSUB 230,280,330
220 PRINT:PRINT:PRINT:PRINT"Ende der Berechnungen":END
230 INPUT"Grundwert ";gw:PRINT
240 INPUT"Prozentsatz ";ps
250 pw=gw*ps/100
260 PRINT:PRINT:PRINT"Der Prozentwert ist gleich :";pw
270 RETURN
280 INPUT"Prozentwert ";pw:PRINT
290 INPUT"Prozentsatz ";ps
300 gw=pw*100/ps
310 PRINT:PRINT:PRINT"Der Grundwert ist gleich :";gw
320 RETURN
330 INPUT"Prozentwert ";pw:PRINT
340 INPUT"Grundwert ";gw
350 ps=pw*100/gw
360 PRINT:PRINT:PRINT"Der Prozentsatz ist gleich :";ps
370 RETURN
380 REM UP Warten
390 LOCATE 7,25
400 PRINT"Bitte eine Taste druecken"
410 x#=INKEY#:IF x#="" THEN 410
420 CLS:RETURN

```

Variablenliste:

a = Antwortvariable für das Auswahlmenü

gw = Grundwert

ps = Prozentsatz

pw = Prozentwert

x\$ = Hilfsvariable, die den Wert der gedrückten Taste zurückgibt

Da die Variable x\$ in jedem Programm dieses Buches die gleiche Funktion hat, wird sie in den folgenden Kapiteln nicht mehr gesondert aufgeführt.

Programmbeschreibung:

Satz 10 : Kommentar (wird nicht ausgegeben)

Satz 20 : Der Bildschirm wird freigemacht

Satz 30- 70 : Titelausdruck

Satz 80 : Das Unterprogramm "Warten" wird aufgerufen

Satz 90-150 : Ausgabe des Auswahlmenüs

Satz 160 : Ein dem Auswahlmenü entsprechender Wert wird angefordert

Satz 170 : Es wird zum Programmende verzweigt, wenn ein zu großer oder ein zu kleiner Wert eingegeben wird

Satz 180-200 : Der Bildschirm wird gelöscht und es wird ein Kommentar bezüglich der noch zu machenden Eingaben ausgegeben

Satz 210 : In Abhängigkeit von der Eingabe zur Menüauswahl erfolgt ein Sprung in das entsprechende Unterprogramm

Satz 220 : Programmende

Satz 230-270 : Unterprogramm zur Berechnung des Prozentwertes

230-240 : Für den bekannten Grundwert und für den bekannten Prozentsatz werden Werte angefordert

250 : Der Prozentwert wird berechnet

260 : Das Ergebnis wird ausgegeben

270 : Rücksprung aus dem Unterprogramm

Satz 280-320 : Unterprogramm zur Berechnung des Grundwertes. Die Vorgehensweise entspricht der der Sätze 230-270.

Satz 330-370 : Unterprogramm zur Berechnung des Prozentsatzes. Es wird entsprechend der beiden vorangegangenen Unterprogramme verfahren

Satz 380-420 : Unterprogramm zum Warten

380 : Kommentar

390 : Der Cursor wird auf die angegebene Position gebracht

400 : Ein Kommentar wird ausgegeben

410 : Unterbrechung des Programmablaufs bis eine Taste gedrückt wird

420 : Der Bildschirm wird freigemacht und es erfolgt ein Rücksprung aus dem Unterprogramm

Programmergebnisse:

Starten wir das Programm mit dem Kommando RUN, so erscheint auf dem Bildschirm die Programmüberschrift. Drücken wir dann eine Taste, wird das Auswahlmenü ausgegeben und eine Eingabe für den unbekanntenen Wert wird angefordert. Wenn wir z. B. den Wert 1 eingeben, wird ein Kommentar ausgegeben und es wird eine Zahl für den Grundwert angefordert. Geben wir beispielsweise den Wert 898 ein, so wird noch ein Wert für den Prozentsatz verlangt. Nach Eingabe der Zahl 96 antwortet der Rechner mit:

Der Prozentwert ist gleich : 862.08

Ende der Berechnungen
Ready

Es sei noch einmal erwähnt, daß das Programm so allgemein gehalten ist, daß es für beliebige Eingaben den Prozentwert, den Grundwert oder den Prozentsatz berechnet. Die Problematik dieses Programmvorschlags erfordert eigentlich nicht eine Programmbeschreibung, die so ausführlich ist wie hier praktiziert. Es sollte auch vielmehr gezeigt werden, wie in den einzelnen Kapiteln prinzipiell vorgegangen wird. Im folgenden werden daher relativ einfache Programme knapper kommentiert werden.

2.3. Der kleinste, der größte und der Durchschnittswert

Stellen Sie sich vor, Sie sind Vorsitzender eines Vereins und wollen etwas zur Altersstruktur der Vereinsmitglieder wissen. Was liegt näher, als nach dem Alter des jüngsten und des ältesten Mitglieds, sowie nach dem Durchschnittsalter (gemeint ist das arithmetische Mittel) zu fragen.

Während Sie früher Papier, Bleistift und etwas Zeit benötigten um diese Fragen zu beantworten, ist es heutzutage ausreichend, Ihren CPC richtig einzusetzen. Sie brauchen nur ein entsprechendes Programm. Und das finden Sie direkt auf der nächsten Seite.

Natürlich läuft das Programm nicht nur für Altersangaben, sondern für beliebige Zahlen. Sie können es also auch benutzen, wenn Sie nicht Vorsitzender eines Vereins sind.

Falls Sie einen größeren Datenbestand haben, ist es u. U. sinnvoller, die einzelnen Werte nicht mit dem INPUT-Statement, sondern mit den Statements DATA und READ einzugeben (siehe Kapitel 2.4), da bei der Eingabe mehrerer Werte ziemlich schnell Fehler beim Eintippen auftreten.

Programm:

```
10 REM I2
20 CLS
30 PRINT"I2 - Programm zum Aufsuchen des"
40 PRINT TAB(6)"kleinsten und des groessten Wertes"
50 PRINT TAB(6)"eines Datenbestandes, sowie"
60 PRINT TAB(6)"zur Berechnung des Durchschnitts-"
70 PRINT TAB(6)"wertes"
80 PRINT:PRINT:PRINT
90 PRINT TAB(11)"Bernd Kowal, 1985"
100 GOSUB 280
110 INPUT"Wieviele Daten ";n
120 IF n=0 THEN 270
130 DIM a(n)
140 PRINT:PRINT"Dateneingabe :":PRINT
150 FOR i=1 TO n
160 PRINT i;".ter Wert";:INPUT a(i)
170 s=s+a(i)
180 NEXT i:CLS
190 d=s/n:k=a(1):g=a(1)
200 FOR i=2 TO n
210 k=MIN(k,a(i))
220 g=MAX(g,a(i))
230 NEXT i
240 PRINT:PRINT"Kleinster Wert      =";k
250 PRINT:PRINT"Groesster Wert      =";g
260 PRINT:PRINT"Durchschnittswert  =";d
270 PRINT:PRINT:PRINT"Programmende":END
280 REM UP Warten
290 LOCATE 7,25
300 PRINT"Bitte eine Taste druecken"
310 x#=INKEY#:IF x#="" THEN 310
320 CLS:RETURN
```

Variablenliste:

$a(i)$ = die eingegebenen Werte

d = Durchschnittswert

g = größter Wert

i = Laufindex

k = kleinster Wert

n = Anzahl der eingegebenen Werte

s = Summe der eingegebenen Werte

Programmbeschreibung:

Satz 10-100 : Überschrift

Satz 110 : Ein Wert für die Anzahl der einzugebenden
Zahlen wird angefordert

Satz 120 : Es wird zum Programmende verzweigt, wenn
keine Zahlen eingegeben werden sollen

Satz 130 : Für das Feld $a(n)$ wird Speicherplatz be-
reitetgestellt

Satz 140-180 : Eingabe der Daten. Im Feld s werden die
einzelnen Werte direkt aufsummiert.

Satz 190 : Der Durchschnittswert wird berechnet und für die Variablen k und g wird als Startgröße der zuerst eingegebene Wert $a(1)$ angenommen

Satz 200-230 : Alle eingegebenen Werte werden mit dem kleinsten - (Satz 210) und dem größten Wert (Satz 220) verglichen. Ist ein Wert kleiner bzw. größer als der bisherige kleinste bzw. größte Wert, so ändert sich der Wert der Variablen k bzw. g entsprechend. Die Programmschleife beginnt erst bei dem Wert $i=2$, da in Satz 190 die Variablen k und g mit dem ersten Wert $a(1)$ belegt worden sind.

Satz 240-260 : Ergebnisausdruck

Satz 270 : Programmende

Satz 280-320 : Unterprogramm "Warten"

Programmergebnisse:

Geben wir nach dem Programmstart 4 Werte ein, zum Beispiel die Zahlen 19, 63, 79 und 56, so erhalten wir folgendes Ergebnis:

Kleinster Wert = 19

Groesster Wert = 79

Durchschnittswert = 54.25

Programmende
Ready

Ein derartiges Ergebnis können wir natürlich auch noch interpretieren. Es ist nämlich auffallend, daß drei Werte oberhalb, aber nur ein Wert unterhalb des Durchschnittswertes liegen, und daß der kleinste Wert viel weiter entfernt ist vom Durchschnittswert als der größte Wert. Die Ursache hierfür ist in der ungleichmäßigen Verteilung der untersuchten Werte über das gesamte Zahlenspektrum hinweg zu sehen. Auf unseren Vereinsvorsitzenden bezogen hieße das, daß zwar auch die Jugend im Verein vertreten ist, sonst aber die älteren Damen und Herren eindeutig dominieren.

2.4. Die alphabetische Ordnung

Bleiben wir bei unserem Vereinsvorsitzenden. Nehmen wir einmal an, Sie als Vorsitzender wollten die Mitgliederkartei in Ordnung bringen. Dazu gehört sicherlich das alphabetische Sortieren aller Namen. Dies kann bei einem umfangreichen Mitgliederbestand schon recht mühsam sein. Aber richtig unangenehm wäre eine derartige Sortierarbeit, wenn zum Beispiel die Namen in Telefonbüchern "von Hand" sortiert werden müßten.

Recht unproblematisch ist diese Arbeit mit Computerunterstützung. Das vorliegende Programm sortiert eine beliebige Anzahl von Wörtern oder allgemein von Strings. Die Daten werden hier im Gegensatz zum letzten Programm mit Hilfe des DATA-Statements eingegeben. Wollen Sie also andere Daten benutzen, so müssen Sie die entsprechenden DATA-Statements umändern und für die Variable n eine entsprechende Zahl eingeben.

Zu beachten ist, daß für die Daten nur Großbuchstaben verwendet wurden. Eine gemischte Verwendung von großen und kleinen Buchstaben führt aufgrund einer Rechnervoreinstellung immer dazu, daß Großbuchstaben als die "kleineren" betrachtet werden. So ist z. B. die Bedingung $C < c$ immer erfüllt. Mit Hilfe der Funktionen LOWER\$ bzw. UPPER\$ ist jedoch eine Umwandlung von Großbuchstaben in Kleinbuchstaben bzw. umgekehrt möglich, so daß auch Groß- und Kleinbuchstaben gemischt verwendet werden könnten.

Das vorliegende Programm läßt sich auch ohne Probleme umändern, wenn Sie keine Strings, sondern Zahlen sortieren möchten. Sie müssen dann nur die Variablen w\$ und hi\$ ändern (aus w\$ wird w und aus hi\$ wird hi) und die entsprechenden Daten bereitstellen.

Programm:

```
10 REM I3
20 CLS
30 PRINT"I3 - Programm zum alphabetischen"
40 PRINT TAB(6)"Sortieren von Woertern"
50 PRINT:PRINT:PRINT
60 PRINT TAB(11)"Bernd Kowal, 1985"
70 GOSUB 260
80 PRINT"Wieviele Woerter werden im Programm-"
90 INPUT"ablauf eingelesen ";n
100 IF n=0 THEN 250
110 CLS:DIM w$(n)
120 REM in diesem Programm ist n=7
130 DATA DER,RICHTIGE,COMPUTER,FUER,SIE
140 DATA DER,SCHNEIDER CPC 464
150 FOR i=1 TO n
160 READ w$(i):NEXT i
170 FOR i=1 TO n-1
180 FOR j=i+1 TO n
190 IF w$(i)<=w$(j) GOTO 210
200 hi#=w$(j):w$(j)=w$(i):w$(i)=hi#
210 NEXT j
220 PRINT w$(i)
230 NEXT i
240 PRINT w$(n)
250 END
260 REM UP Warten
270 LOCATE 7,25
280 PRINT"Bitte eine Taste druecken"
290 x#=INKEY#:IF x#="" THEN 290
300 CLS:RETURN
```

Variablenliste:

hi\$ = Hilfsfeld

i = Laufindex

j = Laufindex

n = Anzahl der String-Variablen

w\$(i) = die Wörter bzw. Strings

Programmbeschreibung:

Satz 10- 70 : Überschrift

Satz 80- 90 : Eingabe der Stringanzahl

Satz 100 : Verzweigung zum Programmende, wenn keine
Strings eingelesen werden sollen

Satz 110 : Der Bildschirm wird gelöscht und es wird
Speicherplatz reserviert

Satz 120 : Kommentar

Satz 130-160 : Die Daten werden bereitgestellt und einge-
lesen

Satz 170-240 : Die Wörter werden alphabetisch sortiert und
das Ergebnis wird ausgegeben.
Sortieren heißt hier, daß je zwei Wörter
miteinander verglichen werden. Zunächst

wird das zuerst eingelesene Wort $w(i)$, für $i=1$, mit allen übrigen Wörtern $w(j)$, für $j=2\dots n$, verglichen. Folgt dabei $w(1)$ im Alphabet einem anderen Wort $w(j)$, so werden die beiden verglichenen Wörter vertauscht (Satz 200). Nach diesem Durchlauf (Satz 180-210) steht dann an erster Stelle ($w(1)$) das Wort, das im Alphabet vorne steht.

Im folgenden wird dann die Laufvariable i jeweils um 1 erhöht, d. h., daß zuerst das an zweiter Stelle stehende Wort $w(2)$ mit allen übrigen Wörtern (nur nicht mit $w(1)$) verglichen und ggf. vertauscht wird, anschließend das an dritter Stelle stehende Wort usw.

Insgesamt werden somit alle nur möglichen Vergleiche angestellt.

Satz 250 : Programmende

Satz 260-300 : Unterprogramm "Warten"

Programmergebnisse:

COMPUTER

DER

DER

FUER

RICHTIGE

SCHNEIDER CPC 464

SIE

Ein richtiger Satz alphabetisch geordnet !

2.5. Spielen Sie Lotto ?

Sie sind 36 Jahre alt, wurden am 12.9.48 geboren und vergessen am 20.3 immer Ihren Hochzeitstag? Dann werden Sie vielleicht mit folgenden Zahlen Woche für Woche Ihr Glück beim Lottospiel versuchen:

36 , 12 , 9 , 48 , 20 , 3

Diese Zahlen können Sie jetzt vergessen (nur in dieser Anordnung natürlich). Wahrscheinlich haben sie Ihnen auch nicht viel Glück gebracht. Ihr CPC liefert Ihnen jetzt Woche für Woche 6 Glückszahlen. Ob Sie mit diesen "Computerzahlen" mehr Glück haben werden als mit Ihrer alten Kombination ist allerdings fraglich. Die Wahrscheinlichkeit für 6 Richtige bleibt nämlich die gleiche. Um Ihnen da jegliche Hoffnungen zu nehmen, sagt Ihnen der Computer direkt, wieviele verschiedene Möglichkeiten es gibt, 6 aus 49 Zahlen auszuwählen.

Zu dem folgenden Programm sind einige mathematische Erläuterungen erforderlich:

Es ist zwar recht unproblematisch 6 Zufallszahlen zwischen 1 und 49 zu erzeugen (dies geschieht mit Hilfe der Funktion RND), jedoch ist die Bestimmung der Anzahl der Möglichkeiten 6 bzw. k Elemente aus 49 bzw. n Elementen auszuwählen komplizierter, da wir dazu die mathematische Größe des Binomialkoeffizienten $\binom{n}{k}$ benötigen (lies n über k).

Es gilt folgende Berechnungsformel:

$$\binom{n}{k} = \frac{n!}{(n-k)! * k!}$$

Für den mathematisch interessierten Leser sei an dieser Stelle noch erwähnt, daß der Binomialkoeffizient den Faktoren entspricht, die sich bei der Auflösung der sog. Binome ergeben. So ist:

$$(a+b)^n = a^n + \binom{n}{1} a^{n-1} b + \binom{n}{2} a^{n-2} b^2 + \dots + \binom{n}{n-1} a b^{n-1} + b^n$$

Aber zurück zur Problematik des vorliegenden Programms. Da mit den Werten n=49 und k=6 bei der Fakultätenberechnung in der obigen Formel Zahlenwerte erreicht werden, die größer als der größte darstellbare Wert sind und die Fehlermeldung "Overflow" zur Folge hätten, ist der Quotient der Formel im Programm gekürzt worden:

$$\begin{aligned} \binom{n}{k} &= \frac{1 * 2 * 3 * 4 * \dots * n}{1 * 2 * 3 * \dots * (n-k) * 1 * 2 * 3 * \dots * k} \\ &= \frac{(k+1) * (k+2) * (k+3) * \dots * n}{1 * 2 * 3 * 4 * \dots * (n-k)} \end{aligned}$$

Berücksichtigt man, daß $\binom{n}{k} = \binom{n}{n-k}$ - wie man leicht aus der obigen Berechnungsformel ersehen kann - , so ergibt sich für $n=49$ und $k=6$ (bzw. $k=43$):

$$\binom{49}{6} = \frac{49 \cdot 48 \cdot 47 \cdot 46 \cdot 45 \cdot 44}{6 \cdot 5 \cdot 4 \cdot 3 \cdot 2 \cdot 1}$$

Diese mathematischen Erläuterungen sollten in erster Linie ein Nachvollziehen der einzelnen Programmschritte erleichtern. Das Programm selbst ist so gehalten, daß eine Änderung des Satzes 190 die Berechnung beliebiger Binomialkoeffizienten $\binom{n}{k}$ ermöglicht. Der Ergebnisausdruck ist dann jedoch entsprechend zu ändern.

Programm:

```

10 REM I4
20 CLS
30 PRINT"I4 - Programm zur Erzeugung von 6 Zu-"
40 PRINT TAB(6)"fallszahlen zwischen 1 und 49,"
50 PRINT TAB(6)"sowie zur Bestimmung der Anzahl"
60 PRINT TAB(6)"der Moeglichkeiten 6 aus 49"
70 PRINT TAB(6)"Zahlen auszuwaehlen (Lotto)"
80 PRINT:PRINT:PRINT
90 PRINT TAB(11) "Bernd Kowal, 1925"
100 GOSUB 340
110 PRINT"6 Zufallszahlen zwischen 1 und 49 :":PRINT
120 FOR i=1 TO 6
130 x(i)=INT(RND*49+1)
140 FOR j=1 TO i-1
150 IF x(i)=x(j) THEN 130

```

```

140 NEXT j
170 PRINT i; ".te Zahl :";x(i)
180 NEXT i:PRINT:PRINT
190 n=49:k=6
200 IF k>n THEN z=0:ne=1:GOTO 240
210 IF n-k>k THEN k=n-k
220 aw=n:ew=k:GOSUB 280:z=f
230 aw=n-k:ew=0:GOSUB 280:ne=f
240 PRINT"Es gibt";z/ne;"verschiedene"
250 PRINT"Moeglichkeiten 6 aus 49 Zahlen"
260 PRINT"auszuwaehlen !"
270 PRINT:PRINT:PRINT"Ende der Berechnungen":END
280 REM UP Fakultaeten
290 f=1
300 FOR i=aw TO ew+1 STEP -1
310 f=f*i
320 NEXT i
330 RETURN
340 REM UP Warten
350 LOCATE 7,25
360 PRINT"Bitte eine Taste druecken !"
370 x#=INKEY#
380 IF x#="" THEN 370
390 CLS:RETURN

```

Variablenliste:

aw = Schleifenanfangswert

ew = Schleifenendwert

f = Produkt im Unterprogramm "Fakultäten"

i = Laufindex und Faktor im Unterprogramm "Fakultäten"

j = Laufindex

k = Anzahl der auszuwählenden Elemente

n = Zahl der Elemente

ne = Nenner in der Berechnungsformel für $\binom{n}{k}$

x(i) = Zufallszahlen zwischen 1 und 49

z = Zähler in der Berechnungsformel für $\binom{n}{k}$

Programmbeschreibung:

Satz 10-100 : Überschrift

Satz 110-180 : 6 Zufallszahlen zwischen 1 und 49 werden bestimmt und ausgegeben.
In den Sätzen 140-160 wird jede erzeugte Zufallszahl mit den schon ausgegebenen Zahlen verglichen und im Falle einer vorliegenden Gleichheit wird, ohne Erhöhung der Laufvariablen, eine neue Zufallszahl x(i) gebildet.

Satz 190 : Die Variablen n und k werden mit Werten belegt

Satz 200 : Ist $k > n$, so ist $\binom{n}{k} = 0$. In diesem Fall erfolgt eine Verzweigung zur Ergebnisausgabe.

Satz 210 : Wenn $n - k > k$ ist, wird k gleich n-k gesetzt, um die Berechnung zu vereinfachen (siehe oben)

Satz 220-230 : Zähler und Nenner der gekürzten Formel werden berechnet

Satz 240-260 : Ergebnisausgabe

Satz 270 : Programmende

Satz 280-330 : Unterprogramm "Fakultäten"
Fakultäten werden nur berechnet, wenn die Variable ew gleich 0 ist. Ansonsten werden beliebige Produkte gebildet.

Satz 340-390 : Unterprogramm "Warten"

Programmergebnisse:

Die Ergebnisausgabe könnte folgendes Aussehen haben:

6 Zufallszahlen zwischen 1 und 49:

1 .te Zahl : 32

2 .te Zahl : 5

3 .te Zahl : 39

4 .te Zahl : 41

5 .te Zahl : 49

6 .te Zahl : 7

Es gibt 13983816 verschiedene Möglichkeiten
6 aus 49 Zahlen auszuwählen !

Da kann man eigentlich nur noch viel Glück wünschen !

2.6. Schaufensterwerbung

Wenn Sie ein Geschäft haben und gelegentlich bereit sind auf Ihren CPC zu verzichten, dann können Sie ihn auch ins Schaufenster stellen. Ihre Ehefrau wird dies wahrscheinlich sehr begrüßen. (Sie müssen ja auch nicht Ihre gesamte Freizeit am Rechner verbringen oder ...)

Natürlich stellen Sie Ihren CPC nicht einfach so ins Schaufenster um den Leuten zu zeigen, daß Sie einen Computer besitzen, sondern Sie lassen ein spezielles Werbeprogramm ablaufen. Optische Werbeträger sind schließlich immer ein besonderes Lockmittel.

Die Gestaltung eines Programms zur Schaufensterwerbung muß interessant und auffallend sein, da eine zufällig am Schaufenster vorbeigehende Person dazu angehalten werden soll, sich die Werbeinformationen genauer anzusehen. Um dies zu erreichen ist es ratsam mit Farben und mit Laufschrift zu arbeiten. Gute Effekte kann man auch mit einer Mischung aus Graphik und Textzeilen erzielen. Dazu dienen die Anweisungen TAG und TAGOFF. Denkbar wäre es zum Beispiel, mit Hilfe des Programms in Kapitel 8.2 eine interessante Graphik zu erstellen und diese dann in einem Werbeprogramm einzulesen, Das gesamte Programm in Kapitel 8.2 könnte dann der Einfachheit halber als Unterprogramm dienen.

Im Beispielprogramm zu diesem Kapitel wird hingegen eine einfachere Lösung vorgestellt. Nach der großformatigen Ausgabe der Zahlen 1,2 und 3 (rot auf weiß), erscheint dreimal der Werbetext auf dem Bildschirm, zuerst auch rot auf weiß, dann blau-rot blinkend und schließlich blau-rot blinkend auf weiß-gelb blinkendem Hintergrund. Die Zahlen 1,2 und 3 werden jeweils analog zu den Rastern definiert, welche in Ihrem Bedienungshandbuch aufgeführt sind. Während dort die Zahlen, wie auch jedes andere Zeichen, mit einem Punktmuster dargestellt werden, setzen sich die großformatigen Zahlen hier aus einzelnen Rechtecken zusammen. Diese Rechtecke werden mit Hilfe der WINDOW-Anweisung definiert und mit der für den Hintergrund festgelegten Farbe ausgefüllt. Die zeitliche Steuerung des Programmablaufs erfolgt mit den Anweisungen TIME und AFTER. Während die großformatigen Zahlen jeweils eine halbe Sekunde auf dem Bildschirm erscheinen, bleiben dem Betrachter jeweils fünf Sekunden für den Werbetext. Weiterhin kann der Benutzer vorgeben, wie lange das Programm insgesamt laufen soll. Zu beachten ist, daß nach dem Aufruf eines Unterprogramms mit dem AFTER-Statement, der Programmablauf solange fortgesetzt wird, bis die angegebene Zeit abgelaufen ist.

Der Werbetext des vorliegenden Programms kann natürlich beliebig geändert werden. Auch ist es möglich nicht nur die letzte Zeile "laufen" zu lassen. Es könnte z. B. durchaus interessant sein, mehrere Textzeilen nacheinander in Bewegung zu setzen. Auf jeden Fall lohnt es sich, verschiedene Varianten durchzuspielen.

Programm:

```
10 REM IS
20 CLS
30 PRINT "IS - Programm zur Schaufensterwerbung"
40 PRINT:PRINT:PRINT:PRINT:PRINT
50 PRINT TAB(9) "Bernd Kowal, 1985"
60 GOSUB 870
70 PRINT"Wie lange soll der Werbespot laufen ?"
80 PRINT:INPUT"Stunden ";st
90 PRINT:INPUT"Minuten ";mi
100 PRINT:INPUT"Sekunden ";se
110 ze=st*3600+mi*60+se+TIME/300
120 INK 2,26:PAPER 2:BORDER 26
130 INK 1,0:INK 0,26:INK 15,1,3:INK 14,26,24
140 CLS
150 FOR i=1 TO 7
160 PAPER #i,3
170 NEXT i
180 WHILE TIME/300<ze
190 GOSUB 270
200 AFTER 25,1 GOSUB 340
210 AFTER 50,2 GOSUB 450
220 AFTER 75,3 GOSUB 560
230 z1=TIME
240 IF TIME<z1+1000 THEN 240
250 WEND
260 MODE 1:END
270 REM Die Zahl 1
280 MODE 1
290 WINDOW #1,12,15,6,8
300 WINDOW #2,16,23,3,20
310 WINDOW #3,8,31,21,23
320 FOR i=1 TO 3:CLS #i:NEXT i
330 RETURN
340 REM Die Zahl 2
350 CLS
360 WINDOW #1,8,15,6,8
370 WINDOW #2,12,27,3,5
380 WINDOW #3,24,31,6,11
390 WINDOW #4,12,27,12,14
400 WINDOW #5,8,15,15,20
410 WINDOW #6,8,31,21,23
420 WINDOW #7,24,31,18,20
430 FOR i=1 TO 7:CLS #i:NEXT i
440 RETURN
450 REM Die Zahl 3
```

```

460 CLS
470 WINDOW #1,8,11,6,8
480 WINDOW #2,12,27,3,5
490 WINDOW #3,24,31,6,11
500 WINDOW #4,16,27,12,14
510 WINDOW #5,24,31,15,20
520 WINDOW #6,12,27,21,23
530 WINDOW #7,8,15,18,20
540 FOR i=1 TO 7:CLS #i:NEXT i
550 RETURN
560 REM Sonderangebot
570 MODE 0
580 z2=TIME
590 GOSUB 660
600 IF TIME<z2+1500 THEN 600
610 z3=TIME:FEN 15:CLS:GOSUB 660
620 IF TIME<z3+1500 THEN 620
630 z4=TIME:PAPER 14:BORDER 26,24:CLS:GOSUB 660
640 IF TIME<z4+1500 THEN 640
650 FEN 1:PAPER 2:BORDER 26:RETURN
660 PRINT"  Sonderangebot":PRINT:PRINT:PRINT
670 PRINT"  Schneider CPC 464":PRINT:PRINT:PRINT
680 PRINT"  mit Monitor":PRINT
690 PRINT"  und":PRINT
700 PRINT"  Datenrecorder":PRINT:PRINT:PRINT
710 PRINT"  zum Preis von :":PRINT:PRINT
720 PRINT TAB(7)"898,-DM"
730 a$=" ist das nicht toll "
740 z5=TIME
750 IF TIME<z5+300 THEN 750
760 FOR i=1 TO 20
770 b$=RIGHT$(a$,i)
780 LOCATE 1,23
790 PRINT b$;
800 NEXT i
810 FOR i=1 TO 20
820 b$=LEFT$(a$,20-i)
830 LOCATE i,23
840 PRINT b$;
850 NEXT i
860 RETURN
870 REM Unterprogramm Warten
880 LOCATE 5,25
890 PRINT"Bitte eine Taste druecken !"
900 x$=INKEY$
910 IF x$="" THEN 900
920 CLS:RETURN

```

Variablenliste:

a\$ = String, in dem der "laufende" Text abgelegt wird

b\$ = der "bearbeitete" String a\$

i = Laufvariable

mi = Minuten

se = Sekunden

st = Stunden

z1 }
z2 }
z3 } = Zeit für eine zeitlich begrenzte Warteschleife
z4 }
z5 }

ze = Zeitbegrenzung in Sekunden für den gesamten Programmablauf

Programmbeschreibung:

Satz 10- 60 : Überschrift

Satz 70-100 : Eingabe der Dauer des Programmablaufs

Satz 110 : Berechnung der Zeitbegrenzung in Sekunden für den gesamten Programmablauf. Die Funktion TIME gibt dabei die Zeit zurück, die seit dem Einschalten des Gerätes vergangen ist (Einheit: 1/300 einer Sekunde).

Satz 120 : Die Hintergrundfarbe und die Randfarbe wird leuchtendweiß

Satz 130 : PEN- bzw. PAPER-Nummern werden Farben zugeordnet

Satz 140 : Der Bildschirm wird freigemacht

Satz 150-170 : Die Hintergrundfarbe für die noch zu definierenden Bildschirmbereiche wird festgelegt (hellrot)

Satz 180-250 : Aufruf der Unterprogramme

Solange die Zeitbegrenzung nicht erreicht ist, werden die Statements in den Sätzen 190-240 ausgeführt.

190-240 : Nach den vorgegebenen Zeiten werden die einzelnen Unterprogramme aufgerufen. Aufgrund der Warteschleifen im Unterprogramm Sonderangebot, kann in Satz 190 auf die AFTER-Anweisung verzichtet werden.

230-240 : Zeitabhängige Warteschleife, die verhindert, daß die Sätze 190-220 abgearbeitet werden, bevor die Unterprogramme aufgerufen sind

Satz 260 : Der Bildschirmmodus wird geändert (40 Zeichen pro Zeile), da im Unterprogramm Sonderangebot mit 20 Zeichen pro Zeile gear-

beitet wird. Gleichzeitig wird dabei der Bildschirm freigemacht und das Programmende ist erreicht.

Satz 270-330 : Unterprogramm zur großformatigen Ausgabe der Zahl 1

270-280 : Kommentar und Änderung des Bildschirmmodus

290-310 : Es werden rechteckige Bildschirmfenster definiert die zusammengesetzt die Zahl 1 darstellen

320 : Die definierten Bildschirmfenster werden mit der Farbe hellrot ausgefüllt (vgl. Satz 150-170)

330 : Rücksprung aus dem Unterprogramm

Satz 340-550 : Unterprogramme zur großformatigen Darstellung der Zahlen 2 und 3. Die Vorgehensweise entspricht der des letzten Unterprogramms.

Satz 560-860 : Unterprogramm "Sonderangebot" zur Ausgabe des Werbetextes

560-570 : Kommentar und Umschaltung auf den 80-Zeichen-Modus

580-650 : In Abhängigkeit von den Zeitbegrenzungsvariablen z_2 , z_3 und z_4 erfolgt dreimal ein Sprung in das Unterprogramm Satz 660-860 zur Textausgabe. Bei der zweiten und dritten Verzweigung in Satz 610 bzw. 630 wird die Farbe des Hintergrundes bzw. des Schreibstiftes und des Hintergrundes (inklusive Rand) geändert. In Satz 650 wird zum Hauptprogramm zurück verzweigt.


660-860 : Unterprogramm zur Ausgabe des Werbetextes. Nach der Ausgabe des Haupttextes (Satz 660-720) und der Belegung der Variablen a\$ (Satz 730), folgt eine zeitabhängige Warteschleife (Satz 740-750). Anschließend wird in den Sätzen 760-850 der Inhalt des Strings a\$ bearbeitet und "laufend" ausgegeben.

Satz 870-920 : Unterprogramm "Warten"

2.7. Ideenfundgrube

- Berechnung des Argumentes einer neuen Mehrheit beim einfachen Dreisatz mit direktem Verhältnis
- Tabellen und Statistiken erstellen
- Darstellung von Kurven und Diagrammen (Kuchen, Balken usw.)
- Zeichnen einer beliebigen Funktion
- Biorhythmus, biorhythmische Partnerschaftsanalyse
- Astrologie und Horoskop
- Persönlichkeitstests oder IQ-Tests
- Fragebogenauswertung
- Adreßverwaltung, Mitgliederverwaltung
- Stichwortkartei

- Vermittlungen jeglicher Natur
- Datenbanken für beliebige Anwendungen
- Schulverwaltung: Kurswahl, Kurslisten
- Kalendererstellung
- Kursus fürs Maschineschreiben
- Sprachsynthesizer
- Darstellung von Schaltplänen
- und noch vieles mehr



3. Auto und Computer

3.1. Vorbemerkung

Gehören Sie auch zu den vielen Menschen in unserem Land die ein Auto besitzen? Wenn nicht, blättern Sie nicht direkt weiter. Die Programme in diesem Kapitel sind recht interessant. Einmal wird die interaktive Nutzung Ihres Cassetten- bzw. Diskettengerätes vorgestellt und zum anderen können Sie das Programm zur Fahrtstreckenoptimierung auch dann einsetzen, wenn Sie beispielsweise eine Fahrradtour planen.

Der Begriff "interaktiv" sei hier kurz erläutert. Er besagt, daß vom Programm aus auf den externen Speicher Cassetten- bzw. Diskettenlaufwerk zugegriffen wird, um ihn zu beschreiben oder um von ihm zu lesen. Sie können also Informationen langfristig sichern und sich diese Informationen jederzeit ansehen bzw. erweitern. Im weiteren Verlauf dieses Buches werden Sie noch einige Programme finden, die auf diese Art und Weise das Cassetten- bzw. Diskettengerät nutzen.

3.2. Benzinverbrauch - langfristig kontrolliert

In dem Programm zu diesem Kapitel bestehen die einzugebenden Informationen aus dem Benzinverbrauch Ihres Autos und dem dazugehörigen Datum der Messung. Dabei müssen Sie natürlich nicht Ihren Verbrauch auf 100 Km selbst berechnen. Selbstverständlich reicht die Angabe der gefahrenen Kilometer und der absolut verbrauchten Kraftstoffmenge. Die Umrechnung auf verbrauchte Liter pro 100 Kilometer erfolgt im Programmablauf. Und falls Sie vergessen haben sollten wie eine solche Rechnung aussieht, hier die Formel:

$$\text{Verbrauch auf 100 Km} = \frac{\text{absoluter Verbrauch} * 100}{\text{gefahrte Kilometer}}$$

Das Ergebnis dieser Berechnung (natürlich ist es auch möglich mehrere Berechnungen hintereinander auszuführen) können Sie dann, neben dem dazugehörigen Datum, mit Hilfe Ihres Cassetten- bzw. Diskettengerätes sichern. Sie können natürlich auch gespeicherte Informationen einlesen und diese Informationsreihe (genauer: Datei) ergänzen. In einem weiteren Schritt besteht dann noch die Möglichkeit der Ausgabe sämtlicher Informationen, um einen u. U. vorliegenden "Verbrauchstrend" erkennen zu können. Dabei wird dann auch der Durchschnittsverbrauch berechnet und das entsprechende Ergebnis wird ausgegeben.

Ein Wort noch zum Programmablauf:

Die Auswahlmöglichkeiten für den Programmbenutzer erfolgen in der Regel durch ein sog. Auswahlmenü. In diesem Programm sind die Wahlmöglichkeiten des Benutzers anders gelöst und zwar durch eine Anzahl von Fragen die mit "j" (für ja) oder "n" (für nein) beantwortet werden müssen.

Es ist dabei zu beachten, daß ein einmal eingegebenes "n" nicht mehr rückgängig gemacht werden kann (es sei denn man unterbricht das Programm durch zweimaliges Drücken der ESC-Taste und benutzt dann die Anweisung GOTO als Kommando, damit die Daten nicht verloren gehen). Ein neuer Programmstart hat ja bekanntlich einen Verlust der vorher eingegebenen Daten zur Folge.

Eine derartige Abfragetechnik scheint auf den ersten Blick weniger komfortabel zu sein als die Menütechnik. Sie hat jedoch den einen Vorteil, daß der Benutzer wichtige Programmschritte - wie zum Beispiel das Sichern von Daten - weniger leicht vergißt, da er auf die jeweiligen Fragen konkret mit "j" oder "n" antworten muß, um den Programmablauf fortzusetzen.

Programm:

```
10 REM I6
20 CLS
30 PRINT"I6 - Programm zur langfristigen"
40 PRINT TAB(6)"Kontrolle Ihres Benzinverbrauchs"
50 PRINT:PRINT:PRINT
60 PRINT TAB(11)"Bernd Kowal, 1985"
70 GOSUB 640
80 DIM v(1000),ta(1000),mo(1000),ja(1000)
90 PRINT"Sollen gespeicherte Daten"
100 INPUT"eingelassen werden (j/n) ";a#:PRINT
110 IF a#="n" THEN 180
120 IF a#<>"j" THEN 90
130 OPENIN "Benzinv.":i=1
140 INPUT #9,v(i),ta(i),mo(i),ja(i)
150 IF EOF=-1 THEN 170
160 i=i+1:GOTO 140
170 CLOSEIN
180 CLS:PRINT"Sollen neue Werte eingegeben"
190 INPUT"werden (j/n) ";a#:PRINT:PRINT
```

```

200 IF a#="n" THEN 370
210 IF a#<>"j" THEN 180
220 i=i+1
230 INPUT"Gefahrene Kilometer      ";gkm:PRINT
240 INPUT"Benzinverbrauch in Liter ";bv:PRINT
250 PRINT:PRINT"Datum ":PRINT
260 INPUT"Tag      ";ta(i)
270 INPUT"Monat    ";mo(i)
280 INPUT"Jahr     ";ja(i):PRINT
290 PRINT:v(i)=bv*100/gkm
300 PRINT"Der Verbrauch auf 100 Kilometer ":PRINT
310 PRINT,USING "##.##";v(i);PRINT" Liter":PRINT
320 PRINT:PRINT"Sollen weitere Werte eingegeben"
330 INPUT"werden (j/n) ";a#
340 IF a#="n" THEN 370
350 IF a#<>"j" THEN 320
360 CLS:GOTO 220
370 CLS: PRINT"Sollen alle Verbrauchswerte (inklusive"
380 PRINT"der eventuell vorher eingelesenen Wer-"
390 PRINT"te) mit dem entsprechendem Datum"
400 INPUT"gespeichert werden (j/n) ";a#
410 IF a#="n" THEN 480
420 IF a#<>"j" THEN 370
430 OPENOUT "Benzinv."
440 FOR j=1 TO i
450 PRINT #9,v(j),ta(j),mo(j),ja(j)
460 NEXT j
470 CLOSEOUT
480 CLS:PRINT"Wollen Sie sich alle Werte einmal"
490 INPUT"anschauen (j/n) ";a#
500 IF a#="n" THEN 630
510 IF a#<>"j" THEN 480
520 CLS:PRINT" Verbrauch",TAB(28)"Datum":PRINT
530 FOR j=1 TO i
540 PRINT;USING "##.##";v(j);PRINT" Liter  ",
550 PRINT ;USING "##.";ta(j),
560 PRINT ;USING "##.";mo(j),
570 PRINT ;USING "##";ja(j)
580 IF j=15 THEN GOSUB 640
590 sv=sv+v(j):NEXT j
600 PRINT:PRINT"Das ist ein durchschnittlicher Ver-"
610 PRINT"brauch von : ";USING "##.##";sv/i;
620 PRINT" Liter"
630 PRINT:PRINT:PRINT:PRINT"Ende des Programms":END
640 REM UP Warten
650 LOCATE 7,25
660 PRINT"Bitte eine Taste druecken"
670 x#=#INKEY#:IF x#="" THEN 670
680 CLS:RETURN

```

Variablenliste:

a\$ = Antwortstring (j/n)

bv = absoluter Benzinverbrauch

gkm = gefahrene Kilometer

i = Laufindex

j = Laufindex

ja(i) = Jahr

mo(i) = Monat

sv = Summationsfeld für v(i)

ta(i) = Tag

v(i) = Verbrauch auf 100 Kilometer

Programmbeschreibung:

Satz 10- 70 : Überschrift

Satz 80 : Für die indizierten Variablen wird Platz
reserviert

Satz 90-120 : Erste Frage "Einlesen"

Diese Frage wird (wie auch die folgenden Fragen) wiederholt ausgegeben, wenn weder "j" noch "n" eingegeben wird. Bei der Eingabe von "n" wird zur nächsten Frage (Satz 180) verzweigt.

- Satz 130 : Die Eingabedatei "Benzinv." wird eröffnet
und der Laufindex i wird gleich 1 gesetzt
- Satz 140 : Die gespeicherten Informationen werden ein-
gelesen
- Satz 150 : Ist das Dateiende erreicht (dann ist EOF =
-1), wird zum Satz 170 verzweigt
- Satz 160 : Der Laufindex i wird um 1 erhöht und es
wird zum Einlesen weiterer Daten zurückver-
zweigt zum Satz 140
- Satz 170 : Die Eingabedatei wird geschlossen
- Satz 180-210 : Die zweite Frage "Eingeben"
- Satz 220 : Der Laufindex i wird erhöht
- Satz 230-280 : Die gefahrenen Kilometer, der absolute Ben-
zinverbrauch und Angaben bezüglich des Da-
tums werden angefordert
- Satz 290-310 : Der Verbrauch auf 100 Kilometer wird be-
rechnet und das Ergebnis wird auf zwei De-
zimalstellen genau ausgegeben
- Satz 320-360 : Abfrage bezüglich der weiteren Eingabe, Lö-
schen des Bildschirms und ggf. Verzweigung
zum Satz 220
- Satz 370-420 : Die dritte Frage "Sichern" (die Sätze 320-
360 gehörten sozusagen zur zweiten Frage)
- Satz 430-470 : Die Ausgabedatei "Benzinv." wird eröffnet,
Informationen werden peripher gespeichert
(es befinden sich jeweils i verschiedene
Werte im Arbeitsspeicher des Rechners) und
die Ausgabedatei wird wieder geschlossen

Satz 480-510 : Die vierte Frage "Ergebnisse"

Satz 520 : Ausgabe der Tabellenüberschrift

Satz 530-590 : Die Verbrauchswerte werden mit dem zugehörigen Datum formatiert ausgegeben. Nach 15 Ausgaben erfolgt ein Sprung ins Unterprogramm Warten, da die Ergebnisse ansonsten über den oberen Bildschirmrand hinaus wandern würden. Im Feld sv (Satz 590) werden die einzelnen Verbrauchswerte aufsummiert, um später das arithmetische Mittel aller Verbrauchswerte berechnen zu können.

Satz 600-620 : Berechnung des arithmetischen Mittels und Ausgabe des Ergebnisses

Satz 630 : Programmende

Satz 640-680 : Unterprogramm "Warten"

Programmergebnisse:

Werden nach dem Programmstart Werte eingelesen und/oder eingegeben (und evtl. natürlich direkt abgespeichert), so könnte der Ergebnisausdruck auf dem Bildschirm folgendermaßen aussehen:

Verbrauch Datum

8.05 Liter 1. 2.85

8.55 Liter 9. 2.85

8.13 Liter 20. 2.85

7.81 Liter 1. 3.85

Das ist ein durchschnittlicher
Verbrauch von : 8.14 Liter

3.3. Fahrtstreckenoptimierung

Und wieder steht das Auto im Mittelpunkt unserer Überlegungen. Es geht auch wieder um Kraftstoff bzw. Energie, diesmal jedoch in indirekter Form.

Kernpunkt dieses Kapitels ist die Planung einer Tour. Vorgegeben werden einzelne Orte, die Sie anfahren wollen (bzw. müssen, wenn Sie beruflich im Außendienst tätig sind) und die Entfernungen zwischen diesen Orten. Ihr CPC sagt Ihnen dann, welche Orte Sie nacheinander anzufahren haben, um möglichst wenige Kilometer zurückzulegen. Es geht also darum, die Energie Ihres Autos und auch Ihre Energie sinnvoll einzusetzen.

Nun ist es allerdings nicht so einfach, einen geeigneten Algorithmus (Rechenvorgang, der nach einem bestimmten sich wiederholenden Schema abläuft) zur Lösung dieses Problems zu finden. Vielleicht werden Sie sagen, daß Ihr Computer alle Möglichkeiten der Anordnung berücksichtigen sollte. Das wäre auch im Falle der Eingabe weniger Orte richtig. Zum Beispiel gibt es bei der Eingabe von 3 Orten a, b und c (der Ausgangsort wird dabei nicht variiert) 2 verschiedene Möglichkeiten der Anordnung: a-b-c und a-c-b. Werden 4 Orte a, b, c und d eingegeben, gibt es 6 verschiedene Möglichkeiten, nämlich: a-b-c-d, a-b-d-c, a-d-b-c, a-d-c-b, a-c-b-d und a-c-d-b.

Diese Anordnungen unterscheiden sich nur nach der Stellung der Elemente (bzw. Orte). Die Anzahl der Anordnungen von n Elementen, d. h. die Anzahl der verschiedenen Möglichkeiten die n vorgegebenen Orte anzufahren, ist gleich $n!$ ($n! = 1 * 2 * 3 * \dots * n$). Wollen Sie beispielsweise $10!$ berechnen, so erhalten Sie: $10! = 3628800$. Eine Berücksichtigung wirklich aller Möglichkeiten hätte daher bei der Eingabe mehrerer Orte enorme Rechenzeiten zur Folge, die sicherlich nicht mehr zu vertreten wären.

Man kommt in der Regel ebenfalls zu einem vernünftigen Ergebnis, wenn man nacheinander von einem Ort zum jeweils nächstliegenden Ort fährt. Will man, von Hamburg losfahrend, die Orte Hannover, Düsseldorf, Köln, Frankfurt und München besuchen, so kommt man sehr leicht zu der richtigen Anordnung HH - H - D - K - F - M. Nun kann es aber sein, daß z. B. 4 Orte wie folgt im Raum verteilt sind:



Von A ausgehend käme man zu der Anordnung A - B - C - D, obwohl die Anordnung A - C - B - D günstiger ist.

Daher wird in dem folgenden Programm der Möglichkeit Rechnung getragen, daß es jeweils, d. h. bei jeder Zuordnung eines Ortes zu den schon in eine Reihenfolge gebrachten Orten, günstiger sein kann, nicht zuerst den nächstliegenden Ort, sondern erst einmal den zweitnächsten Ort anzufahren.

Im Programmablauf werden somit $n-1$ ($n =$ Anzahl der eingegebenen Orte, einschließlich des Ausgangsortes) Fahrstrecken miteinander verglichen und die günstigste Alternative wird als Ergebnis ausgegeben.

Bezogen auf das obige Beispiel würden die folgenden Anordnungen überprüft:

A - B - C - D

A - C - B - D

A - B - D - C

Der an zweiter und dritter Stelle stehende Ort ist also jeweils einmal nicht der nächste, sondern der zweitnächste Ort.

Dieser Algorithmus findet in der Regel die für Sie günstigste Tour und befreit Sie damit von der - insbesondere wenn viele Orte anzufahren sind - mühevollen Arbeit, handschriftlich die beste Route auszuknobeln.

Programm:

```
10 REM I7
20 CLS
30 PRINT"I7 - Programm zur Fahrtstrecken-"
40 PRINT TAB(6)"optimierung"
50 PRINT:PRINT:PRINT
60 PRINT TAB(11) "Bernd Kowal, 1985"
70 GOSUB 520
80 PRINT"Wieviele Orte sollen insgesamt be-"
90 INPUT"ruecksichtigt werden ";n:PRINT
100 DIM h$(n),s$(n),o$(n),e(n,n),fs(n)
110 PRINT"Bitte geben Sie die Ortsnamen bzw."
120 PRINT"ein Ortskennzeichen ein :":PRINT
130 INPUT"Zunaechst der Ausgangsort ";o$(1):PRINT
140 FOR i=2 TO n
150 PRINT i;:INPUT".ter Ortsname ";o$(i)
160 NEXT i:CLS
170 PRINT"Geben Sie bitte folgende Entfernungen"
180 PRINT"in Kilometern an :":PRINT
190 FOR i=1 TO n-1
200 FOR j=i+1 TO n
```

```

210 PRINT o$(i);"-";o$(j),:INPUT e(i,j)
220 e(j,i)=e(i,j)
230 NEXT j:NEXT i:CLS
240 FOR u=1 TO n-1
250 FOR i=2 TO n:s$(i)="":NEXT i
260 i=1:s$(i)=o$(i):l=2
270 kw=10^30
280 FOR j=1 TO n
290 FOR k=1 TO n
300 IF s$(k)=o$(j) THEN 330
310 NEXT k
320 IF e(i,j)<kw THEN kw=e(i,j):q=j
330 NEXT j
340 IF u=1 AND kr=0 THEN e(i,q)=10^30:kr=1:GOTO 270
350 s$(l)=o$(q)
360 fs(u)=fs(u)+kw
370 IF l<n THEN i=q:l=l+1:GOTO 270
380 IF u>1 THEN 410
390 FOR i=1 TO n:h$(i)=s$(i):NEXT i
400 fs=fs(u):GOTO 440
410 IF fs(u)>=fs THEN 440
420 FOR i=1 TO n:h$(i)=s$(i):NEXT i
430 fs=fs(u)
440 kr=0:NEXT u
450 PRINT"Folgende Strecke ist optimal :":PRINT
460 FOR i=1 TO n
470 IF i=n THEN PRINT h$(i):GOTO 490
480 PRINT h$(i);"-";
490 NEXT i
500 PRINT:PRINT"Das sind";fs;"Kilometer."
510 PRINT:PRINT"Ende der Berechnungen":END
520 REM UP Warten
530 LOCATE 7,25
540 PRINT"Bitte eine Taste druecken !"
550 x$=INKEY$
560 IF x$="" THEN 550
570 CLS:RETURN

```

Variablenliste:

$e(i,j)$ = die Entfernung zwischen jeweils zwei Orten

f_s = Anzahl der Kilometer für die optimale Fahrtstrecke

$f_s(i)$ = Anzahl der Kilometer für eine Fahrtstreckenalternative

$h(i)$ = die Orte in der optimalen Rangreihe

i = Laufindex

j = Laufindex

k = Laufindex

kr = Kontrollvariable für die Zuordnung des zweitnächsten Ortes

kw = kürzeste Entfernung eines Ortes zu einem noch nicht zugeordneten Ort

l = Laufvariable für die jeweils (für jede Fahrtstreckenalternative) erfolgenden $n-1$ Zuordnungen

n = Anzahl der Orte

$o(i)$ = die eingegebenen Ortsnamen

q = Index für die Zuordnung des nächsten bzw. zweitnächsten Ortes

$s(i)$ = die für jede Fahrtstreckenalternative in eine Rangreihe gebrachten Orte

u = Laufvariable für die $n-1$ Fahrtstreckenalternativen

Programmbeschreibung:

Satz 10- 70 : Überschrift

Satz 80- 90 : Die Anzahl der Orte wird eingegeben

Satz 100 : Für die indizierten Variablen wird Speicherplatz reserviert

Satz 110-160 : Eingabe der Ortsnamen

Satz 170-230 : Eingabe der Entfernungen

In Satz 220 wird eine Entfernung $e(i,j)$ gleich der nicht eingegebenen Entfernung $e(j,i)$ gesetzt, da in der Schleife Satz 280-330 die Laufvariable j , unabhängig von der Größe der Variablen i , von 1 bis n läuft.

Satz 240 : Eröffnung der Programmschleife für die $n-1$ Fahrtstreckenalternativen

Satz 250 : Die Variable $s\$(i)$ wird für die neue Fahrtstreckenalternative von den alten Zuordnungen "befreit"

Satz 260-270 : Belegung der Variablen $i, l, s\$(1)$ und kw ($s\$(1) = \text{Ausgangsort } o\(1)). Für die Variable kw wird ein willkürlich gewählter, sehr hoher Wert angenommen ($10 \uparrow 30$), der bei den Entfernungsangaben nicht erreicht werden darf.

Satz 280-330 : Die kleinste Entfernung eines Ortes i zu einem "noch nicht angefahrenen Ort" j wird gesucht und im Feld kw zwischengespeichert. Welcher Ort j zugeordnet wurde, wird im

Feld q registriert. In den Sätzen 290-310 wird untersucht, ob der aktuelle Ort j schon zugeordnet bzw. "angefahren" worden ist.

Satz 340 : Im ersten Durchgang werden jeweils die kürzesten Entfernungen gesucht (dann ist $u=1$ und l immer größer u). Die folgenden Durchgänge sind dadurch gekennzeichnet, daß an immer wechselnden Stellen (in Abhängigkeit von u und l) die gefundenen kürzesten Entfernungen $e(i,q)$ gleich $10+30$ gesetzt werden, um dann erneut die kürzeste Entfernung (das ist dann natürlich die zweit-kürzeste, da die kürzeste Entfernung mit dem Wert $10+30$ belegt wurde) zu suchen. Die Kontrollvariable kr dient dazu, diesen Prozeß nur einmal zu durchlaufen.

Satz 350-360 : Der gefundene Ort wird im Feld $s(l)$ zwischengespeichert und die gefundene Entfernung in Kilometern wird zu dem Wert im Feld $fs(u)$ hinzuaddiert.

Satz 370 : Ist $l < n$, so wird der Zuordnungsprozeß je Fahrtstreckenalternative fortgesetzt

Satz 380-400 : Beim ersten Durchgang (dann ist $u=1$) wird die gefundene Fahrtstrecke zu späteren Vergleichszwecken (Satz 410) als optimal angesehen (Belegung der Variablen $h(i)$ und fs)

Satz 410-430 : Die Kilometeranzahl der gefundenen Fahrtstreckenalternative wird mit dem entsprechenden Wert des vorangegangenen Durchgangs verglichen. Gegebenenfalls werden die Variablen $h(i)$ und fs mit neuen Werten belegt.

Satz 440 : Die Kontrollvariable kr wird gleich 0 gesetzt und es erfolgt ein euer Durchgang (ab Satz 240)


Satz 450-500 : Ergebnisausgabe

Satz 510 : Programmende

Satz 520-570 : Unterprogramm "Warten"

Programmergebnisse:

Gehen wir von 6 Orten mit folgender geographischer Lage (D ist der Ausgangsort)



und den exakten Entfernungen

	A	B	C	D	E	F
A		40	46	13	39	12
B			30	22	47	32
C				30	26	29
D					34	12
E						23
F						

aus,

so erhalten wir nach den entsprechenden Eingaben:

Folgende Fahrtstrecke ist optimal:

D - A - F - E - C - B

Das sind 104 Kilometer.

Im Vergleich zu den vorangegangenen Programmen ist die erhöhte Rechenzeit auffallend. Wenn Sie dieses Programm für relativ viele Ortseingaben laufen lassen, müssen Sie schon etwas Geduld mitbringen.

3.4. Autokosten - fest im Griff

In Kapitel 7.2 dieses Buches finden Sie ein recht allgemein gehaltenes Programm zu den Haushaltskosten. Da Sie dort selbst Kategorien bilden können, denen dann einzelne Ausgaben zuzuordnen sind, läßt sich dieses Programm auch für den Bereich Autokosten verwenden. Als Kategorien kommen zum Beispiel in Frage: Versicherung, Steuern, Benzin-kosten, Reparaturen, Pflege und Wartung, Abschreibungen, usw. usf. Wollen Sie dieses Programm ausschließlich für Ihre Autokosten verwenden, ist eine Programmerweiterung bezüglich der Kosten pro gefahrenen Kilometer sinnvoll. Es gibt verschiedene Möglichkeiten, Kilometerangaben in diesem Programm unterzubringen:

- 1) Kilometereingabe bei Bedarf, durch einen gesonderten Programmteil, der vom Menü aus aufrufbar ist
- 2) Eingabe des Kilometerstandes bei jeder Geldausgabe
- 3) Bei der Eingabe eines Zeitraumes für den Programmteil "Berechnungen" wird ein Wert für die gefahrenen Kilometer angefordert und die Kosten pro Kilometer werden direkt anschließend ausgegeben

Die erste Möglichkeit hat den Nachteil, daß im Programmteil "Berechnungen" nicht immer die wirklich gefahrenen Kilometer eingehen werden, da der Kilometerstand nicht bei jeder Eingabe aktualisiert wird. Diesen Nachteil hat die zweite Möglichkeit zwar nicht, dafür ist es aber sehr unwahrscheinlich, daß Sie den Kilometerstand Ihres Autos immer im Kopf haben. Die dritte Möglichkeit ist programmtechnisch am einfachsten einzufügen und weist auch nicht die genannten Nachteile auf. Sie müssen sich dann den Wert der gefahrenen Kilometer nur anderweitig merken. Gerade in bezug auf Autokosten ist es aber sowieso sinnvoll, im Programmteil "Berechnungen" Jahres- oder Halbjahreszeiträume anzugeben, da z. B. Steuern und Versicherungsbeiträge auch nur jährlich oder halbjährlich anfallen.

Um auf diese Art und Weise das Haushaltskostenprogramm umzuschreiben, müssen Sie folgende Programmzeilen einfügen:


```
1395 PRINT:PRINT:PRINT"Wieviel Kilometer haben Sie in"  
1400 INPUT"diesem Zeitraum zurueckgelegt "rkm  
1514 PRINT:PRINT:PRINT"Auf den Kilometer bezogen"  
1515 PRINT"ergeben sich Kosten von"  
1516 PRINT INT(su/km*100+0.5)/100;"DM pro Km."  
1517 GOSUB 2060
```

Hierbei wird nur der Satz 1400 überschrieben. Auf den Kommentar "Ich rechne" läßt sich aber auch leicht verzichten.

Ansonsten ist Ihnen viel Spaß bei der Nutzung des Programms zu wünschen. Und ... fallen Sie nicht vom Hocker, wenn Sie sehen wieviel Sie für Ihr Vehikel ausgeben...

3.5. Ideenfundgrube

- Simulation der Fahrt eines 700 PS starken Porsche 935 über den neuen Nürburgring
- exakte Dokumentation eines Rennverlaufs (der Wagen-durchlauf kann dabei mit Lichtschranken erfasst werden)
- Autos aus dem Computer (Design, elektronisch gesteuerte Prüfstände, Dauerprüfungen im Simulator)
- Verkehrstrainer (Bewältigung von simulierten Verkehrssituationen)
- Fahrschule: Schüler, Stunden, Kassenbuch und Tagesnachweis
- Datenbank für Auto-Vergleichsdaten (Höchstgeschwindigkeit, Beschleunigung, Testergebnisse etc.)
- Ortskenntnisprüfungen mit dem Computer
- und noch vieles mehr



Zust

4. Geld und Kredit

4.1. Vorbemerkung

Wenn's ums Geld geht ... dann hört für viele der Spaß auf, und die meisten wollen alles ganz genau wissen. Sparkassen und Banken setzen schon lange Rechner ein, um auch komplizierte Dinge schnell und zuverlässig auszurechnen. Sie als Kunde sind da häufig auf die Angaben des Geldinstituts angewiesen, wenn es zum Beispiel um die Aufnahme eines Kredits geht. Und diese Angaben können schon verwirrend oder von Institut zu Institut verschieden sein, so daß ein Vergleich der Angebote recht schwierig sein kann. Daher ist es nur angemessen, wenn wir in diesem Kapitel die Chancengleichheit für den Bankkunden verbessern wollen.

Natürlich sind die hier vorgestellten Probleme und Programme nicht mit der umfangreichen Software großer Geldinstitute zu vergleichen; dennoch sind sie auf jeden Fall hilfreich, wenn sich Ihre Geldausgaben und -einnahmen nicht immer decken, und Sie darauf angewiesen sind, einen Kredit aufzunehmen oder Ihr Geld zinsgünstig anzulegen.

4.2. Die Zinsrechnung

Die Grundlage des ganzen Bereichs "Geld und Kredit" ist die Zinsrechnung. Rein mathematisch ist das natürlich eine vergleichsweise einfache Angelegenheit. Der eine oder andere Leser wird aber unter Umständen nicht mehr so genau wissen, wie die elementaren Größen der Zinsrechnung berechnet werden. Daher dient das folgende Programm mehr der Einübung des Umgangs mit diesen Größen, als der praktischen Anwendung bezüglich Ihrer Finanzplanung.

Die folgenden Berechnungsformeln liegen dem Programm zugrunde:

$$\text{Zins} = \frac{\text{Kapital} * \text{Zeit} * \text{Zinssatz}}{100}$$

$$\text{Kapital} = \frac{\text{Zins} * 100}{\text{Zinssatz} * \text{Zeit}}$$

$$\text{Zinssatz} = \frac{\text{Zins} * 100}{\text{Kapital} * \text{Zeit}}$$

$$\text{Zeit} = \frac{\text{Zins} * 100}{\text{Kapital} * \text{Zinssatz}}$$

Die Angabe der Zeit erfolgt hier in Jahren. Ein Jahr hat dabei 360 Tage oder 12 Monate zu je 30 Tagen. Wollen Sie Monate und/oder Tage berücksichtigen, so müssen Sie diese entsprechend umrechnen.

Programm:

```

10 REM IB
20 CLS
30 PRINT"IB - Programm zur Berechnung"
40 PRINT TAB(6)"der Zinsen,des Zinssatzes,"
50 PRINT TAB(6)"des Kapitals oder der Zeitdauer"
60 PRINT TAB(6)"der Kapitalueberlassung"
70 PRINT:PRINT:PRINT
80 PRINT TAB(11) "Bernd Kowal, 1985"
90 GOSUB 460
100 PRINT TAB(12)"Auswahlmenue":PRINT:PRINT
110 PRINT"Was soll berechnet werden?":PRINT
120 PRINT TAB(30)"Eingabe":PRINT
130 PRINT"Zinsen",TAB(33)"1":PRINT
140 PRINT"Zinssatz",TAB(33)"2":PRINT
150 PRINT"Kapital",TAB(33)"3":PRINT
160 PRINT"Dauer der Kapital-"
170 PRINT"ueberlassung",TAB(33)"4"
180 PRINT:PRINT:PRINT:INPUT"Ihre Wahl";a
190 IF a<1 OR a>4 THEN CLS:GOTO 100
200 CLS:PRINT"Eingabe der bekannten Groessen :":PRINT

```

```

210 IF a<>1 THEN INPUT"Zinsen (in DM)           ";Zi
220 IF a<>2 THEN INPUT"Zinssatz (in Prozent)    ";Zs
230 IF a<>3 THEN INPUT"Kapital (in DM)         ";K
240 IF a<>4 THEN INPUT"Zeitdauer (in Jahren)   ";Zd
250 PRINT:PRINT
260 IF a>1 THEN 310
270 Zi=K*Zd*Zs/100
280 PRINT"Nach";Zd;"Jahren bringt das Kapital"
290 PRINT Zi;"DM an Zinsen."
300 GOTO 410
310 IF a>2 THEN 350
320 Zs=Zi*100/K/Zd
330 PRINT"Der Zinssatz betraegt";Zs;"%."
340 GOTO 410
350 IF a>3 THEN 390
360 K=Zi*100/Zs/Zd
370 PRINT"Das Kapital belaeuft sich auf";K;"DM."
380 GOTO 410
390 Zd=Zi*100/K/Zs
400 PRINT"Die Zeitdauer betraegt";Zd;"Jahre."
410 PRINT:PRINT:INPUT"Noch eine Berechnung (j/n) ";a#
420 IF a#="j" THEN CLS:GOTO 100
430 IF a#<>"n" THEN 410
440 PRINT:PRINT:PRINT"Dann eben nicht !"
450 END
460 REM UP Warten
470 LOCATE 7,25
480 PRINT"Bitte eine Taste druecken !"
490 x#=INKEY#
500 IF x#="" THEN 490
510 CLS:RETURN

```

Variablenliste:

a = Antwortvariable bei der Menüauswahl

a\$ = Antwortstring (j/n)

K = Kapital

Zd = Zeitdauer der Kapitalüberlassung

Zi = Zins

Zs = Zinssatz

Programmbeschreibung:

Satz 10- 90 : Überschrift

Satz 100-190 : Menüauswahl

Satz 200-250 : Es werden Werte für die bekannten Größen
angefordert

Satz 260-400 : Die unbekannte Größe wird berechnet und das
Ergebnis wird ausgegeben

Satz 410-440 : Verzweigung zur Menüauswahl, wenn noch eine
Berechnung durchgeführt werden soll. Anson-
sten wird ein Kommentar ausgegeben.

Satz 450 : Programmende

Satz 460-510 : Unterprogramm "Warten"

Programmergebnisse:

Geben wir bei der Menüauswahl eine 1 ein, so werden daraufhin Werte für den Zinssatz, für das Kapital und für die Zeitdauer angefordert. Nehmen wir einen Zinssatz von 8.95%, ein Kapital von 347,- DM und eine Zeitdauer von 4 Jahren, so meldet der Rechner:

Nach 4 Jahren bringt das Kapital
124.226 DM an Zinsen.

Noch eine Berechnung (j/n) ?

4.3. Ihr Geld auf Ihrem Sparkonto (Zinseszinsrechnung)

Bringen Sie einen bestimmten Kapitalbetrag zu einer Bank, so werden nach Ablauf eines Jahres Zinsen zu dem Kapitalbetrag hinzuaddiert. Im darauffolgenden Jahr - und dies ist im letzten Kapitel nicht berücksichtigt worden - werden die schon angefallenen Zinsen wiederum verzinst (Zinseszinsrechnung). Insofern ist das Programm im vorangehenden Kapitel für praktische Anwendungen weniger geeignet. Hingegen rechnet Ihnen das hier vorliegende Programm genau aus, wie sich Ihr Geld auf Ihrem Sparkonto "entwickelt", wie es sich also vermehrt. Sie müssen nur einen Ausgangsbetrag und einen beliebigen Zinssatz vorgeben. Als Ergebnis erhalten Sie dann den Endbetrag nach einer beliebigen Anzahl von Jahren.

Programm:

```
10 REM I9
20 CLS
30 PRINT"I9 - Programm zur Berechnung von"
40 PRINT TAB(6)"Zinseszinsen bei jaehrlicher"
50 PRINT TAB(6)"Zinszahlung und gegebenem"
60 PRINT TAB(6)"Anfangskapital"
70 PRINT:PRINT:PRINT
80 PRINT TAB(11) "Bernd Kowal, 1985"
90 GOSUB 260
100 INPUT"Anfangskapital      ";Ak
110 INPUT"Jaehrlicher Zinssatz ";Zs
120 PRINT"Dauer der Kapital-"
130 INPUT"ueberlassung in Jahren ";Zd
140 j=20:l=1
150 CLS:PRINT" Jahr          Zins          Kapital"
160 PRINT:FOR i=1 TO Zd
170 Zb=Ak*Zs/100
180 Ek=Ak+Zb
190 PRINT; USING "####      ";i;
200 PRINT; USING "#####.##      ";Zb;
210 PRINT; USING "#####.##";Ek
220 Ak=Ek
230 IF i=j THEN l=j+1:j=j+20:GOSUB 260:GOTO 150
240 NEXT i
250 END
260 REM UP Warten
270 LOCATE 7,25
280 PRINT"Bitte eine Taste druecken !"
290 x#=INKEY#
300 IF x#="" THEN 290
310 CLS:RETURN
```

Variablenliste:

Ak = Ausgangskapital

Ek = Endkapital

i = Laufindex

j = Hilfsvariable, die eine Programmunterbrechung bei
vollem Bildschirm ermöglicht

l = variabler Schleifenanfangswert

Zb = Zinsbetrag pro Jahr

Zd = Zeitdauer der Kapitalüberlassung

Zs = jährlicher Zinssatz in Prozent

Programmbeschreibung:

Satz 10- 90 : Überschrift

Satz 100-130 : Anforderung der Ausgangsdaten

Satz 140 : Der Schleifenanfangswert wird auf 1 - und
die Hilfsvariable j auf 20 gesetzt

Satz 150 : Der Bildschirm wird geräumt und eine Ta-
bellenüberschrift wird ausgegeben

Satz 160-240 : Für jedes Jahr wird der Zinsbetrag und das

Endkapital berechnet und die Ergebnisse werden formatiert ausgegeben. Nach der Ausgabe einer Ergebniszeile wird aus dem berechneten Endkapital das Ausgangskapital für die nächste Periode (Satz 220). Satz 230 dient der Programmunterbrechung nach 20 Perioden (Unterprogramm "Warten"). In diesem Falle wird auch der Schleifenanfangswert I entsprechend geändert.

Satz 250 : Programmende

Satz 260-310 : Unterprogramm "Warten"

Programmergebnisse:

Bei einem Anfangskapital von 1000,- DM, und einem jährlichen Zinssatz von 4.8% und einer Zeitdauer von 10 Jahren, ergibt sich:

Jahr	Zins	Kapital
1	48.00	1048.00
2	50.30	1098.30
3	52.72	1151.02
4	55.25	1206.27
5	57.90	1264.17
6	60.68	1324.85
7	63.59	1388.45
8	66.65	1455.09
9	69.84	1524.94
10	73.20	1598.13

4.4. Der effektive Jahreszins

Wenn Sie bei einem Geldinstitut nach Kreditkonditionen fragen, werden Ihnen die Berater die unterschiedlichsten Angebote unterbreiten. Ohne Effektivzinsangaben ist das Vergleichen der Angebote schwer, für den Laien sogar oft unmöglich. Der Effektivzins gibt nämlich Auskunft darüber, welches Kreditangebot das günstigere ist.

Selbstverständlich berechnen die Kreditinstitute für ihre interne Kalkulation einen Effektivzins. Jedoch verhalten sich einige Banken ziemlich zurückhaltend, wenn es darum geht, diesen Effektivzins auch ihren Kunden zu nennen. Sie als Bankkunde haben dann das Nachsehen.

Was ist denn nun genau der Effektivzins und wie läßt er sich berechnen?

Im Effektivzins sind alle Preisbestandteile eines Kredits enthalten. Er ist sozusagen eine Maßzahl für den Preis eines Kredits. Dabei werden sämtliche Kreditkosten in Prozent zur je einjährigen Nutzung des tatsächlich ausgezahlten Betrags ausgewiesen.

Bis zum 1.1.1981 war die Berechnung des effektiven Jahres-

zinses recht unproblematisch. Die Gleichung für Kreditverträge, die vor dem 1.1.1981 abgeschlossen wurden, sieht folgendermaßen aus:

$$e = \frac{24 * (\text{Monatzzinssatz} * \text{Laufzeitmonate} + b)}{\text{Laufzeitmonate} + 1}$$

Dabei ist:

e = Effektivzinssatz in % p. a.

b = Bearbeitungsgebühr in % vom ursprünglichen Kreditbetrag

Seit dem 1.1.1981 wird zur Berechnung des effektiven Jahreszinses die "360-Tage-Methode" angewandt. Mit Hilfe folgender Gleichung wird seither der effektive Jahreszins bei einem Kredit mit monatlicher Ratenzahlung und Pro-Monats-Zinsberechnung (bezogen auf den aufgenommenen Kreditbetrag) berechnet:

$$\frac{1 + \frac{b}{100}}{M} + \frac{p}{100} = \frac{q^j}{\left(\frac{12}{i} \frac{11}{2}\right) * (q^j - 1) + \left(1 + \frac{m-1}{24} * i\right) * \frac{m}{1 + \frac{m}{12} * i}}$$

Dabei ist:

e = gesuchter Effektivzins in % p. a. (=100*i)

i = e/100

q = 1+i

M = Laufzeit in Monaten

= Anzahl der vereinbarten Monatsraten (=12*J+m)

J = Laufzeit in vollen Jahren
= Vorkommensanteil von M/12

m = restliche Laufzeit in Monaten = M - 12 * J

p = nomineller p. M.-Zinssatz in % vom ursprünglichen Kreditbetrag

b = Bearbeitungsgebühr in % vom ursprünglichen Kreditbetrag

Diese Gleichung läßt sich nicht in die Form Effektivzins $e = \dots$ bringen. Abgesehen von recht anspruchsvollen mathematischen Näherungslösungen ist e nur durch mehrfaches Probieren zu bestimmen. Nicht umsonst werden im Bankgewerbe Computer eingesetzt (oder Tabellenwerke mit rund 6000 Seiten), um den Effektivzins zu ermitteln. Während die alte Formel den Vorteil hatte, daß auch Nicht-Mathematiker mit ihr umgehen konnten, liefert die neue Methode einfach genauere Ergebnisse.

Mit Unterstützung Ihres CPC und einem geeigneten Programm ist die Bestimmung des effektiven Zinssatzes gar nicht so schwierig. Dabei gehen wir von folgender Überlegung aus:

Bringen wir die links vom Gleichheitszeichen stehenden Terme auf die rechte Seite (dadurch ändern sich bekanntlich die Vorzeichen), so erhalten wir eine Gleichung der Form

$$0 = \dots \dots \dots$$

Setzen wir nun verschiedene Werte für e ein, so ist der gesamte Ausdruck rechts vom Gleichheitszeichen kleiner als 0, wenn der eingesetzte Wert für e zu klein ist. Er ist größer als 0, wenn wir einen zu großen Wert gewählt haben.

In dem folgenden Programm werden daher immer größere Werte für e eingesetzt, so lange, bis der Ausdruck rechts vom Gleichheitszeichen größer als 0 ist. Dann werden zur näheren Lokalisierung des wahren Wertes für e immer kleinere Intervalle und Schrittweiten für die Versuchswerte angenommen, so daß letztendlich ein auf 4 Stellen (hinter dem Dezimalpunkt) genaues Ergebnis herauskommt.

Eine derartige Vorgehensweise ist quasi nur mit einem Computer möglich. Ansonsten wäre der Zeitaufwand unverhältnismäßig groß. Vorgeben müssen Sie im Programmablauf lediglich Werte für die Größen M, p und b der obigen Gleichung.

Programm:

```
10 REM I10
20 CLS
30 PRINT"I10 - Programm zur Berechnung des"
40 PRINT TAB(7)"effektiven Jahreszinses"
50 PRINT:PRINT:PRINT
60 PRINT TAB(11) "Bernd Kowal, 1985"
70 GOSUB 310
80 PRINT"Anzahl der vereinbarten"
90 INPUT"Monatsraten ";m:PRINT
100 PRINT"Nomineller Monatszinssatz in % vom"
110 INPUT"urspruenglichen Kreditbetrag ";p:PRINT
120 PRINT"Bearbeitungsgebuehr in % vom"
130 INPUT"urspruenglichen Kreditbetrag ";b
140 IF m=0 OR p=0 THEN CLS:GOTO 80
150 PRINT:PRINT:PRINT:PRINT
```

```

160 k1=1:k2=100:s=1
170 FOR e=k1 TO k2 STEP s
180 i=e/100:q=1+i:j=FIX(m/12):r=m-12*j
190 lt=(1+b/100)/m+p/100;z=q^j
200 n1=(12/i+11/2)*(q^j-1)
210 n2=(1+(r-1)/24*i)*r/(1+r/12*i)
220 vg=z/(n1+n2)-1t
230 IF vg>0 THEN GOSUB 280:GOTO 170
240 NEXT e
250 PRINT"Effektiver Jahreszins :":PRINT
260 PRINT"e = ";USING "##.####";e;:PRINT"%"
270 PRINT:PRINT"Ende der Berechnungen":END
280 k1=e-s:k2=e:s=s/10
290 IF s=0.000001 THEN 250
300 RETURN
310 REM UP Warten
320 LOCATE 7,25
330 PRINT"Bitte eine Taste druecken !"
340 x$=INKEY$
350 IF x$="" THEN 340
360 CLS:RETURN

```

Variablenliste:

b = Bearbeitungsgebühr

e = gesuchter Effektivzinssatz

i = $e/100$

j = Laufzeit in vollen Jahren

$k1$ = Anfangswert der Programmschleife (untere Intervallgrenze)

$k2$ = Endwert der Programmschleife (obere Intervallgrenze)

lt = linker Teil der Berechnungsformel (links vom Gleichheitszeichen)

m = Laufzeit in Monaten

$n1$ = erster Summand des Nenners (rechts vom Gleichheitszeichen)

$n2$ = zweiter Summand des Nenners (rechts vom Gleichheitszeichen)

p = nomineller p. M.-Zinssatz

q = $1+i$

rm = restliche Laufzeit in Monaten

s = Schrittweite der Programmschleife

vg = Vergleichswert (größer oder kleiner als 0)

z = Zähler (rechts vom Gleichheitszeichen)

Programmbeschreibung:

Satz 10- 70 : Überschrift

Satz 80-130 : Die Ausgangswerte werden angefordert

Satz 140 : Wiederholung der Werteanforderung, wenn für m oder p eine 0 eingegeben wird

Satz 150 : Vier Leerzeilen werden ausgegeben

Satz 160 : Anfangs-, Endwert und Schrittweite der Programmschleife werden jeweils mit einem Ausgangswert belegt

Satz 170-240 : Programmschleife zur Berechnung des aus allen Termen der Berechnungsformel bestehenden Vergleichswertes vg für alternative Werte e (e ist der Laufindex der Schleife).
Ist in Satz 230 der Vergleichswert größer als 0, erfolgt ein Sprung ins Unterprogramm (Satz 280), um Anfangs- und Endwert (welche als Intervallgrenzen aufzufassen sind) und Schrittweite der Programmschleife zu verändern. Weiterhin wird in diesem Fall zum Satz 170 verzweigt (neuer Schleifendurchlauf).

Satz 250-260 : Ergebnisausgabe

Satz 270 : Programmende

Satz 280-300 : Unterprogramm zur Neubelegung der Variablen k_1, k_2 und s . Die Werte für k_1 und k_2 werden so gewählt, daß zwischen k_1 und k_2 der wahre Wert für e liegen muß (beim Schlei-

fendurchlauf für e-s war vg noch negativ!).
In Satz 290 befindet sich ein Kriterium
für den Programmabbruch.

Satz 310-360 : Unterprogramm "Warten"

Programmergebnisse:

Geben wir nach dem Programmstart die Werte 50 (Anzahl der Monatsraten), 0.5 (nomineller p. M.-Zinssatz) und 1 (Bearbeitungsgebühr in %) ein, so erhalten wir folgendes Ergebnis:

Effektiver Jahreszins :

e=12.0329 %

Mit der alten Formel ermitteln wir hingegen einen geringfügig höheren Effektivzins:

$$e = \frac{24 * (0.5 * 50 + 1)}{50 + 1} = 12.2353$$

4.5. Schulden und ihre Tilgung (Tilgungsplan)

Auch in diesem Kapitel geht es um das Thema Schulden. Wollen Sie beispielsweise einen Kredit von nominal 100000 DM aufnehmen, berechnet Ihnen Ihre Bank in der Regel ein Disagio. Das Disagio (auch Damnum genannt) muß zwar verzinst und getilgt werden, es wird aber nicht ausbezahlt, sondern als eine Art Gebühr von Ihrer Bank einbehalten. Sie müssen also bei der Berechnung der Zinsen und der Tilgung von der Nominalschuld und nicht von dem Auszahlungskurs ausgehen.

Mit Ihrer Bank werden Sie bei der Kreditaufnahme einen Zinssatz und eine Anfangstilgung (ebenfalls in Prozent vom nominalen Kreditbetrag) vereinbaren. Aus diesen beiden Größen setzt sich dann Ihre Rate zusammen, die Sie monatlich oder vierteljährlich zu entrichten haben. Bei einem Zinssatz von 6% und einer Anfangstilgung von 2% sind das 2000,- DM pro Vierteljahr oder 8000,- DM pro Jahr. Für die Laufzeit des Kredits bleibt diese Rate gleich hoch. Allerdings wächst der Tilgungsanteil in der Rate, während der Zinsanteil ständig abnimmt, da sich das zu verzinsende Kapital von Jahr zu Jahr vermindert. Daher dauert die Rückzahlung des Darlehens in dem obigen Beispiel auch nicht 50 Jahre, sondern nur ca. 23 Jahre (bei unverändertem Zins).

Kreditangebote unterscheiden sich auch in der Dauer des fest vereinbarten Zinssatzes. Nach Ablauf dieser Frist wird der Zinssatz je nach Marktlage erhöht oder gesenkt. Sie können dann aber auch, falls Sie dies wünschen, die Restschuld bar bezahlen, sei es nun aus eigenen Mitteln oder aus Fremdmitteln.

Es kann also für Sie sehr wichtig sein, diese Restschuld zu kennen. Sie wird mit Hilfe eines Tilgungsplans ermittelt, der die Kontoführung des Geldinstituts nachbildet. Es gehen natürlich alle relevanten Kreditparameter, wie Rate, zu verzinsendes Kapital, Zins, Tilgung und Restschuld in diese Kontoführung mit ein.

Einen derartigen Tilgungsplan simuliert das folgende Programm. Gleichzeitig werden Ihnen die Möglichkeiten des CPC demonstriert, mit Hilfe des 80-Zeichen-Modus und der PRINT USING-Anweisung Tabellen zu erstellen.

Zu diesem Tilgungsplan sind jedoch noch einige Anmerkungen zu machen:

- 1) Ausgangsgröße ist die Nominalschuld und nicht der ausbezahlte Betrag.
- 2) Verzinst und getilgt wird diese Nominalschuld.
- 3) Obwohl regelmäßig (im vorliegenden Programm vierteljährlich) Raten gezahlt werden, wird jeweils ein Jahr lang der Zins auf einen unveränderten Kapitalbetrag berechnet. Erst nach Ablauf eines Jahres vermindert sich das zu verzinsende Kapital.
- 4) Die Restschuld hingegen verringert sich mit jeder Ratenzahlung um den Tilgungsbetrag. Sie ist nach Ablauf eines Jahres identisch mit dem zu verzinsenden Kapital.
- 5) Nach Ablauf eines Jahres werden die Kontenbewegungen abgeschlossen. Die geleisteten Zins-, Tilgungs- und Ratenzahlungen werden aufsummiert und das zu verzinsende Kapital wird um den Tilgungsbetrag vermindert. Die Restschuld bleibt beim Abschluß unberührt, da sie sich mit jeder Ratenzahlung verkleinert (natürlich nur, wenn auch wirklich Schulden getilgt werden).

- 6) Der Tilgungsplan wird solange fortgesetzt, bis die Restschuld kleiner als die Rate ist. Im vorliegenden Programm erfolgt dann ein entsprechender Ausdruck.
- 7) Um die Bildschirm Ausgabe zu optimieren, ist im vorliegenden Programm bei der Tabellenerstellung darauf verzichtet worden, die Erläuterung "Bitte eine Taste drücken !" gesondert auszugeben. Weiterhin ist auf die Ausgabestellen der PRINT USING-Anweisung zu achten. Geben Sie nach dem Programmstart Milliardenbeträge ein, so gerät die Tabellenausgabe durcheinander.

Programm:

```

10 REM I11
20 CLS
30 PRINT"I11 - Programm zur Aufstellung"
40 PRINT TAB(7)"eines Tilgungsplans"
50 PRINT:PRINT:PRINT
60 PRINT TAB(11) "Bernd Kowal, 1985"
70 GOSUB 640
80 INPUT"Kreditbetrag (nominal) ";Kb:PRINT
90 INPUT"Nominaler Jahreszinssatz ";Zs:PRINT
100 INPUT"Wieviel % Anfangstilgung ";t:PRINT
110 PRINT"Eingabe des Auszahlungsdatums : "
120 INPUT"Tag ";tg
130 INPUT"Monat ";m
140 INPUT"Jahr ";j#:PRINT
150 j#=RIGHT$(j#,2):j=VAL(j#)
160 PRINT"Es wird somit der";tg;".";m;".";j
170 PRINT"als Auszahlungsdatum angenommen."
180 GOSUB 640:CLS:MODE 2
190 GOSUB 550
200 k1=1:k2=5:R=0:Zi=0:T1=0:Rs=Kb
210 FOR i=k1 TO k2
220 IF i=k1 THEN 260
230 IF m>9 AND j=99 THEN j=0:GOTO 250
240 IF m>9 THEN j=j+1
250 IF m>9 THEN m=m-9 ELSE m=m+3

```

```

260 PRINT;USING"##.";tg;:PRINT;USING"##.";m;
270 PRINT;USING"##";j;
280 IF i=1 THEN PRINT TAB(11)"Auszahlung";:GOTO 310
290 IF i=k1 THEN PRINT TAB(11)"Abschluss";:GOTO 310
300 PRINT TAB(11)i-1;". Rate";
310 PRINT TAB(23);USING"#####.##";R;
320 PRINT;USING"      #####.##";Kb;
330 PRINT;USING"      #####.##";Zi;
340 PRINT;USING"      #####.##";Ti;
350 PRINT;USING"      #####.##";Rs
360 IF i>k1 THEN 410
370 IF i=1 THEN R=Kb*(Zs+t)/400 ELSE R=R/4
380 PRINT
390 Zi=Kb*Zs/400
400 Ti=R-Zi
410 Rs=Rs-Ti
420 IF Rs<0 THEN 510
430 IF i<k2 THEN 460
440 R=R*4;Zi=Zi*4;Ti=Ti*4
450 Kb=Kb-Ti;Rs=Rs+Ti/4
460 IF zae=3 AND k1=i THEN zae=0;GOSUB 640;GOSUB 560
470 NEXT i
480 zae=zae+1
490 k1=k2;k2=k2+4
500 GOTO 210
510 PRINT;PRINT"Mit der Zahlung von";
520 PRINT INT((Rs+Ti)*100+0.5)/100;"DM sind ";
530 PRINT"die Schulden abgetragen."
540 GOSUB 640;MODE 1:END
550 REM UP Tabelleneberschrift
560 PRINT"Datum      Vorgang      Rate      Zu ver-";
570 PRINT"      Zins      Tilgung      Rest-";
580 PRINT TAB(25)"in DM";TAB(35)"zinsendes";
590 PRINT TAB(48)"in DM";TAB(58)"in DM";TAB(71)"schuld"
600 PRINT TAB(35)"Kapital";TAB(71)"in DM"
610 PRINT TAB(35)"in DM"
620 PLOT 0,330:DRAW 620,330:PRINT
630 RETURN
640 REM UP Warten
650 LOCATE 7,25
660 IF i=0 THEN PRINT"Bitte eine Taste druecken !"
670 x#=INKEY#
680 IF x#="" THEN 670
690 CLS:RETURN

```

Variablenliste:

i = Laufindex

j = Jahr

$j\$$ = Eingabefeld für j

k_1 = Anfangswert der Programmschleife

k_2 = Endwert der Programmschleife

K_b = nominaler Kreditbetrag bzw. zu verzinsendes Kapital

m = Monat

R = Rate in DM

R_s = Restschuld in DM

t = Anfangstilgung in %

tg = Tag

T_i = Tilgung in DM

z_{ae} = Zählvariable

Z_i = Zins in DM

Z_s = nominaler Jahreszins in %

Programmbeschreibung:

Satz 10- 70 : Überschrift

Satz 80-170 : Eingabe der Ausgangsgrößen

Die Jahresangabe wird als Stringvariable aufgefaßt, um in Satz 150 nur die beiden zuletzt eingegebenen Zeichen in einen numerischen Ausdruck umzuwandeln. Diese etwas umständliche Vorgehensweise hat zur Folge, daß von einer Eingabe "1985" nur der Wert "85" als Jahr in die Tabellenerstellung eingeht.

Satz 180 : Es wird zum Unterprogramm "Warten" verzweigt, der Bildschirm wird geräumt und der Bildschirmmodus wird geändert (80 Zeichen pro Zeile)

Satz 190 : Sprung ins Unterprogramm (Satz 550) zur Ausgabe der Tabellenüberschrift

Satz 200 : Belegung der Variablen mit Ausgangswerten für den ersten Schleifendurchlauf

Satz 210-470 : Bildung einer Programmschleife zur Berechnung der einzelnen Größen und zur tabellarischen Ausgabe der Ergebnisse für jeweils eine Periode (= 1 Jahr). Beim jeweils ersten Durchlauf wird der Vorgang "Auszahlung" bzw. "Abschluß" berücksichtigt.
220-250 : Im 2. - 5. Durchlauf je Periode (dann werden Raten gezahlt) verändert sich das Datum
260-270 : Ausgabe des Datums

280-300 : Ausgabe des jeweiligen Vorgangs
(1. Durchgang der 1. Periode:
Auszahlung, 1. Durchgang der fol-
genden Perioden: Abschluß,
2. - 5. Durchgang aller Perioden:
Ratenzahlung)

310-350 : Ausgabe der übrigen Tabellenwerte

360-400 : Im jeweils 1. Durchgang werden
Rate, Zins und Tilgung berechnet
(bleiben für die jeweilige Pe-
riode konstant) und eine Leer-
zeile wird ausgegeben. In Satz
370 ist zu beachten, daß eigent-
lich nur im 1. Durchgang der 1.
Periode die Rate berechnet wird.
Ansonsten wird der letzte Ab-
schlußwert einfach durch 4 ge-
teilt.

430-450 : Im jeweils letzten Durchgang wird
die Rate, der Zins, das zu ver-
zinsende Kapital und die Rest-
schuld für den Abschluß berech-
net.

460 : Sind die Ergebnisse für 3 Peri-
oden ($z_{ae}=3$) einschließlich Ab-
schluß ($k_1=i$, 1. Durchgang) aus-
gegeben, wird der Zählindex wie-
der gleich 0 gesetzt und es wird
in die Unterprogramme "Warten"
und "Tabellenüberschrift" ver-
zweigt.

Satz 480-500 : Nach Beendigung des Schleifendurchlaufs für
eine Periode wird der Zählindex um 1 er-
höht, Anfangs- und Endwert der Programm-
schleife werden neu belegt und es wird zum
Satz 210 (erneuter Schleifendurchlauf) ver-
zweigt.

Satz 510-530 : Es wird ein Kommentar bezüglich der Restzahlung ausgegeben, wenn keine Rate mehr zu zahlen ist.

Satz 540 : Sprung ins Unterprogramm "Warten", Umschaltung auf den 40-Zeichen-Modus und Programmende

Satz 550-630 : Unterprogramm "Tabellenüberschrift"
In Satz 620 wird die Überschrift unterstrichen. Mit der Anweisung PLOT wird der Graphikcursor in die entsprechende Position gebracht und mit der Anweisung DRAW wird die Linie gezeichnet.


Satz 640-690 : Unterprogramm "Warten"
Der Kommentar "Bitte eine Taste drücken" wird nur ausgegeben, wenn noch nicht mit der Tabellenerstellung begonnen wurde.

Programmergebnisse:

Starten Sie das Programm für einen nominalen Kreditbetrag von 10000 DM, einem Jahreszins von 7% und einer Anfangstilgung von 3%, und lassen Sie sich das Geld am 30.12.1985 auszahlen, so werden Sie sehen, daß Sie bei einer vierteljährlichen Belastung von 225,- DM, 88 Raten bis zum 30.12.07 (also genau 22 Jahre) entrichten müssen, und daß dann noch ein Restbetrag von 198.85 DM zu zahlen ist. Sie zahlen dann also insgesamt 19998.85 DM zurück.

4.6. Ideenfundgrube

- Aufstellung eines Vergleichskontos, um Kreditangebote vergleichbar zu machen (dabei spielen - im Gegensatz zum Tilgungsplan - die tatsächlichen Zahlungen zu den tatsächlichen Terminen die entscheidende Rolle)
- Bestimmung der Bewertungszahl bei Bausparverträgen
- Baufinanzierung und Bauherrenmodelle
- Kostenverwaltung und Kostenrechnung
- Wertpapieranalyse: Kursanalyse und Börsentrend
- Aktien-Depotverwaltung
- und noch vieles mehr



5. Texten und Drucken

5.1. Vorbemerkung

Ein Mikrocomputer ist nicht nur zum Rechnen da, sondern er kann auch Texte verarbeiten. Dieses Einsatzgebiet der Computer dürfte nicht nur jedem Büroangestellten bekannt sein.

Die wichtigsten Aufgabenstellungen der Textverarbeitung sind:

- Speichern und Archivieren von Texten
- wiederholtes Ausgeben von Texten
- beliebiges und schnelles Verändern von Texten
(korrigieren, ergänzen, zusammenfügen usw.)

Wollen Sie wirklich ernsthaft Textverarbeitung betreiben, so sollten Sie sich auf dem professionellen Software-Markt umschauen. Sie werden sich wundern, was zum Beispiel TEXTOMAT von DATA BECKER leistet. Allerdings benötigen Sie dann auch ein Diskettenlaufwerk als externes Speichermedium. Für den gelegent-

lichen Hausgebrauch reichen aber auch selbstgestrickte Programme und ein Cassettenrecorder. Ein derartiges Programm kann entweder die wichtigsten Funktionen der professionellen Textverarbeitungssoftware simulieren (siehe Kapitel 5.4) oder nur für bestimmte Anwendungen geeignet sein (siehe Kapitel 5.3).

Ohne große Schwierigkeiten können auch bestimmte Tasten Ihres Rechners mit deutschen Sonderzeichen, wie zum Beispiel ü,ä,ö oder ß belegt werden. Gerade bei der Textverarbeitung ist dies ja sicherlich sinnvoll. In Kapitel 5.2 befindet sich dazu ein entsprechendes Programm. Eigentlich selbstverständlich ist die Möglichkeit, mit der SHIFT- und der CAPS LOCK-Taste auf Groß- bzw. Kleinbuchstaben umzuschalten. Dies sei nur kurz erwähnt, da Sie bei mathematischen Anwendungen auch sehr gut ohne diese Umschaltmöglichkeit zurecht kommen.

Zum Thema Drucker lassen sich generelle Empfehlungen nicht machen. Der Schneider Matrix Printer NLQ 401 liefert für sein Geld sicherlich ein sehr gutes Schriftbild, aber - egal welchen Drucker Sie haben - Sie werden immer einen finden, der noch schöner druckt. Insbesondere die Frage "Typenraddrucker oder Matrixprinter ?" ist immer eine Frage des Geldbeutels. Für den mehr professionellen Bereich ist ein Typenraddrucker natürlich geeigneter.

Ein letzter Hinweis noch zum Thema Monitor. Wenn Sie einen Farbmonitor haben, so probieren Sie am besten aus, mit welcher Farbkombination Sie gut arbeiten können. Vermeiden Sie jeglichen Farbenzauber. Ihre Augen werden es Ihnen danken. Einfacher haben Sie es da, wenn Sie einen Grünmonitor besitzen. Er eignet sich sehr gut für die Textverarbeitung, da er nicht vom Wesentlichen ablenkt.

5.2. Deutsche Sonderzeichen

Beim Vergleich der Tastatur Ihres Rechners mit der einer Schreibmaschine ist Ihnen sicherlich aufgefallen, daß da einiges anders ist. Für die Textverarbeitung ist das Fehlen der deutschen Umlaute ü,ä,ö,Û,Ä,Ö und des ß zu bemängeln. In Schneider-BASIC ist es jedoch recht unproblematisch, diese Sonderzeichen für bestimmte Tasten zu definieren. Man sollte sich dabei nur für Tasten entscheiden, deren ASCII-Wert mit dem entsprechenden Wert in der Character Code Tabelle des Druckers übereinstimmen. So wird gewährleistet, daß ein Umlaut auf dem Bildschirm auch als Umlaut ausgedruckt wird.

In dem folgenden Programm wird diese Tastenbelegung vollzogen. Der Umgang mit diesen Tasten bedarf jedoch einer gewissen Eingewöhnungsphase, da die belegten Tasten nur zum Teil an die Schreibmaschinentastatur erinnern.

Die Neubelegung erfolgt mit Hilfe der SYMBOL-Anweisung. Zu beachten ist dabei, daß die Parameter dieser Anweisung, welche bitabhängig von oben nach unten die 8x8 Punktmatrix des neuen Zeichens beschreiben, Hexadezimalzahlen sind. So bewirkt zum Beispiel der Parameter C3, daß die ersten und die letzten beiden Spalten der jeweiligen Zeile die aktuelle Schreibstiftfarbe und die mittleren

vier Spalten die Schreibflächenhintergrundfarbe erhalten,
da die Hexadezimalzahl C3 in binärer Schreibweise gleich

11000011

0 = Schreibflächenhintergrundfarbe
1 = Schreibstiftfarbe

ist.

Für jede der 8 Zeilen müssen Sie eine "passende" Hexadezimalzahl angeben. Im Bedienungshandbuch zu Ihrem CPC finden Sie eine hübsche Tabelle, in der das Binär- dem Hexadezimalsystem gegenübergestellt wird, so daß Sie sich aufwendige Umrechnungen ersparen können. Weiterhin erleichtert es die Definitionsarbeit, wenn Sie den in Ihrem Handbuch aufgeführten CPC-Zeichensatz zu Rate ziehen. Dort finden Sie auch die jeweiligen Zeichennummern (erster Parameter der SYMBOL-Anweisung).

Programm:

```
10 REM I12
20 CLS
30 PRINT "I12 - Programm zur Definition von"
40 PRINT TAB(7)"deutschen Sonderzeichen"
50 PRINT:PRINT:PRINT:PRINT:PRINT
60 PRINT TAB(9) "Bernd Kowal, 1985"
70 GOSUB 350
80 PRINT"Betaetigen Sie bitte langsam und"
90 PRINT"nacheinander folgende Tasten : "
100 PRINT:PRINT"ohne SHIFT und CTRL:  A,Ü,ö"
110 PRINT:PRINT"und mit SHIFT      :  $,A,Ü"
120 PRINT:PRINT"und mit CTRL       :  2 (links oben)"
130 SYMBOL AFTER 90
140 REM klein 'ue'
150 SYMBOL 125,&0,&66,&0,&66,&66,&66,&3E,&0
```

```

160 REM gross 'UE'
170 SYMBOL 93,&66,&0,&66,&66,&66,&66,&3C,&0
180 REM klein 'oe'
190 SYMBOL 124,&66,&0,&3C,&66,&66,&66,&3C,&0
200 REM gross 'OE'
210 SYMBOL 92,&BA,&6C,&C6,&C6,&6C,&38,&0
220 REM klein 'ae'
230 SYMBOL 123,&C6,&0,&7B,&C,&7C,&CC,&76,&0
240 REM gross 'AE'
250 SYMBOL 91,&5A,&3C,&66,&66,&7E,&66,&66,&0
260 REM scharfes 'S'
270 SYMBOL 126,&7C,&C6,&C6,&FC,&C6,&C6,&F8,&C0
280 PRINT:PRINT:PRINT
290 PRINT"Falls Ihnen das 'Design' der Umlaute"
300 PRINT"nicht gefallen sollte, ändern Sie"
310 PRINT"doch einfach die entsprechenden"
320 PRINT"SYMBOL-Anweisungen."
330 PRINT:PRINT:PRINT
340 PRINT"Ende des Programms":END
350 REM UP Warten
360 LOCATE 5,25
370 PRINT"Bitte eine Taste druecken !"
380 x$=INKEY$
390 IF x$="" THEN 380
400 CLS:RETURN

```

Im Programm werden keine Variablen benötigt

Programmbeschreibung:

Satz 10- 70 : Überschrift

Satz 80-120 : Ausgabe einer Benutzeranweisung

Satz 130 : Festlegung des Bereichs, in dem die deutschen Sonderzeichen definiert werden

Satz 140-270 : Definition der deutschen Sonderzeichen und Neubelegung der Tasten

Satz 280-330 : Ausgabe eines Kommentars

Satz 340 : Programmende

Satz 350-400 : Unterprogramm zum Warten

Programmergebnisse:

Immer dann, wenn Sie die deutschen Sonderzeichen benutzen wollen, können Sie die SYMBOL-Anweisungen anbringen. Dies kann - wie oben demonstriert - im Programmablauf geschehen oder auch im Direktmodus (auf der Kommandoebene).

5.3. Ein individueller Geschäftsbrief

Sie sind Geschäftsmann und Inhaber eines kleinen Ladens mit dem hübschen Namen "Handel Allerlei" in Hintertupfingen, Dorfstraße 1? Dann kommt es doch auch gelegentlich vor, daß ein Kunde das Begleichen einer Rechnung "vergibt". Diesem lieben Kunden (nennen wir ihn Herrn Geizig) müssen Sie dann eine Zahlungsaufforderung zusenden, die etwa so aussehen könnte:

(bitte umblättern)

Handel Allerlei
Dorfstraße 1

4999 Hintertupfingen

Heini Geizig
Hauptstraße 2

4998 Vordertupfingen Hintertupfingen, den 20.4.1985

Betr.: Lieferung einer Schubkarre vom 1.4.1985

Sehr geehrter Herr Geizig!

Sie haben immer noch nicht gezahlt. Das ist eine Unverschämtheit. Wenn der Betrag von 147.50 DM nicht bis zum 30.4.1985 auf unserem Konto eingegangen ist, werden wir zu Rechtsmitteln greifen müssen.

Mit freundlichen Grüßen



Sollte nun ein anderer Kunde ebenfalls mit seiner Zahlung im Rückstand sein, so würde es ausreichen, die unterstrichenen Teile des Briefes lediglich auszuwechseln.

Genau so können Sie verfahren, wenn Sie Ihren CPC, einen geeigneten Drucker und ein entsprechendes Programm einsetzen. Das passende Programm zu dem obigen Geschäftsbrief finden Sie auf den nächsten Seiten. Die unterstrichenen Teile des Briefes geben Sie als Variablen über INPUT-Statements ein; der Rest des Briefes ist hingegen fest im Programm verankert. Falls Ihnen der Text des Briefes nicht gefallen sollte (in Düsseldorf sollen die Geschäftsleute andere Gepflogenheiten kennen als in Hintertupfingen), verändern Sie ihn einfach. Das ist eine Arbeit, die Sie nur einmal erledigen müssen. Bedenken Sie aber, daß zum Beispiel im folgenden Programm jede Textzeile genau 56 Anschläge hat. Das macht einen professionellen Eindruck und ist durch zusätzliche Leerzeichen nicht schwer zu programmieren.

Auf diese Art und Weise können Sie Ihre gesamte Geschäftskorrespondenz durchforsten. Sie werden sich wundern, wieviele Textteile sich in Ihren Briefen wiederholen. Und - die Individualität Ihrer Geschäftsbriefe bleibt erhalten.

Programm:

```
10 REM I13
20 CLS
30 PRINT "I13 - Programm zum Schreiben eines"
40 PRINT TAB(7)"individuellen Geschaefts-"
50 PRINT TAB(7)"briefes (Mahnung)"
60 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
70 PRINT TAB(9) "Bernd Kowal, 1985"
80 GOSUB 680
90 REM deutsche Sonderzeichen
100 SYMBOL AFTER 90
```

```

110 SYMBOL 125,&O,&66,&O,&66,&66,&66,&3E,&O
120 SYMBOL 93,&66,&O,&66,&66,&66,&66,&3C,&O
130 SYMBOL 124,&66,&O,&3C,&66,&66,&66,&3C,&O
140 SYMBOL 92,&BA,&6C,&C6,&C6,&6C,&6C,&38,&O
150 SYMBOL 123,&C6,&O,&78,&C,&7C,&CC,&76,&O
160 SYMBOL 91,&5A,&3C,&66,&66,&7E,&66,&66,&O
170 SYMBOL 126,&7C,&C6,&C6,&FC,&C6,&C6,&FB,DO
180 REM Eingaben
190 MODE 2
200 PRINT"Name des Kunden : "
210 INPUT"Vorname ";vn#
220 INPUT"Nachname ";nn#
230 PRINT:PRINT"Ist der Kunde Mann oder Frau"
240 INPUT"(geben Sie m oder f ein) ";g#
250 PRINT:PRINT"Adresse des Kunden : "
260 INPUT"Straße mit Hausnummer ";sh#
270 INPUT"Ort mit Postleitzahl ";op#
280 PRINT:INPUT"Absendedatum der Mahnung ";ad#
290 PRINT:INPUT"Betreff ";b#
300 PRINT:INPUT"Um welchen DM-Betrag geht es ";dm
310 PRINT:PRINT"Bis wann werden Zahlungen"
320 INPUT"berücksichtigt ";zd#
330 REM Ausgabe
340 CLS:PRINT"Ausgabe auf";TAB(30)"Eingabe"
350 PRINT:PRINT"Bildschirm";TAB(33)"0"
360 PRINT:PRINT"Drucker";TAB(33)"8"
370 PRINT:PRINT:INPUT "Ihre Wahl ";a:CLS
380 PRINT #a:PRINT #a:PRINT #a:PRINT #a
390 PRINT #a, "Handel Allerlei"
400 PRINT #a, "Dorfstraße 1"
410 PRINT #a:PRINT #a, "4999 Hintertupfingen"
420 PRINT #a:PRINT #a:PRINT #a
430 PRINT #a, vn#;" ";nn#
440 PRINT #a, sh#
450 PRINT #a:PRINT #a, op#;
460 q=57-21-LEN(ad#)
470 PRINT #a, TAB(q)"Hintertupfingen, den ";ad#
480 PRINT #a:PRINT #a:PRINT #a:PRINT #a:PRINT #a
490 PRINT #a, "Betr. : ";b#
500 PRINT #a:PRINT #a
510 IF a=0 THEN GOSUB 680
520 IF g#="m" THEN 540
530 PRINT #a,"Sehr geehrte Frau ";nn#;"!":GOTO 550
540 PRINT #a,"Sehr geehrter Herr ";nn#;"!"
550 PRINT #a:PRINT #a, "Sie haben immer noch nicht ";
560 PRINT #a, "gezahlt. Das ist eine Unver-";
570 PRINT #a, "schämtheit. Wenn der Betrag von ";

```

```
580 PRINT #a, USING "####.##";dm;
590 PRINT #a, " DM nicht bis zum"
600 PRINT #a, USING"ö        ö";zd#;
610 PRINT #a," auf unserem Konto eingegangen ";
620 PRINT #a, "ist, werden "
630 PRINT #a, "wir zu Rechtsmitteln greifen müssen."
640 PRINT #a:PRINT #a:PRINT #a:PRINT #a:PRINT #a
650 PRINT #a,"Mit freundlichen Grüßen"
660 PRINT #a:PRINT #a:PRINT #a:PRINT #a:PRINT #a
670 GOSUB 680:MODE 1:END
680 REM UP Warten
690 LOCATE 5,25
700 PRINT"Bitte eine Taste druecken !"
710 x#=#INKEY#
720 IF x#="" THEN 710
730 CLS:RETURN
```

Variablenliste:

a = Ausgabegerät (0 für Bildschirm, 8 für Drucker)

ad\$ = Absendedatum des Briefes

b\$ = Betreff

dm = der noch nicht gezahlte DM-Betrag

g\$ = Geschlecht des Kunden

nn\$ = Nachname des Kunden

op\$ = Ort mit Postleitzahl

q = berechneter Parameter für die TAB-Funktion bei der
Ausgabe des Absendedatums

sh\$ = Straße mit Hausnummer

v\$ = Vorname des Kunden

zd\$ = Datum, bis zu dem Zahlungen berücksichtigt werden

Programmbeschreibung:

Stz 10- 80 : Überschrift

Satz 90-170 : Definition der deutschen Sonderzeichen
(siehe Kapitel 5.2)

Satz 180-320 : Eingabe der "unterstrichenen" Teile des Briefes. Dazu wird in Satz 190 der Bildschirmmodus auf 80 Zeichen pro Zeile gesetzt (für längere Eingaben).

Satz 330-660 : Ausgabe des kompletten Briefes

In den Sätzen 340-370 können Sie auswählen, ob der Brief auf dem Bildschirm oder auf dem Drucker ausgegeben werden soll. Beachten Sie, daß eine falsche Eingabe für a im Programm nicht "abgefangen" wird.

In Satz 460 wird der Parameter der TAB-Funktion für die Ausgabe des Absendedatums in Satz 470 berechnet, damit die gesamte Ausgabezeile genau 56 Zeichen lang ist.

Im Falle einer Ausgabe auf dem Bildschirm wird in Satz 510 der Programmablauf unterbrochen (Unterprogramm "Warten"), da der Brief mehr Zeilen als der Bildschirm hat.


In den Sätzen 520-540 wird die Anrede in Abhängigkeit von der Eingabe "Geschlecht des Kunden" ausgegeben.

Die Ausgabe des DM-Betrages und des Datums in den Sätzen 580-600 erfolgt formatiert.

Wenn Sie mit höheren DM-Beträgen zu tun haben, müssen Sie das Format ändern. Zu beachten ist dabei, daß dann die entsprechende Zeile mehr Ausgabezeichen hat. Für das Datum sind maximal 10 Zeichen vorgesehen, ausreichend etwa für die Eingabe "24.12.1985".

Satz 670 : Sprung ins Unterprogramm "Warten", Änderung des Bildschirmmodus und Programmende

Satz 680-730 : Unterprogramm "Warten"



5.4. Textverarbeitung

Im Vergleich zur professionellen Textverarbeitungs-Software weist das Programm in diesem Kapitel natürlich Mängel auf. Aber es ist ja schließlich auch nur für den Hausgebrauch und nicht fürs Büro erstellt worden.

Folgende Bearbeitungsschritte können Sie mit dem Programm durchführen:

- Texte eingeben (über die Tastatur)
- Texte ansehen (auf dem Bildschirm)
- Texte korrigieren (Textzeilen verbessern, einfügen oder löschen)
- Texte einlesen (von Cassette oder Diskette)
- Texte abspeichern (auf Cassette oder Diskette)
- Texte drucken

Wollen Sie ein Diskettenlaufwerk als externes Speichermedium einsetzen, müssen Sie die folgenden Programmzeilen ändern bzw. einfügen:

```
1745 INPUT "Wie heisst der Text ";n$  
1750 OPENIN n$
```

Ansonsten sind keine Änderungen bezüglich des zu verwendenden Speichers notwendig.

Die Anzahl der Zeichen pro Zeile können Sie selbst wählen (bis zu 70 Zeichen). Ebenfalls können Sie die Anzahl der Zeilen pro Seite und die Breite des linken Randes bei der Druckausgabe frei bestimmen. Weitere Formatangaben werden nicht gebraucht. So wie Sie den Text eingeben und auf dem Bildschirm sehen, wird er auch ausgedruckt.

Im Gegensatz dazu geschieht das Formatieren, also das Aufteilen eines Textes zum anschließenden Druck, bei vielen professionellen Textverarbeitungsprogrammen für den CPC in einem separaten Arbeitsgang. Der Monitor ist nun einmal nicht dazu in der Lage, alles darzustellen, was auch der Drucker drucken kann. Weiterhin ist bei vielen Textverarbeitungs-Paketen nicht auf die Länge einer Textzeile zu achten. Eine entsprechende Angabe können Sie dann vor der Druckausgabe machen. Natürlich können Sie sich den formatierten Text auch erst einmal auf dem Bildschirm zeigen lassen. Zusammenfassend läßt sich sagen, daß Sie bei professionellen Programmen - neben einer komfortableren Handhabung des Textes - viele Möglichkeiten haben, einen eingegebenen Text aufs Papier zu bringen.

Bei dem hier vorliegenden Programm ist das ein wenig anders. Zwar wird die Anzahl der Zeichen pro Zeile bei der Druckausgabe auch begrenzt, Sie müssen aber bei der Texteingabe selbst darauf achten, nicht zuviele Zeichen pro Zeile einzugeben. Es wird lediglich eine senkrechte Linie auf dem Bildschirm gezeichnet, die Ihnen anzeigt, wann eine Zeilenlänge erreicht ist. Das ist immer dann der

Fall, wenn der Cursor diese Linie berührt. Durch die Betätigung der ENTER-Taste kann dann die nächste Zeile geschrieben werden. In welcher Zeile und auf welcher Seite Sie sich befinden, wird natürlich während der Texteingabe angezeigt. Zu beachten ist bei diesem Programm weiterhin, daß Anführungszeichen zu Zeilenumbrüchen führen. Für Hervorhebungen ist daher das Apostroph zu verwenden.

Insgesamt kann ein mit dem vorliegenden Programm bearbeiteter Text maximal 300 Textzeilen umfassen (unabhängig von der Seitenaufteilung). Das sind ca. 6 Schreibmaschinenseiten. Indem Sie die Sätze 180 und 570 ändern, können Sie auch mehr Textzeilen eingeben. Sie müssen dann nur auf die begrenzte Speicherkapazität Ihres Rechners achten. Entweder unterbrechen Sie dazu den Programmablauf und fragen mit der Funktion FRE den freien Speicherplatz ab, oder Sie nehmen diese Funktion in geeigneter Weise in den Programmablauf auf, um ein "Überlaufen" des Speichers zu verhindern.

Wollen Sie einen abgespeicherten Text ergänzen, so können Sie diesen einlesen und dann neue Textzeilen hinzufügen. Die Formatangaben "Zeichen pro Zeile" und "Zeilen pro Seite" müssen Sie dann aber übernehmen. Nur die Angabe "Zeilen pro Seite" ist bei der Druckausgabe noch zu ändern.

Es ist nicht möglich, einen Text zusätzlich zu einem eingegebenen Text einzulesen. Der eingegebene Text würde dann überschrieben. Jedoch kann man einen eingelesenen Text an jeder beliebigen Stelle ergänzen.

Ansonsten ist das Programm weitestgehendst selbsterklärend. Am besten Sie probieren es gleich einmal aus. Und ... ziehen Sie nicht den Stecker bevor Sie Ihren Text gesichert haben ...

Programm:

```
10 REM I14
20 CLS
30 PRINT "I14 - Programm zur Textverarbeitung"
40 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
50 PRINT TAB(9) "Bernd Kowal, 1985"
60 GOSUB 2490
70 REM
80 REM deutsche Sonderzeichen
90 REM
100 SYMBOL AFTER 90
110 SYMBOL 125,&0,&66,&0,&66,&66,&66,&3E,&0
120 SYMBOL 93,&66,&0,&66,&66,&66,&66,&3C,&0
130 SYMBOL 124,&66,&0,&3C,&66,&66,&66,&3C,&0
140 SYMBOL 92,&BA,&6C,&C6,&C6,&6C,&3B,&0
150 SYMBOL 123,&C6,&0,&78,&C,&7C,&CC,&76,&0
160 SYMBOL 91,&5A,&3C,&66,&66,&7E,&66,&66,&0
170 SYMBOL 126,&7C,&C6,&C6,&FC,&C6,&C6,&FB,&C0
180 DIM t$(300)
190 OPENOUT "d":MEMORY HIMEM-1:CI OSEOUT
200 REM
210 REM Menue
220 REM
230 PRINT TAB(15) "Menue":PRINT
240 PRINT TAB(33) "Eingabe":PRINT
250 PRINT "Texte":PRINT
260 PRINT " eingeben";TAB(36) "1":PRINT
270 PRINT " ansehen";TAB(36) "2":PRINT
280 PRINT " korrigieren";TAB(36) "3":PRINT
290 PRINT " einlesen";TAB(36) "4":PRINT
300 PRINT " abspeichern";TAB(36) "5":PRINT
310 PRINT " drucken";TAB(36) "6":PRINT
320 PRINT "Arbeit beenden";TAB(36) "7"
330 PRINT:PRINT:PRINT:INPUT "Ihre Wahl ";a:CLS
340 IF a<1 OR a>7 THEN 230
350 ON a GOSUB 380,650,850,1730,1830,1950,2260
360 MODE 1:GOTO 230
370 REM
380 REM Texte eingeben
390 REM
400 IF zz>0 THEN 430
410 INPUT "Wieviele Zeichen pro Zeile ";zz:PRINT
420 IF zz<1 OR zz>70 THEN 410
430 IF zs>0 THEN 470
440 INPUT "Wieviele Zeilen pro Seite ";zs:PRINT
450 PRINT:PRINT"Geben Sie fuer eine Zeile '##' ein,"
460 PRINT"so wird die Eingabe beendet.":GOSUB 2490
470 GOSUB 2370
```

```

480 WINDOW #0,1,7+zz,4,25
490 IF s=0 THEN s=1
500 IF z<1 THEN 550
510 k1=s*zs-zs+1:k2=s*zs-zs+z
520 FOR i=k1 TO k2
530 PRINT #0, USING " ##      ";i-s*zs+zs;
540 PRINT #0, t$(i):NEXT i
550 IF z=zs THEN z=0:s=s+1
560 z=z+1:az=az+1
570 IF az=300 THEN PRINT:PRINT"Es ist kein
      Speicherplatz mehr verfuegbar":RETURN
580 LOCATE #1, zz/2+7,1:PRINT #1, s
590 PRINT #0, USING " ##      ";z;
600 LINE INPUT t$(az)
610 IF t$(az)<>"##" THEN 550
620 t$(az)=" "
630 z=z-1:az=az-1:WINDOW #0,1,40,1,25:RETURN
640 REM
650 REM Texte ansehen
660 REM
670 INPUT"Welche Seite ";x
680 IF x=0 THEN 670
690 l=0:GOSUB 2370
700 WINDOW #0,1,7+zz,4,25
710 LOCATE #1, zz/2+7,1:PRINT #1, x
720 k1=x*zs-zs+1:k2=x*zs
730 FOR i=k1 TO k2
740 l=l+1
750 PRINT #0, USING " ##      ";i-x*zs+zs;
760 PRINT #0, t$(i)
770 IF l=22 THEN l=0:GOSUB 2530
780 NEXT i:GOSUB 2530:MODE 1
790 WINDOW #0,1,40,1,25
800 INPUT"Nach eine Seite (j/n) ";a$
810 IF a$="n" THEN RETURN
820 IF a$<>"j" THEN 800
830 PRINT:PRINT:GOTO 670
840 REM
850 REM Texte korrigieren
860 REM
870 PRINT TAB(15)"Menue":PRINT:PRINT
880 PRINT TAB(30)"Eingabe":PRINT:PRINT
890 PRINT"Textzeilen":PRINT:PRINT
900 PRINT" verbessern";TAB(33)"1":PRINT
910 PRINT" einfuegen";TAB(33)"2":PRINT
920 PRINT" loeschen";TAB(33)"3":PRINT:PRINT
930 PRINT"Korrektur beenden";TAB(33)"4":PRINT
940 PRINT:PRINT:INPUT"Ihre Wahl ";a:MODE 2
950 IF a<1 OR a>4 THEN 870

```

```

960 IF a<4 THEN INPUT"Welche Seite ";x:PRINT
970 ON a GOTO 990,1150,1470,1700
980 REM
990 REM verbessern
1000 REM
1010 PRINT"Welche Zeile soll"
1020 INPUT"verbessert werden ";y
1030 i=(x-1)*zs+y
1040 PRINT:PRINT"Die Zeile davor : "
1050 PRINT t$(i-1);:PRINT TAB(zz+1) CHR$(211):PRINT
1060 PRINT"Die zu verbessernde Zeile : "
1070 PRINT t$(i);:PRINT TAB(zz+1) CHR$(211):PRINT
1080 PRINT"Die Zeile danach : "
1090 PRINT t$(i+1);:PRINT TAB(zz+1) CHR$(211):PRINT
1100 PRINT:PRINT"Wie soll die";y;".te Zeile lauten ?"
1110 LOCATE zz+1,17:PRINT CHR$(211):PRINT
1120 LOCATE 1,17:LINE INPUT t$(i)
1130 MODE 1:GOTO 870
1140 REM
1150 REM einfuegen
1160 REM
1170 PRINT"Zwischen welchen Zeilen soll etwas"
1180 PRINT"eingefuegt werden ?"
1190 PRINT:INPUT"Kleinere Zeilennummer ";y1
1200 z1=y1:y1=(x-1)*zs+y1
1210 PRINT"Die Zeile lautet : "
1220 PRINT t$(y1);:PRINT TAB(zz+1) CHR$(211)
1230 PRINT:INPUT"Groessere Zeilennummer ";y2
1240 IF y2-z1=1 THEN 1270
1250 PRINT:PRINT"Es muessen zwei beieinanderliegende"
1260 PRINT"Zeilen angegeben werden.":GOTO 1170
1270 y2=(x-1)*zs+y2
1280 PRINT"Die Zeile lautet : "
1290 PRINT t$(y2);:PRINT TAB(zz+1) CHR$(211)
1300 PRINT:PRINT:PRINT"Wieviele Zeilen sollen einge-"
1310 INPUT"fuegt werden ";y3
1320 az=az+y3:z=z+y3
1330 IF z>zs THEN z=z-zs:s=s+1
1340 FOR i=az TO y2 STEP -1
1350 t$(i+y3)=t$(i)
1360 NEXT i:GOSUB 2370
1370 WINDOW #0,1,7+zz,4,25
1380 FOR i=y1+1 TO y1+y3
1390 z1=z1+1
1400 LOCATE #1, zz/2+7,1:PRINT #1, x
1410 PRINT #0, USING " ## " ;z1;
1420 LINE INPUT t$(i)

```

```

1430 NEXT i
1440 WINDOW #0,1,40,1,25
1450 MODE 1:GOTO 870
1460 REM
1470 REM loeschen
1480 REM
1490 PRINT"Welche Zeilen sollen geloescht werden?"
1500 PRINT:INPUT"von: Zeilennummer ";y1
1510 z1=y1:y1=(x-1)*zs+y1
1520 PRINT"Die Zeile lautet : "
1530 PRINT t$(y1);:PRINT TAB(zz+1) CHR$(211)
1540 PRINT:INPUT"bis: Zeilennummer ";y2
1550 IF y2>=z1 THEN 1580
1560 PRINT:PRINT"Die zweite Zeilennummer muss groesser"
1570 PRINT"als die erste sein.":GOTO 1500
1580 y2=(x-1)*zs+y2
1590 PRINT"Die Zeile lautet : "
1600 PRINT t$(y2);:PRINT TAB(zz+1) CHR$(211)
1610 PRINT:PRINT:PRINT TAB(10)"Ich loesche !"
1620 GOSUB 2490
1630 y3=y2-y1+1
1640 FOR i=y1 TO az
1650 t$(i)=t$(i+y3)
1660 NEXT i
1670 az=az-y3:z=z-y3
1680 IF z<1 THEN z=z+zs:s=s-1
1690 MODE 1:GOTO 870
1700 REM Korrektur beenden
1710 MODE 1:RETURN
1720 REM
1730 REM Texte einlesen
1740 REM
1750 OPENIN ""
1760 INPUT #9,zz,zs,s,z,az
1770 FOR i=1 TO az
1780 INPUT #9,t$(i)
1790 NEXT i
1800 CLOSEIN
1810 CLS:RETURN
1820 REM
1830 REM Texte abspeichern
1840 REM
1850 PRINT"Wie soll der abzuspeichernde"
1860 INPUT"Text heissen ";n$:PRINT
1870 OPENOUT n$
1880 PRINT #9,zz,zs,s,z,az
1890 FOR i=1 TO az
1900 WRITE #9,t$(i)

```

```

1910 NEXT i
1920 CLOSEOUT
1930 CLS:RETURN
1940 REM
1950 REM Texte drucken
1960 REM
1970 PRINT"Soll der Wert fuer die Anzahl"
1980 PRINT"der Zeilen pro Seite geaendert"
1990 INPUT"werden (j/n) ";a#
2000 IF a#="n" THEN 2030
2010 IF a#<>"j" THEN 1970
2020 PRINT:INPUT"Wie gross soll er sein ";zs
2030 PRINT:PRINT"Wie breit soll der linke Rand sein ?"
2040 INPUT"Anzahl der Leerzeichen ";lr
2050 WIDTH zz+lr
2060 PRINT:PRINT"Soll der gesamte Text gedruckt"
2070 INPUT"werden (j/n) ";a#
2080 IF a#="n" THEN 2140
2090 IF a#<>"j" THEN 2060
2100 FOR i=1 TO az
2110 PRINT #8, TAB(lr+1) t$(i)
2120 NEXT i
2130 GOTO 2240
2140 PRINT:PRINT"Welche Seite soll"
2150 INPUT"gedruckt werden ";x
2160 k1=x*zs-zs+1;k2=x*zs
2170 FOR i=k1 TO k2
2180 PRINT #8, TAB(lr+1) t$(i)
2190 NEXT i
2200 PRINT:PRINT"Soll eine weitere Seite"
2210 INPUT"gedruckt werden (j/n) ";a#
2220 IF a#="j" THEN 2140
2230 IF a#<>"n" THEN 2200
2240 CLS:RETURN
2250 REM
2260 REM Arbeit beenden
2270 REM
2280 PRINT"Haben Sie auch nicht vergessen den"
2290 PRINT"Text abzuspeichern ?"
2300 PRINT"Ist das der Fall, so geben Sie einfach"

```

```
2310 PRINT"ein 'oh' ein. Ansonsten reicht die"
2320 PRINT"Leertaste.":INPUT a$
2330 IF a$="oh" THEN GOSUB 1830
2340 PRINT:PRINT:PRINT
2350 PRINT"Sie koennen mich jetzt ausschalten."
2360 PRINT:PRINT:PRINT TAB(8)"Auf Wiedersehen":END
2370 REM
2380 REM Unterprogramm Bildschirmmaske
2390 REM
2400 MODE 2
2410 WINDOW #1,1,80,1,3
2420 PRINT #1,TAB(zz/2)"Seite"
2430 PRINT #1,"Zeile";
2440 WINDOW #2,8+zz,9+zz,4,25
2450 FOR i=1 TO 25
2460 PRINT #2, CHR$(211):NEXT i
2470 RETURN
2480 REM
2490 REM Unterprogramm Warten
2500 REM
2510 LOCATE 5,25
2520 PRINT"Bitte eine Taste druecken !"
2530 x$=INKEY$
2540 IF x$="" THEN 2530
2550 CLS:RETURN
```

Variablenliste:

a = Antwort "Menüauswahl"

a\$ = Antwortstring

az = absolute Zeilenanzahl bzw. Zeilnummer (bezogen
auf alle im Programmspeicher befindlichen Text-
zeilen)

i = Laufindex und absolute Zeilennummer im Programm-
teil "verbessern"

k1 = untere Begrenzung einer Programmschleife

k2 = obere Begrenzung einer Programmschleife

l = Hilfsgröße im Programmteil "Texte ansehen"

lr = Breite des linken Randes bei der Druckausgabe

n\$ = Name des abgespeicherten Textes

s = Seitenzahl

t\$(i) = Textzeile

x = Seitenzahl in den Programmteilen "ansehen" und
"korrigieren"

y = Nummer einer zu verbessernden Zeile

y1 = kleinere Zeilennummer in den Programmteilen "ein-
fügen" und "löschen"

y2 = größere Zeilennummer in den Programmteilen "ein-
fügen" und "löschen"

y3 = Anzahl der einzufügenden Zeilen bzw. Anzahl der
zu löschenden Zeilen

z = Zeilennummer einer Seite

z1 = Zeilennummer einer Seite im Programmteil "korri-
gieren"

zs = Zeilen pro Seite

zz = Zeichen pro Zeile

Programmbeschreibung:

Satz 10- 60 : Überschrift

Satz 70-170 : Definition der deutschen Sonderzeichen

Satz 180 : Speicherplatzreservierung für den Text

Satz 190 : Durch diese Programmzeile werden die noch
zu eröffnenden Textdateien vor einer zeit-
raubenden Speicher-Reorganisation ge-
schützt. Die Zeile kann aber auch wegge-
lassen werden, da sie bei einem wiederhol-
ten Programmstart eine Fehlermeldung in
Satz 100 verursacht. Es ist jedoch auch
möglich, die Sätze 70-170 vor einem zweiten

Programmstart mit dem Kommando DELETE zu löschen oder den Satz 100 in ein REM-Statement umzuwandeln:

Satz 200-350 : Auswahlmenü und entsprechende Programmverzweigungen in die einzelnen Unterprogramme

Satz 360 : Da in den Unterprogrammen häufig der Bildschirmmodus gewechselt wird, geschieht dies auch hier, um eine gleichbleibende Menüausgabe zu gewährleisten

Satz 370-630 : Unterprogramm "Texte eingeben"

370-390 : Kommentar

400-460 : Belegung der Parameter zz und zs mit Werten, falls dies noch nicht geschehen ist, und Ausgabe einer Erläuterung für die Beendigung der Eingabe

470 : Verzweigung zum Unterprogramm "Bildschirmmaske"

480 : Definition eines Bildschirmbereichs für die Texteingabe (die rechte Fensterbegrenzung ist abhängig von dem Parameter "Zeichen pro Zeile")

490 : Bei der Eingabe der allerersten Zeile (dann ist s=0) wird die Variable s gleich 1 gesetzt.

500-540 : Der Text einer teilweise beschriebenen Seite wird ausgegeben

550 : Ist eine Seite voll (dann ist z=zs), wird eine neue Seite "aufgeschlagen" (s=s+1) und die Variable z gleich 0 gesetzt

560 : Vor jeder Zeileneingabe werden die Variablen z und az um 1 erhöht

- 570 : Ausgabe eines Kommentars und Rücksprung zum Menü, wenn sich 300 Textzeilen im Programmspeicher befinden
- 580-590 : Ausgabe der Seitenzahl und der Zeilennummer (bezüglich einer Seite)
- 600 : Eingabe einer Textzeile. Der Index einer Textzeile ist immer die absolute Zeilennummer az.
- 610-630 : Bei der Eingabe des Abbruchkriteriums werden die Variablen z und az um 1 vermindert (da das Abbruchkriterium ja nicht zum eigentlichen Text gehört), der Bildschirmbereich wird neu definiert und es erfolgt ein Rücksprung aus diesem Unterprogramm. Ansonsten wird zur weiteren Texteingabe zum Satz 550 verzweigt.

Satz 640-830 : Unterprogramm "Texte ansehen"

- 640-660 : Kommentar
- 670-680 : Eingabe einer Seitenzahl
- 690-700 : Die Hilfsvariable l wird gleich 0 gesetzt, es wird ins Unterprogramm "Bildschirmmaske" verzweigt und das Textfenster wird neu definiert
- 710 : Ausgabe der Seitenzahl
- 720 : Die (absoluten) Zeilennummern des Anfangs und des Endes einer Seite werden bestimmt
- 730-780 : Ausgabe des Textes. Nach der Ausgabe von 22 Zeilen (siehe Hilfsvariable l) wird ins Unterprogramm "Warten" verzweigt. Auf die Erläuterung "Bitte eine Taste drücken" wird hier verzichtet.
- 790-830 : Das Textfenster wird neu defi-

niert und es können weitere Seiten angesehen werden. Ansonsten erfolgt ein Rücksprung aus dem Unterprogramm.

Satz 840-1710 : Unterprogramm "Texte korrigieren"

840- 860 : Kommentar

870- 970 : Auswahlmenü und entsprechende Programmverzweigungen, sowie Anforderung einer Seitenzahl

980-1130 : Programmteil "verbessern"

Eine Zeilennummer wird angefordert (Satz 1010-1020), die dazugehörige absolute Zeilennummer wird berechnet (1030), die zu verbessernde Zeile wird neben den benachbarten Zeilen ausgegeben (1040-1090) und dann anschließend verbessert (1100-1120). Schließlich wird der Bildschirm geräumt und es erfolgt eine Verzweigung zum Menü.

1140-1450 : Programmteil "einfügen"

Zeilennummern werden angefordert, die absoluten Kennzahlen werden berechnet und die Textzeilen werden ausgegeben (1170-1290). Dann wird ein Wert für die Anzahl der einzufügenden Zeilen eingegeben (1300-1310), die Variablen z und az werden berichtigt (1320-1330, dabei wird in Satz 1330 berücksichtigt,

daß die zusätzlichen Zeilen eine neue Seite erfordern können), die Indizes der vorhandenen Textzeilen werden korrigiert (1340-1360) und die neuen Textzeilen können mit Hilfe der Bildschirmfensterdefinition und dem Unterprogramm "Bildschirmmaske" eingefügt werden (1360-1440). Anschließend wird wieder zum Menü verzweigt (1450).

1460-1690 : Programmteil "löschen"

Es werden Zeilennummern angefordert, umgerechnet und die entsprechenden Textzeilen werden ausgegeben (1490-1600). Dann wird nach einer Kommentarausgabe (1610) die Anzahl der zu löschenden Zeilen berechnet (1630), welche dann überschrieben werden (1640-1660). Anschließend werden die Variablen z und az neu berechnet (1670), wobei wiederum berücksichtigt wird, daß sich die Seitenzahl ändern kann (1680). Zum Schluß erfolgt ein Rücksprung zur Menüauswahl (1690).

1700-1710 : Beendigung der Korrekturen durch Löschen des Bildschirms und Rücksprung aus dem Unterprogramm

Satz 1720-1810 : Unterprogramm "Texte einlesen"

Das Einlesen erfolgt für die numerischen und die Textvariablen in getrennten Arbeitsschritten. Für den Diskettenbetrieb

ist der Satz 1750 zu ändern, da dann ein konkreter Dateiname anzugeben ist.

Satz 1820-1930 : Unterprogramm "Texte abspeichern"

Das Abspeichern vollzieht sich genauso wie das Einlesen. Den Namen des abzuspeichernden Textes können Sie frei wählen. Zu beachten ist jedoch, daß hier in Satz 1900 nicht mit der PRINT- sondern mit der WRITE-Anweisung gearbeitet wird, d. h. die Textzeilen werden in Doppelanführungszeichen eingeschlossen abgespeichert, um zu verhindern, daß zum Beispiel abgespeicherte Kommata zu unerwünschten Zeilenumbrüchen bei der Ausgabe führen.

Satz 1940-2240 : Unterprogramm "Texte drucken"

Nach einigen Formatangaben (1970-2040), wird die Anzahl der Zeichen pro Zeile für die Druckausgabe begrenzt (2050) und es wird entweder der ganze Text oder jeweils eine Seite ausgedruckt. Der Satz 2160 ist dabei identisch mit dem Satz 720.

In diesem Programmteil können auch noch zusätzliche Anweisungen für die Druckersteuerung (Druckermenü) eingefügt werden. Insbesondere die TAB-Anweisung macht bei längeren Textzeilen Probleme. Man nimmt dann besser druckerspezifische Anweisungen wie CHR\$(27);'I';CHR\$(n); (Epson-Drucker), wobei "n" den Spaltenwert der Ausgabe angibt.

Satz 2250-2360 : Unterprogramm "Arbeit beenden"

Falls Sie es versäumt haben sollten Ihren Text abzuspeichern, können Sie dies hier noch tun. Ansonsten ist das Programmende erreicht.

Satz 2370-2470 : Unterprogramm "Bildschirmmaske"

Mit Hilfe der Definition zweier Bildschirmfenster wird die "Maske" erstellt. Das erste Fenster (definiert in Satz 2410) dient der Ausgabe der aktuellen Zeilen- und Seitenzahl. Im zweiten Fenster (definiert in Satz 2440) wird die Zeilenbegrenzung eingezeichnet (2450-2460). Die Definition der Textfenster hat den Vorteil, daß auch bei einem "Wegrollen" der Textzeilen (Bildschirmbereich #0), die "Maske" erhalten bleibt.

Satz 2480-2550 : Unterprogramm "Warten"

5.5. Ideenfundgrube

- Geburtstagsgrüße
- Einladung zur Party
- Standard- und Geschäftsbriefe (Rechnungen, Mahnungen etc.)
- Bewerbungen (Anschreiben und Lebenslauf)
- individuelle Werbebriefe
- Zeugnisse und Urkunden
- Adressen speichern und ausdrucken
- Ausdruck selbstklebender Computer-Etiketten
- Referate, Diplom- und Doktorarbeiten professionell gestalten
- und noch vieles mehr

6. Der Staat und das Geld

6.1. Vorbemerkung

Wahrscheinlich gehören Sie auch zu den Leuten, die Monat für Monat Steuern und Sozialversicherungsbeiträge zahlen. Dazu sind Sie in der Regel ja auch verpflichtet. Wer aber Pflichten hat, der hat auch Rechte. So haben Sie zum Beispiel das Recht, Ihr Steuerkonto nach Ablauf eines Jahres abzurechnen, d. h. einen Antrag auf Lohnsteuerjahresausgleich zu stellen. Weiterhin werden Sie irgendwann einmal in den "Genuß" kommen, Rente zu beziehen. Ob dies ein Genuß sein wird bleibt natürlich fraglich. Es ist daher auf jeden Fall vorteilhaft, über sein Rentenkonto Bescheid zu wissen.

"Was nützt mir dabei mein Computer ?" werden Sie vielleicht fragen. Nun ja, die Antwort ist einfach: er hilft Ihnen dabei.

Natürlich müssen Sie Ihre persönlichen Angaben selbst zusammenstellen, um sie dann in den Rechner eingeben zu können. Damit hat es sich dann aber auch schon. Der Rest, die eigentliche Berechnung also, wird vom Computer erledigt. Und diese Berechnungen können recht aufwendig sein, wenn Sie beispielsweise an die Ermittlung des durchschnittlichen Vomhundertsatzes bei der Rentenberechnung oder an die Berechnung der Vorsorgepauschale beim Lohnsteuerjahresausgleich denken. Außerdem - und dies gilt insbesondere für den Lohnsteuerjahresausgleich - gehen Ihnen die Informationen eines auf Ihre Bedürfnisse abgestimmten Programms nicht wieder verloren. Sie müssen sich also nicht immer erneut durch die Materie "hindurchbeißen" (der Mensch ist vergeßlich), sondern es reicht, gesetzgeberische Änderungen zu berücksichtigen. Schließlich und endlich können Sie behördliche Bescheide auch besser kontrollieren, wenn Sie zu jeder Zeit und ohne großen Aufwand Ihre eigenen Berechnungen durchführen können.

Ein grundsätzliches Problem besteht nur darin, daß die hier vorgestellten Programme nicht jede Kleinigkeit berücksichtigen. Es ist im Rahmen dieses Buches schier unmöglich, die programmtechnischen Möglichkeiten Ihrer Finanzbehörde oder Ihrer Landesversicherungsanstalt nachzuvollziehen. Sie bleiben daher aufgefordert, das in den folgenden Programmen dargebotene "Gerüst" auf Ihre persönlichen Bedürfnisse hin zu überprüfen und gegebenenfalls Programmweiterungen vorzunehmen.

6.2. Die Rentenberechnung

Grundlage der Rentenberechnung ist die Rentenformel. In ihr ist festgelegt, wie im einzelnen die vom Versicherten erbrachte Leistung zusammen mit anderen Faktoren zur Rentenberechnung herangezogen wird. Für die Berechnung einer Versichertenrente ohne Berücksichtigung eines etwaigen Steigerungsbetrages für Höherversicherung und ohne Kinderzuschuß lautet sie:

$$JR = DvHS * Ba * VJ * StS$$

Dabei ist:

JR = Jahresrente

DvHS = Durchschnittlicher persönlicher Vomhundertsatz

Ba = Allgemeine Bemessungsgrundlage

VJ = Versicherungsjahre

StS = Steigerungssatz

Für jedes Kalenderjahr wird die allgemeine Bemessungsgrundlage neu festgelegt. Sie betrug für das Jahr 1983 25445 DM. Falls Sie einen aktuelleren Wert zur Hand haben, ändern Sie bitte die entsprechende Programmzeile.

Die Versicherungsjahre ergeben sich aus den anrechenbaren Beitragszeiten zuzüglich eventueller Ersatzzeiten (Militärdienst, politische Haft in der DDR usw.), Ausfallzeiten (Zeiten des Bezugs von Arbeitslosengeld, Zeiten der allgemeinen Schulbildung nach dem 16. Lebensjahr und der abgeschlossenen Hochschulausbildung usw.) oder Zurechnungszeiten (z. B. bei Berufs- oder Erwerbsunfähigkeit vor dem 55. Lebensjahr).

Der Steigerungssatz beträgt für Altersruhegelder und für Erwerbsunfähigkeitsrenten 1.5%, für Berufsunfähigkeitsrenten 1%.

Der durchschnittliche persönliche Vomhundertsatz wird zunächst für jedes Kalenderjahr und dann für die gesamte Beitragszeit berechnet. Er zeigt an, zu wieviel Prozent der persönliche Bruttoarbeitsverdienst mit dem Durchschnittsverdienst aller Versicherten im gleichen Zeitraum übereinstimmt. Die Obergrenze liegt bei 200%.

Im vorliegenden Programm sind zur Berechnung des durchschnittlichen persönlichen Vomhundertsatzes die durchschnittlichen Bruttojahresarbeitsentgelte aller Versicherten für die Jahre 1924 bis 1985 gespeichert. Dabei sind die Werte für die Jahre 1983 bis 1985 (Satz 440 des Programms) geschätzt. Sie sind gegebenenfalls zu aktualisieren.

Zu beachten ist, daß im Programmablauf die Berechnung des durchschnittlichen Vomhundertsatzes nur funktioniert, wenn Sie den persönlichen Verdienst exakt eingeben. Bis 1942 wurden jedoch ausschließlich Beiträge in Form von Beitragsmarken "geklebt". Diese Beiträge müßten Sie gesondert berücksichtigen, genauso wie nach 1942 "geklebte" Beitragsmarken, insbesondere für freiwillige Beiträge.

Das Programm in diesem Kapitel deckt also nicht alle Besonderheiten und Möglichkeiten der Rentenberechnung ab.

Aber wir wollen hier ja auch nicht mit den Landesversicherungsanstalten konkurrieren. Zudem kann das hier Gesagte durch gesetzliche Änderungen schon bald hinfällig werden. Wer weiß?

Also, insbesondere wenn es um

- Beitragsmarken
- Höherversicherung
- Kinderzuschuß oder
- Besonderheiten wie z. B. Zusatzbeträge für Wehrdienstpflichtige neben der Wehrdienstdauer oder besondere Regelungen für die ersten fünf Jahre der Versicherung zur Vermeidung einer Belastung des DvHS infolge geringerer Verdienste

geht, sollten Sie weitere Informationen einholen.

Programm:

```
10 REM I15
20 CLS
30 PRINT "I15 - Programm zur Rentenberechnung"
40 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
50 PRINT TAB(9) "Bernd Kowal, 1985"
60 GOSUB 650
70 DIM dbv(100),bjv(100)
80 PRINT"Der durchschnittliche persoenliche"
90 PRINT:PRINT"Vomhundertsatz"
100 PRINT:PRINT:PRINT:PRINT TAB(30)"Eingabe":PRINT
110 PRINT:PRINT"ist bekannt";TAB(33)"1"
120 PRINT:PRINT"soll berechnet werden";TAB(33)"2"
130 PRINT:PRINT:PRINT:INPUT"Ihre Wahl ";a:CLS
140 IF a<1 OR a>2 THEN 80
150 ON a GOTO 160,200
160 PRINT"Wie gross ist der durchschnittliche"
170 INPUT"persoenliche Vomhundertsatz ";DvHS
```

```

180 IF DvHS>200 THEN DvHS=200
190 GOTO 240
200 GOSUB 330
210 IF DvHS>200 THEN DvHS=200
220 PRINT"Der durchschnittliche persoenliche"
230 PRINT"Vomhundertersatz betraegt ";ROUND(DvHS,2);"%
240 PRINT:PRINT"Wieviele Versicherungsjahre"
250 INPUT"haben Sie ";VJ:PRINT
260 INPUT"Wie hoch ist Ihr Steigerungssatz ";STS
270 Ba=25445
280 JR=DvHS/100*Ba*VJ*STS/100
290 PRINT:PRINT:PRINT
300 PRINT"Ihre Jahresrente betraegt";ROUND(JR,2);"DM"
310 PRINT:PRINT"Das sind monatlich";ROUND(JR/12,2);"DM"
320 END
330 REM Unterprogramm zur Berechnung des
340 REM durchschnittlichen persoenlichen
350 REM Vomhundertersatzes
360 DATA 1233,1469,1642,1742,1983,2110,2074,1924
370 DATA 1651,1583,1605,1692,1783,1856,1947,2092
380 DATA 2156,2297,2310,2324,2292,1778,1778,1833
390 DATA 2219,2838,3161,3579,3852,4061,4234,4548
400 DATA 4844,5043,5330,5602,6101,6723,7328,7775
410 DATA 8467,9229,9893,10219,10842,11839,13343
420 DATA 14931,16335,18295,20381,21808,23335
430 DATA 24945,26242,27685,29485,30900,32198
440 DATA 33000,34000,35000
450 FOR i=24 TO 85
460 READ dbv(i):NEXT i
470 PRINT"In welchem Jahr haben Sie begonnen"
480 INPUT"Beitraege zu zahlen ";j1#
490 j1#=RIGHT$(j1#,2):j1=VAL(j1#)
500 IF j1<24 OR j1>85 THEN 470
510 PRINT:INPUT"Bis zu welchem Jahr ";j2#
520 j2#=RIGHT$(j2#,2):j2=VAL(j2#)
530 IF j2<24 OR j2>85 THEN 510
540 IF j1>j2 THEN 470
550 PRINT:PRINT"Geben Sie Ihr Bruttojahresverdienst"
560 PRINT"fuer die folgenden Jahre an ":PRINT
570 FOR i=j1 TO j2
580 PRINT"Jahr ";i;:INPUT bju(i)
590 we=we+bjv(i)/dbv(i)*100
600 NEXT i

```

```
610 PRINT:PRINT:PRINT"Geben Sie bitte den Umfang"  
620 INPUT"der Beitragszeit in Monaten an ";mo  
630 DvHS=we/mo*12  
640 CLS:RETURN  
650 REM Unterprogramm Warten  
660 LOCATE 5,25  
670 PRINT"Bitte eine Taste druecken !"  
680 x#=INKEY#  
690 IF x#="" THEN 680  
700 CLS:RETURN
```

Variablenliste:

a = Antwort "Menüauswahl"

Ba = Allgemeine Bemessungsgrundlage

bjv(i) = Bruttojahresverdienst

dbv(i) = durchschnittlicher Bruttojahresverdienst

DvHS = durchschnittlicher Vomhundertsatz

i = Laufindex

j1 = Jahresangabe (Beginn der Beitragszeit)

j1\$ = Eingabefeld für j1

j2 = Jahresangabe (Ende der Beitragszeit)

j2\$ = Eingabefeld für j2

JR = Jahresrente

mo = Beitragszeit in Monaten

STS = Steigerungssatz

VJ = Versicherungsjahre

**we = Werteinheit (Prozentzahl, die aus dem Vergleich
von bjv(i) mit dbv(i) resultiert)**

Programmbeschreibung:

Satz 10- 60 : Überschrift

Satz 70 : Speicherplatz wird reserviert

Satz 80-150 : Abfrage, ob der Wert für DvHS bekannt ist oder berechnet werden soll und entsprechende Programmverzweigungen

Satz 160-190 : Der Wert für DvHS ist bekannt und wird eingegeben. Ab Satz 240 geht es dann weiter.

Satz 200-230 : Der Wert für DvHS ist unbekannt und zu seiner Berechnung wird ins Unterprogramm (Satz 330-640) verzweigt. Das Ergebnis wird ausgegeben. Der Höchstwert für DvHS liegt bei 200% (siehe auch Satz 180).

Satz 240-260 : Es werden Werte für die Anzahl der Versicherungsjahre und für den Steigerungssatz angefordert

Satz 270 : Das Feld Ba wird belegt

Satz 280 : Die Jahresrente wird berechnet. Zu beachten ist, daß es sich bei DvHS und STS um Prozentzahlen handelt, die durch 100 dividiert werden müssen.

Satz 290-310 : Ergebnisausgabe mit auf zwei Dezimalstellen gerundeten Werten

Satz 320 : Programmende

Satz 330-640 : Unterprogramm zur Berechnung des DvHS

330-350 : Kommentar

360-460 : Bereitstellen und Einlesen der Daten für den durchschnittlichen Bruttojahresverdienst von 1924 bis 1985


470-540 : Eingabe der Beitragszeit (Anfang und Ende). Die Jahresangaben werden als Stringvariablen eingegeben, um in den Sätzen 490 und 520 eine möglicherweise mehrstellige Jahresangabe auf die letzten beiden Stellen zu reduzieren (z. B. 1985 auf 85). Die Sätze 500, 530 und 540 dienen dem Abfangen unsinniger Eingaben.

550-600 : Die individuellen Bruttojahresverdienste werden für den angegebenen Zeitraum angefordert. In Satz 590 wird die Werteinheit jeweils aktualisiert.

610-630 : Eingabe eines Wertes für den Umfang der Beitragszeit in Monaten und Berechnung des DvHS

640 : Löschen des Bildschirms und Ende des Unterprogramms

Satz 650-700 : Unterprogramm "Warten"



6.3. Der Lohnsteuerjahresausgleich

Alle Jahre wieder ... müssen (oder sollten) Sie einen Lohnsteuerjahresausgleich durchführen. Entweder Sie haben von Steuerangelegenheiten überhaupt keine Ahnung und beauftragen Jahr für Jahr einen Steuerberater, oder Sie beraten sich aufgrund Ihrer beruflichen Tätigkeit oder aus Interesse am Thema selbst. In beiden Fällen kann Ihnen das Programm in diesem Kapitel behilflich sein. Einerseits bietet es nämlich eine Einführung in die Materie "Lohnsteuerjahresausgleich", andererseits ist es für den Steuerexperten mit BASIC-Kenntnissen noch beliebig ausbaubar.

Sinn und Zweck des Programms ist es, den Erstattungsbetrag zu berechnen. Dazu ist im Programmablauf die Eingabe vieler persönlicher Daten erforderlich. Das Programm selbst ist zwar sehr umfangreich, aber eigentlich gar nicht so kompliziert, da es zu einem erheblichen Teil aus PRINT-Anweisungen besteht, die der Erläuterung der Eingaben dienen. Diese Eingaben werden dann in Abhängigkeit von den individuellen Daten häufig einfach aufsummiert, um die abzugsfähigen Größen, wie Werbungskosten, Sonderausgaben und Vorsorgeaufwendungen zu ermitteln. Dabei werden in der Regel unsinnige Eingaben nicht abgefangen, d. h., daß beispielsweise die Eingabe von 10000 DM für Arbeitskleidung akzeptiert wird.

Auch die mathematischen Teile des Programms sind durchaus verständlich. Haben Sie zum Beispiel das Hintergrundwissen bezüglich der Kilometergeldberechnung oder der Zumutbarkeitsregelung bei Sonderausgaben, so können Sie die entsprechenden Programmteile auch ganz gut nachvollziehen. Nur die Berechnungen zu den Vorsorgeaufwendungen und der Vorsorgepauschale könnten Schwierigkeiten bereiten. Jedoch sind an diesen Stellen die Variablen z29 bis z77 gebildet worden, welche die entsprechenden Zeilen in dem Schema zur Selbstberechnung (erhältlich bei Ihrem Finanzamt) symbolisieren. Die entsprechenden Programmzeilen wirken dadurch zwar nicht sehr elegant, dafür sind sie aber leichter nachvollziehbar.

Nach der Berechnung des zu versteuernden Einkommens wird ein Wert für die Steuer laut Einkommensteuertabelle angefordert. Sollten Sie keine Tabelle griffbereit haben, so wird die Steuer geschätzt. Diese Schätzung kann unter Umständen etwas ungenau sein, da die Steuer, von einigen gespeicherten Werten ausgehend, durch lineare Interpolation berechnet wird, obwohl der Anstieg der Steuerbeträge bekanntlich progressiv verläuft. Zur ungefähren Angabe des Erstattungsbetrages ist dieses Verfahren jedoch ausreichend.

Bisher ist nur vom Lohnsteuerjahresausgleich und nicht von der Einkommensteuererklärung die Rede gewesen. Zwar ergibt sich in beiden Fällen grundsätzlich die gleiche Jahressteuer, allerdings können die verschiedensten Tatbestände - neben dem Einkommen - eine Einkommensteuererklärung erforderlich machen. Eine Berücksichtigung dieser Besonderheiten hätte eine enorme Ausweitung des Programms zur Folge. Sie bleiben daher aufgefordert, in das Programm "einzusteigen" und Erweiterungen (bzw. Verbesserungen) selbst vorzunehmen. Die beim Finanzamt erhältliche Anleitung zum Antrag auf Lohnsteuerjahresausgleich und zur Einkommensteuererklärung kann da schon eine gewisse Hilfestellung

leisten. Dies gilt auch immer dann, wenn die im Programmablauf ausgegebenen Erläuterungen nicht ausreichend sein sollten.

Ansonsten sind Ihnen umfangreiche Steuerabsetzungen zu wünschen. Ob Sie aber den Kauf dieses Buches als Sonderausgabe von der Steuer absetzen können, wie es für Steuerfachliteratur möglich ist, das erfragen Sie bitte bei Ihrem Finanzamt oder bei Ihrem Steuerberater.

Programm:

```
10 REM I16
20 CLS
30 PRINT"I16 - Programm zur Berechnung des"
40 PRINT TAB(7)"Erstattungsbetrages beim"
50 PRINT TAB(7)"Lohnsteuerjahresausgleich"
60 PRINT:PRINT:PRINT
70 PRINT TAB(11)"Bernd Kowal, 1985"
80 GOSUB 6150
90 REM *****
100 REM Angaben zur Person
110 REM *****
120 PRINT"      Fragen zur Person":PRINT:PRINT
130 PRINT"Bei den folgenden Fragen koennen Sie"
140 PRINT"mit der Eingabe von 0 oder 1 antworten."
150 PRINT:PRINT"Dabei ist : 0 = nein und 1 = ja."
160 PRINT:PRINT
170 INPUT"Sind Sie verheiratet          ";ve:PRINT
180 PRINT"Waren Sie im vergangenen Jahr"
190 PRINT"rentenversicherungspflichtig"
200 INPUT"beschaeftigt          ";pk:PRINT
210 PRINT"Haben Sie im vergangenen Jahr"
220 PRINT"Versorgungsbezeuge (welche"
230 PRINT"als Bruttoarbeitslohn anzu-"
240 INPUT"geben sind) erhalten          ";vs:PRINT
250 PRINT"Sind Sie vor dem 1.2.1920"
260 INPUT"geboren          ";al:CLS
270 IF ve=0 THEN 430
280 PRINT" 0 = nein    und    1 = ja":PRINT:PRINT
290 PRINT"Ist Ihre Ehefrau/mann vor dem"
300 INPUT"1.2.1920 geboren          ";ale:PRINT
```

```

310 PRINT"Ist Ihre Ehefrau/mann"
320 INPUT"erwerbstaetig" ";er:PRINT
330 IF er=0 THEN 430
340 PRINT"War Ihre Ehefrau/mann"
350 PRINT"im vergangenen Jahr"
360 PRINT"rentenversicherungspflichtig"
370 INPUT"beschaeftigt" ";pke
380 PRINT:PRINT"Hat Ihre Ehefrau/mann im ver-"
390 PRINT"gangenen Jahr Versorgungs-"
400 PRINT"bezeuge (welche als Brutto-"
410 PRINT"arbeitslohn anzugeben sind)"
420 INPUT"erhalten" ";vse:PRINT
430 CLS:PRINT"Wie hoch war Ihr Bruttoarbeitslohn"
440 INPUT"im vergangenen Jahr (in DM) ";bl(1)
450 IF vs=1 THEN vfb(1)=bl(1)/10*4
460 IF vfb(1)>4800 THEN vfb(1)=4800
470 bl1(1)=bl(1)-vfb(1)-600
480 PRINT:PRINT"Wie hoch war Ihre Lohnsteuer im"
490 INPUT"vergangenen Jahr (in DM) ";lst(1):PRINT
500 INPUT"und Ihre Kirchensteuer (in DM) ";kst(1)
510 PRINT
520 INPUT"Wieviele Kinder haben Sie insgesamt ";ki
530 PRINT:PRINT"Wieviele Kinder werden von Ihnen"
540 PRINT"nur unterhalten"
550 PRINT"(d. h. sie sind dem anderen Elternteil"
560 INPUT"zugeordnet worden) ";kiz
570 ki=ki-kiz
580 PRINT: IF er=0 THEN 690
590 PRINT"Wie hoch war der Bruttoarbeitslohn"
600 PRINT"Ihrer Ehefrau/mann im ver-"
610 INPUT"gangenen Jahr (in DM) ";bl(2)
620 IF vse=1 THEN vfb(2)=bl(2)/10*4
630 IF vfb(2)>4800 THEN vfb(2)=4800
640 bl1(2)=bl(2)-vfb(2)-600
650 PRINT:PRINT"Wie hoch war die Lohnsteuer"
660 PRINT"Ihrer Ehefrau/mann im ver-"
670 INPUT"gangenen Jahr (in DM) ";lst(2)
680 PRINT:INPUT"und die Kirchensteuer (in DM) ";kst(2)
690 CLS
700 REM *****
710 REM Werbungskosten
720 REM *****
730 i=1
740 PRINT TAB(9)"Werbungskosten"
750 IF i=2 THEN PRINT TAB(9)"fuer Ihre Ehefrau/mann"
760 PRINT:PRINT TAB(33)"Eingabe"
770 PRINT:PRINT"Ihre Werbungskosten sind"
780 PRINT" Ihnen bekannt" 1"

```

```

790 PRINT"Sie wollen den Fauschbetrag"
800 PRINT" von 564 DM einsetzen          2"
810 PRINT"Sie wollen Ihre Werbungskosten"
820 PRINT" berechnen                      3"
830 PRINT:INPUT"Ihre Wahl ";a1
840 CLS
850 IF a1<>1 THEN 890
860 PRINT:PRINT"Wie hoch sind Ihre"
870 INPUT" Werbungskosten (in DM) ";wk(i)
880 GOTO 2140
890 IF a1=2 THEN wk(1)=564:GOTO 2180
900 PRINT" Ihre Werbungskostenaufstellung":PRINT
910 PRINT"Aufwendungen fuer Fahrten zwischen"
920 PRINT"Wohnung und Arbeitsstaette"
930 PRINT:PRINT TAB(33)"Eingabe":PRINT
940 PRINT"mit dem eigenen Pkw          1"
950 PRINT"mit oeffentlichen Verkehrsmitteln  2"
960 PRINT:INPUT"Ihre Wahl ";a2
970 PRINT:PRINT"Bei wievielen Arbeitgebern waren"
980 INPUT"Sie im letzten Jahr beschaeftigt ";a3
990 PRINT:FOR j=1 TO a3
1000 IF j=3 THEN CLS
1010 IF a3>1 THEN PRINT j;".ter Arbeitgeber" 4
1020 IF a2=2 THEN INPUT"DM fuer Fahrkarten ";fk
1030 IF a2=2 THEN 1080
1040 INPUT"Arbeitstage je Woche ";ajw
1050 INPUT"Urlaubs- und Krankheitstage ";ukd
1060 INPUT"einfache Entfernung (in km) ";ekm
1070 fk=(ajw*52-ukd)*0.36*ekm
1080 PRINT: NEXT j:CLS
1090 PRINT:PRINT"Wieviel DM hat der Arbeitgeber"
1100 PRINT"im letzten Jahr steuerfrei fuer"
1110 INPUT"Fahrkosten ersetzt ";erf
1120 fk=fk-erf
1130 CLS:PRINT"Wieviel DM haben Sie im letzten"
1140 PRINT"Jahr an Beitraegen zu Berufs-"
1150 PRINT"verbaenden (z. B. Gwerkschaften)"
1160 INPUT"gezahlt (in DM) ";bbv
1170 PRINT:PRINT"und an Aufwendungen fuer"
1180 PRINT"Arbeitsmittel (z. B. Fachliteratur,"
1190 PRINT"Arbeitskleidung ... , bis 180 DM"
1200 INPUT"ohne Nachweis) ";awa
1210 PRINT:PRINT"und an Aufwendungen fuer"
1220 PRINT"berufliche Fortbildungmassnahmen"
1230 INPUT"(in voller Hoehe mit Nachweis) ";fbk
1240 PRINT:PRINT"und aus beruflichen Gruenden Um-"
1250 PRINT"zugskosten (aus Vereinfachungsgruenden)"

```

```

1260 PRINT"sollen 700 DM die Obergrenze fuer Ver-"
1270 PRINT"heiratete und 400 DM fuer Ledige plus"
1280 INPUT"125 DM je Kind sein) ";uzk
1290 CLS
1300 PRINT"Mehraufwendungen fuer Verpflegung"
1310 PRINT"(mit Nachweis des Arbeitgebers)"
1320 LOCATE 33,4:PRINT"Eingabe":PRINT
1330 PRINT"bei ueber 12-stuendiger Abwesen-"
1340 PRINT"heit von der Wohnung 1"
1350 PRINT
1360 PRINT"bei staendig wechselnden Arbeits-"
1370 PRINT"orten und ueber 10-stuendiger Ab-"
1380 PRINT"wesenheit von der Wohnung 2"
1390 PRINT:PRINT"bei Berufskraftfahrern"
1400 PRINT:PRINT" Abwesenheit von"
1410 PRINT" mehr als 6 Stunden 3"
1420 PRINT:PRINT" Abwesenheit von"
1430 PRINT" mehr als 12 Stunden 4"
1440 PRINT
1450 PRINT"es ist nichts geltend zu machen 5"
1460 PRINT:INPUT"Ihre Wahl ";a4
1470 IF a4=5 THEN 1530
1480 PRINT:INPUT"an wievielen Tagen ";mvt
1490 IF a4=1 THEN mfv=mvt*3:GOTO 1530
1500 IF a4=2 THEN mfv=mvt*5:GOTO 1530
1510 IF a4=3 THEN mfv=mvt*8:GOTO 1530
1520 IF a4=4 THEN mfv=mvt*16
1530 CLS
1540 PRINT"Mehraufwendungen fuer doppelte"
1550 PRINT"Haushaltsfuehrung"
1560 PRINT"(aus beruflichen Gruenden)"
1570 LOCATE 33,5:PRINT"Eingabe":PRINT
1580 PRINT"mit eigenem Hausstand 1"
1590 PRINT
1600 PRINT"ohne eigenen Hausstand 2"
1610 PRINT
1620 PRINT"kann nicht geltend gemacht werden 3"
1630 PRINT:PRINT:INPUT"Ihre Wahl ";a5
1640 IF a5=3 THEN 2060
1650 CLS
1660 PRINT"Kosten der ersten Fahrt zum"
1670 PRINT"Beschaeftigungsort und der letzten"
1680 PRINT"Fahrt zum Wohnort nach Beendigung"
1690 PRINT"der Taetigkeit"
1700 PRINT:PRINT TAB(33)"Eingabe":PRINT
1710 PRINT"mit dem eigenen Pkw 1"
1720 PRINT

```

```

1730 PRINT"mit oeffentlichen Verkehrsmitteln 2"
1740 PRINT:INPUT"Ihre Wahl ";a6:PRINT
1750 IF a6=1 THEN 1780
1760 INPUT"Preis der Fahrkarte (in DM) ";mdh
1770 GOTO 1800
1780 INPUT"gefahrne Kilometer ";gkm
1790 mdh=gkm*0.42
1800 IF a5=2 THEN 1890
1810 PRINT:PRINT"Ihre Wahl fuer Ihre woechent-
1820 INPUT"lichen Heimfahrten ";a7:PRINT
1830 INPUT"Anzahl der Heimfahrten ";anz
1840 PRINT: IF a7=1 THEN 1870
1850 INPUT"Preis einer Fahrkarte (in DM) ";pfk
1860 mdh=mdh+pfk*anz:GOTO 1890
1870 INPUT"Entfernung (in Km) ";ekm
1880 mdh=mdh+ekm*0.36*anz
1890 CLS:PRINT"Kosten der Unterkunft (mit Nachweis)
1900 IF a6=2 THEN PRINT"fuer die ersten zwei Wochen"
1910 INPUT"in DM ";kdu
1920 mdh=mdh+kdu
1930 PRINT:PRINT"Verpflegungskosten fuer die"
1940 PRINT"ersten zwei Wochen bzw. 10 Tage bei"
1950 PRINT"geltend gemachter Heimfahrt (mit Nach-
1960 INPUT"weis max. 35 DM pro Tag) ";vpk
1970 mdh=mdh+vpk
1980 IF a6=2 THEN 2030
1990 PRINT:PRINT"fuer die Folgezeit (max. 14 DM"
2000 PRINT"pro Tag der auswaertigen"
2010 INPUT"Unterbringung ";vpf
2020 mdh=mdh+vpf
2030 PRINT:PRINT"Wieviel (in DM) hat der Arbeitgeber"
2040 INPUT"an Ersatzleistungen gezahlt ";abe
2050 mdh=mdh+abe
2060 CLS
2070 PRINT"Wenn Sie weitere Werbungskosten geltend"
2080 PRINT"machen koennen, geben Sie diese"
2090 INPUT"bitte (in DM) an ";wvk
2100 wk(i)=fk+bbv+awa+fbk+sbk+usk+mfv+mdh+wvk
2110 PRINT:PRINT
2120 PRINT"Es sind Werbungskosten in Hoehe von"
2130 PRINT"DM";wk(i);"errechnet worden !"
2140 IF wk(i)>=564 THEN 2180
2150 PRINT:PRINT"Es wird somit der Pauschbetrag"
2160 PRINT"von 564 DM angenommen"
2170 wk(i)=564
2180 b12(i)=b11(i)-480-wk(i)
2190 IF a1=2 THEN CLS ELSE GOSUB 6150

```

```

2200 IF i=2 OR er=0 THEN b1=b12(1)+b12(2):GOTO 2250
2210 ajw=0:ukd=0:ekm=0:fk=0:erf=0:bbv=0:awa=0:fbk=0
2220 sbk=0:uzk=0:mfv=0:mvt=0:mdh=0:gkm=0:anz=0
2230 pfk=0:kdu=0:vpk=0:vpf=0:abe=0:wwk=0
2240 i=2:CLS:GOTO 740
2250 REM *****
2260 REM Sonderausgaben
2270 REM *****
2280 PRINT"Bestimmung der Sonderausgaben,"
2290 PRINT"die nicht Vorsorgeaufwendungen sind !"
2300 PRINT:IF er=0 THEN 2350
2310 PRINT"Es wird vom Gesamtbetrag der Sonderaus-"
2320 PRINT"gaben ausgegangen, d. h. es wird nicht"
2330 PRINT"zwischen Ihnen und Ihrer Ehefrau/mann"
2340 PRINT"unterschieden."
2350 PRINT:PRINT TAB(33)"Eingabe"
2360 PRINT:PRINT"Ihre Sonderausgaben sind"
2370 PRINT" Ihnen bekannt 1"
2380 PRINT"Sie wollen den Pauschbetrag"
2390 PRINT" von 270 DM fuer Ledige bzw."
2400 PRINT" von 540 DM fuer Ehegatten"
2410 PRINT" einsetzen 2"
2420 PRINT"Sie wollen Ihre Sonderausgaben"
2430 PRINT" berechnen 3"
2440 PRINT:INPUT"Ihre Wahl ";a8
2450 CLS
2460 IF ve=0 THEN pb=270 ELSE pb=540
2470 IF a8<>1 THEN 2510
2480 PRINT:PRINT"Wie hoch sind Ihre"
2490 INPUT" Sonderausgaben (in DM) ";sa
2500 GOTO 3300
2510 IF a8=2 THEN sa=pb:GOTO 3340
2520 PRINT" Ihre Sonderausgabenaufstellung":PRINT
2530 PRINT"Geben Sie an (in DM), welche Renten-"
2540 PRINT"zahlungen oder dauernde Lasten Sie"
2550 PRINT"im letzten Jahr geleistet haben."
2560 PRINT:PRINT"Beachten Sie, dass diese Ausgaben"
2570 PRINT"auf einer besonderen Verpflichtung"
2580 PRINT"(sozusagen unfreiwillig)"
2590 PRINT"beruhen muessen.":PRINT
2600 PRINT"Weiterhin ist bei der Zahlung von "
2610 PRINT"Leibrenten zu beruecksichtigen, dass"
2620 PRINT"als Sonderausgabe nur der jeweilige"
2630 PRINT"Ertragsanteil des Rentenrechts abzieh-"
2640 PRINT"abziehbar ist (ggf. entsprechende"
2650 PRINT"Informationen einholen !)."
```

```

2670 CLS
2680 PRINT"Geben Sie (in DM) Ihre Unterhalts-"
2690 PRINT"leistungen an den geschiedenen oder"
2700 PRINT"dauernd getrennt lebenden Ehegatten an."
2710 PRINT:PRINT"Sind diese Leistungen beim"
2720 PRINT"Empfaenger steuerpflichtig (s. Nachweis"
2730 PRINT"Anlage U), so ist ein Hoechstbetrag"
2740 PRINT"von 9000 DM einsetzbar."
2750 PRINT"Ansonsten koennen die Unterhalts-"
2760 PRINT"leistungen nur als aussergewoehnliche"
2770 PRINT"Belastung (Hoechstbetrag 3600 DM)"
2780 PRINT"geltend gemacht werden."
2790 PRINT:INPUT"DM ";ule
2800 CLS
2810 IF kst(1)=0 AND kst(2)=0 THEN 2860
2820 PRINT"Wie hoch war Ihre Kirchensteuerer-"
2830 PRINT"stattung im letzten Jahr."
2840 INPUT"DM ";ker
2850 kae=kst(1)+kst(2)-ker:PRINT:PRINT
2860 PRINT"Geben Sie Ihre Steuerberatungskosten"
2870 PRINT"an, sofern Sie diese nicht als"
2880 PRINT"Werbungskosten geltend gemacht haben."
2890 PRINT:INPUT"DM ";sks:CLS
2900 PRINT"Hatten Sie Aufwendungen fuer Ihre"
2910 PRINT"Berufsausbildung oder fuer die Berufs-"
2920 PRINT"ausbildung Ihres Ehegatten ?"
2930 PRINT"Bis zu einem Hoechstbetrag von je"
2940 PRINT"900 DM (1200 DM bei auswaertiger"
2950 PRINT"Unterbringung) sind diese Aufwendungen"
2960 PRINT"abziehbar."
2970 PRINT"Dazu gehoeren u. a. Studiengebuehren,"
2980 PRINT"Kosten fuer Lernmaterial, aber auch"
2990 PRINT"Aufwendungen fuer Fahrten zwischen"
3000 PRINT"Wohnung und Ausbildungsstaette (ohne"
3010 PRINT"Nachweis werden bei Fahrten mit dem"
3020 PRINT"eigenen Fkw 0.42 DM je Kilometer aner-"
3030 PRINT"kannt), sowie Mehraufwendungen fuer"
3040 PRINT"Unterkunft und Verpflegung bei aus-"
3050 PRINT"waertiger Unterbringung).
3060 PRINT:PRINT"Wie hoch waren Ihre Aufwendungen"
3070 INPUT"(in DM) ";aeb:CLS
3080 PRINT"Haben Sie Spenden und Beitraege fuer"
3090 PRINT"  a) wissenschaftliche und kulturelle"
3100 PRINT"      Zwecke"
3110 PRINT"  b) mildtaetige, kirchliche"
3120 PRINT"      religioese und gemeinnuetzige"
3130 PRINT"      Zwecke"

```

```

3140 PRINT"geleistet (mit Bestaetigung des"
3150 PRINT"Empfaengers) ?"
3160 PRINT:PRINT"Beachten Sie, dass fuer Punkt a) und"
3170 PRINT"b) jeweils hoechstens 5 v.H. Ihrer"
3180 PRINT"Einkuenfte beruecksichtigt werden"
3190 PRINT"koennen. Das sind in Ihrem Fall"
3200 PRINT (b1(1)+b1(2))/100*5;"DM."
3210 PRINT:INPUT"Geben Sie einen DM Betrag an ";spb
3220 CLS
3230 PRINT"Wenn Sie weitere Sonderausgaben geltend"
3240 PRINT"machen koennen, geben Sie diese"
3250 INPUT"bitte (in DM) an ";wsa
3260 sa=rdl+tule+kae+skst+ueb+spb+wsa
3270 PRINT:PRINT
3280 PRINT"Es sind Sonderausgaben in Hoehe von"
3290 PRINT"DM";sa;"errechnet worden !"
3300 IF sa>=pb THEN 3340
3310 PRINT:PRINT"Es wird somit der Pauschbetrag"
3320 PRINT"von";pb;"DM angenommen."
3330 sa=pb
3340 b11=b1-sa
3350 IF a8=2 THEN CLS ELSE GOSUB 6150
3360 REM *****
3370 REM Vorsorgeaufwendungen
3380 REM *****
3390 PRINT"      Vorsorgeaufwendungen":PRINT
3400 PRINT:IF er=0 THEN 3450
3410 PRINT"Es wird vom Gesamtbetrag der Vorsorge"
3420 PRINT"aufwendungen ausgegangen, d. h. es wird"
3430 PRINT"nicht zwischen Ihnen und Ihrer"
3440 PRINT"Ehefrau/mann unterschieden.":PRINT:PRINT
3450 PRINT"Geben Sie den Arbeitnehmeranteil der"
3460 PRINT"Beitraege zur gesetzlichen Sozialver-"
3470 PRINT"sicherung und/oder Beitrage zu einer"
3480 PRINT"befreienden Lebensversicherung (bzw."
3490 PRINT"gleichgestellte Aufwendungen, jeweils"
3500 PRINT"ohne steuerfreie Zuschuesse des"
3510 PRINT"Arbeitgebers) an."
3520 PRINT:INPUT"DM ";ags:PRINT
3530 PRINT"Geben Sie den Arbeitgeberanteil zur ge-"
3540 PRINT"setzlichen Rentenversicherung und/oder"
3550 PRINT"steuerfreie Zuschuesse des Arbeitgebers"
3560 PRINT"zu gleichgestellten Aufwendungen an"
3570 PRINT:INPUT"DM ";agr:CLS
3580 PRINT"Geben Sie zu den folgenden Punkten die"
3590 PRINT"entsprechenden DM-Betraege (ggf. ab-"
3600 PRINT"zueglich der steuerfreien Zuschuesse,"

```

```

3610 PRINT"z. B. des Arbeitgebers) an, soweit sie"
3620 PRINT"keine vermögenswirksamen Leistungen"
3630 PRINT"sind :":PRINT
3640 PRINT"Freiwillige Angestellten-,"
3650 PRINT"Renten- und Hoerher-"
3660 INPUT"versicherung ..... " ;arh
3670 PRINT"Freiwillige Kranken-"
3680 INPUT"versicherung ..... " ;fkv
3690 PRINT"Freiwillige Unfall-"
3700 INPUT"versicherung ..... " ;fuv
3710 PRINT"Lebensversicherung und"
3720 INPUT"Sterbekasse ..... " ;lvs
3730 PRINT"Haftpflichtversicherung (mit"
3740 PRINT"Kfz-, ohne Kasko-, Hausrat-"
3750 INPUT"Rechtsschutzversicherung) ... " ;hvp
3760 PRINT"Beitraege an Bausparkassen"
3770 PRINT"(nur wenn keine Wohnungsbau-"
3780 INPUT"praemie beantragt wurde) .... " ;bbs:CLS
3790 vb=agstarh+fkv+fuv+lvs+hvp
3800 IF ve=1 THEN z29=6000:z36=4680+ki*600+kiz*300
3810 IF ve=0 THEN z29=3000:z36=2340+ki*600+kiz*300
3820 IF pk=0 THEN agr=b1(1)/100*9
3830 IF er=1 THEN 3850
3840 IF pke=0 THEN agr=agr+b1(2)/100*9
3850 z31=z29-agr
3860 z32=vb-z31:IF z32<0 THEN z32=0
3870 IF vb<z31 THEN z33=vb ELSE z33=z31
3880 z35=z32+bbs
3890 IF z35<z36 THEN z37=z35 ELSE z37=z36
3900 z38=z35-z36:IF z38<0 THEN z38=0
3910 IF z38/2<z36/2 THEN z41=z38/2 ELSE z41=z36/2
3920 z42=z33+z37+z41
3930 PRINT"Berechnete abziehbare Vorsorge-"
3940 PRINT"aufwendungen :";z42;"DM":PRINT:PRINT
3950 REM *****
3960 REM Berechnung der Vorsorgepauschale
3970 REM *****
3980 z46=b11(1)+b11(2)
3990 IF pk<>pke AND er=1 THEN 4170
4000 REM Berechnung der ungekuerzten Vorsorge-
4010 REM pauschale (Personenkreis A)
4020 REM oder der gekuerzten Vorsorgepauschale
4030 REM (Personenkreis B)
4040 z47=z46/100*9
4050 IF pk=1 THEN z48=2340*(ve+1):z51=1170*(ve+1)
4060 IF pk=0 THEN z48=1000*(ve+1):z51=1000*(ve+1)
4070 z48=z48+ki*600+kiz*300:z51=z51+ki*300+kiz*150

```

```

4080 IF z47<z48 THEN z49=z47 ELSE z49=z48
4090 IF z47<z51 THEN z52=z47 ELSE z52=z51
4100 vp=z49+z52
4110 GOTO 4340
4120 REM Berechnung der Vorsorgepauschale fuer
4130 REM Ehegatten, die beide Arbeitslohn
4140 REM beziehen und von denen einer zum
4150 REM Personenkreis A und der andere
4160 REM zum Personenkreis B gehoert
4170 IF pk=1 THEN i=1 ELSE i=2
4180 z55=b11(i)/100*18
4190 IF i=1 THEN i=2 ELSE i=1
4200 z56=b11(i)/100*9;z57=1000+ki*600+kiz*300
4210 IF z56<z57 THEN z58=z56 ELSE z58=z57
4220 z60=1000+ki*300+kiz*150
4230 IF z56<z60 THEN z61=z56 ELSE z61=z60
4240 z62=z55+z58+z61;z64=4680+ki*600+kiz*300
4250 IF z62<z64 THEN z65=z62 ELSE z65=z64
4260 z67=(z62-z64)/2;IF z67<0 THEN z67=0
4270 IF z67<z64/2 THEN z69=z67 ELSE z69=z64/2
4280 z70=z65+z69;z72=2000+ki*600+kiz*300
4290 IF z56<z72 THEN z73=z56 ELSE z73=z72
4300 z75=2000+ki*300+kiz*150
4310 IF z56<z75 THEN z76=z56 ELSE z76=z75
4320 z77=z73+z76
4330 IF z70>z77 THEN vp=z70 ELSE vp=z77
4340 REM Der berechnete Wert fuer die Vorsorge-
4350 REM pauschale wird auf den naechsten durch
4360 REM 54 ohne Rest teilbaren vollen DM-Betrag
4370 REM abgerundet
4380 IF INT(vp)/54<>INT(INT(vp)/54) THEN 4400
4390 vp=INT(vp);GOTO 4410
4400 vp=vp-1;GOTO 4380
4410 PRINT"Berechnete Vorsorgepauschale :";vp;"DM"
4420 REM Mindestbetrag
4430 IF vp<300 AND ve=0 THEN vp=300;GOTO 4460
4440 IF vp<600 AND ve=1 THEN vp=600;GOTO 4460
4450 GOTO 4480
4460 PRINT:PRINT"Es wird der Mindestbetrag von";vp;"DM"
4470 PRINT"angesetzt."
4480 IF z42>vp THEN abv=z42 ELSE abv=vp
4490 PRINT:PRINT:PRINT"Abziehbarer Betrag fuer"
4500 PRINT"Vorsorgeaufwendungen :";abv;"DM"
4510 GOSUB 6150
4520 REM *****
4530 REM aussergewoehnliche Belastungen
4540 REM *****

```

```

4550 PRINT" Aussergewoehnliche Belastungen":PRINT
4560 IF er=0 THEN 4610
4570 PRINT"Es wird vom Gesamtbetrag der ausserge-"
4580 PRINT"woehnlichen Belastungen ausgegangen,"
4590 PRINT"d. h. es wird nicht zwischen Ihnen und"
4600 PRINT"Ihrer Ehefrau unterschieden.":PRINT
4610 PRINT"Naehere Informationen zu den folgenden"
4620 PRINT"Punkten koennen Sie z. B. der Anleitung"
4630 PRINT"zum Antrag auf Lohnsteuerjahres-"
4640 PRINT"ausgleich entnehmen.":PRINT:PRINT:PRINT
4650 PRINT"Koerperbehinderte und Hinterbliebene"
4660 PRINT"koennen Pauschbeträge beanspruchen"
4670 PRINT"(je nach Grad der Erwerbsminderung)"
4680 PRINT"oder Einzelnachweise erbringen."
4690 PRINT:INPUT"Geben Sie einen DM-Betrag an ";kub
4700 CLS
4710 PRINT:PRINT"Aufwendungen bis zu 1200 DM im Jahr"
4720 PRINT"koennen abgezogen werden, wenn Sie eine"
4730 PRINT"Haushaltshilfe beschaeftigt hatten"
4740 PRINT"(dann muessen Sie oder Ihr Ehegatte"
4750 PRINT"mindestens 60 Jahre alt gewesen sein"
4760 PRINT"oder ein Familienmitglied war krank"
4770 PRINT"oder schwer koerperbehindert),"
4780 PRINT"oder wenn Sie oder Ihr Ehegatte in"
4790 PRINT"einem Pflegeheim untergebracht waren."
4800 PRINT:INPUT"Geben Sie einen DM-Betrag an ";bhh
4810 PRINT:PRINT
4820 PRINT"Koennen Sie einen Freibetrag fuer"
4830 PRINT"besondere Faelle gelten machen (als"
4840 PRINT"Fluechtling, Vertriebener, Spaet-"
4850 PRINT"aussiedler usw.) ?"
4860 PRINT:INPUT"Geben Sie einen DM-Betrag an ";fbf
4870 CLS:PRINT"Haben Sie Unterhalt fuer ein Kind"
4880 PRINT"geleistet, das Ihnen steuerlich nicht"
4890 PRINT"zugeordnet wird ? (wenn ja, dann"
4900 PRINT"koennen Sie 600 DM angeben)"
4910 PRINT:INPUT"DM-Betrag ";fuk:PRINT:PRINT
4920 PRINT"Wenn Sie alleinstehend und erwerbs-"
4930 PRINT"taetig sind, so koennen Sie, falls Sie"
4940 PRINT"Kinder haben, einen Freibetrag von"
4950 PRINT"1200 DM geltend machen."
4960 PRINT:INPUT"DM-Betrag ";kba:CLS
4970 PRINT"Wenn Sie beduerftige Angehoerige unter-"
4980 PRINT"stuetzt haben, so koennen Sie Ihre"
4990 PRINT"nachgewiesenen Aufwendungen bis zu"
5000 PRINT"3600 DM fuer jede unterhaltene Person"
5010 PRINT"geltend machen. Auf diesen Hoechst-"

```

```

5020 PRINT"betrag werden jedoch die eigenen Ein-"
5030 PRINT"kuenfte der unterhaltenen Person,"
5040 PRINT"soweit sie 4200 DM uebersteigen,"
5050 PRINT"angerechnet."
5060 PRINT"Fuer Angehoerige in Ostblockstaaten"
5070 PRINT"koennen Sie fuer jedes Paket 40 DM und"
5080 PRINT"fuer jedes Paeckchen 30 DM absetzen"
5090 PRINT"(weitere Aufwendungen mit Nachweis,"
5100 PRINT"z. B. wenn Sie Ihre Angehoerigen be-"
5110 PRINT"sucht haben einen Pauschbetrag von"
5120 PRINT"50 DM oder wenn Sie Besuch von An-"
5130 PRINT"gehoeerigen aus dem Ostblock hatten,"
5140 PRINT"10 DM je Besucher und Besuchstag)."

```

```

5490 IF ki=1 OR ki=2 THEN pge=2
5500 IF ki>2 THEN pge=1
5510 IF b1>30000 THEN pge=pge+1:IF ki>2 THEN pge=1
5520 aab=aab-b1/100*pge:IF aab<0 THEN aab=0
5530 agb=kub+bhh+fbf+fuk+ubp+afb+kba+aab
5540 PRINT"Der Gesamtbetrag Ihrer ausserge-"
5550 PRINT"woehnlichen Belastungen betraegt"
5560 PRINT agb;"DM.":GOSUB 6150
5570 REM *****
5580 REM Berechnung des zu versteuernden Einkommens
5590 REM *****
5600 vek=b11-abv-agb-al*720-ale*720-ki*432-kiz*216
5610 IF ve=0 AND ki>0 THEN vek=vek-4212
5620 PRINT"Ihr zu versteuerndes Einkommen"
5630 PRINT"belaeuft sich auf :";ROUND(vek);"DM":PRINT
5640 REM *****
5650 REM Berechnung des Erstattungsbetrages
5660 REM *****
5670 PRINT:PRINT"Geben Sie den Betrag fuer Ihre Lohn-"
5680 PRINT"steuer laut der fuer die Steuer-"
5690 PRINT"berechnung massgebenden Einkommen-"
5700 PRINT"steuerlabelle an. Wenn Sie nur die"
5710 PRINT"ENTER-Taste (ohne Angabe eines"
5720 PRINT"Betrages) druecken, wird ein"
5730 PRINT"Naeherungswert (welcher den wahren"
5740 PRINT"Wert um bis zu 30 DM ueberzeichnen"
5750 PRINT"kann) fuer Ihre Lohnsteuer"
5760 INPUT"angenommen ";stt$
5770 IF stt$<>" " THEN stt=VAL(stt$):GOTO 5940
5780 IF ve=1 THEN vek=vek/2
5790 IF vek<=4265 THEN stt=0:GOTO 5930
5800 IF vek<=17711 THEN stt=(vek-4212)/100*22:GOTO 5930
5810 IF vek<=20465 THEN u=17765:o=20465:a=2970:b=3604:
GOTO 5880
5820 IF vek<=23219 THEN u=20519:o=23219:a=3617:b=4346:
GOTO 5880
5830 IF vek<=25973 THEN u=23273:o=25973:a=4361:b=5176:
GOTO 5880
5840 PRINT"Fuer ein Einkommen dieser Hoehe steht"
5850 PRINT"kein Naeherungsverfahren zur Verfuegung."
5860 PRINT"Versuchen Sie es also mit einer Eingabe."
5870 INPUT"Steuer laut Tabelle ";stt:GOTO 5940
5880 d=b-a:d=d/49:a=a-d
5890 FOR i=u TO o STEP 54
5900 a=a+d
5910 IF vek<=i THEN stt=a:GOTO 5930
5920 NEXT i
5930 IF ve=1 THEN stt=stt*2
5940 jks=(stt-ki*600-kiz*300)/100*9
5950 IF jks<0 THEN jks=0

```

```

5960 zek=kst(1)+kst(2)-jks
5970 CLS:IF zek<0 THEN zek=0
5980 PRINT"          Ein-    Jahres-   Ausgleichs-"
5990 PRINT"          behalten steuer   betrag"
6000 PRINT"Lohn-":PRINT"steuer";
6010 PRINT USING "#####";ROUND(1st(1)+1st(2));
6020 PRINT USING "#####";ROUND(stt);
6030 PRINT USING "#####";ROUND(1st(1)+1st(2)-stt)
6040 PRINT"Kirchen-":PRINT"steuer";
6050 PRINT USING "#####";ROUND(kst(1)+kst(2));
6060 PRINT USING "#####";ROUND(jks);
6070 PRINT USING "#####";ROUND(zek)
6080 PRINT:PRINT:PRINT
6090 PRINT"Erstattungsbetrag :";ROUND(1st(1)+1st(2)
      -stt+zek);"DM"
6100 PRINT:PRINT:PRINT
6110 PRINT"Es handelt sich um gerundete Werte."
6120 PRINT:PRINT:PRINT:PRINT TAB(12)"Programmende"
6130 PRINT:PRINT:PRINT
6140 END
6150 REM UP Warten
6160 LOCATE 7,25
6170 PRINT"Bitte eine Taste druecken"
6180 x#=INKEY#:IF x#="" THEN 6180
6190 CLS:RETURN

```

Variablenliste:

Die Variablen werden hier nicht alphabetisch geordnet, sondern nach ihrem Auftreten in den einzelnen Programmenteilen aufgelistet, um eine bessere Übersicht zu erhalten.

Angaben zur Person:

ve = verheiratet (0=nein, 1=ja)

pk = Personenkreis (1=Personenkreis A, 0=Personenkreis B, siehe Schema zur Selbstberechnung)

vs = Versorgungsbezüge (0=nein, 1=ja)

al = Alter (0=nein, 1=ja, siehe Stichtag zum Altersfreibetrag)

ale = Alter des Ehegatten (0,1)

er = Erwerbstätigkeit des Ehegatten (0=nein, 1=ja)

pke = Personenkreis des Ehegatten (0,1)

vse = Versorgungsbezüge des Ehegatten (0,1)

bl(1) = Bruttoarbeitslohn des Antragstellers

vfb(1) = Versorgungsfreibetrag des Antragstellers

bl1(1) = $bl(1) - vfb(1) - 600$, 600=Weihnachtsfreibetrag

lst(1) = gezahlte Lohnsteuer des Antragstellers

kst(1) = gezahlte Kirchensteuer des Antragstellers

ki = Anzahl der Kinder

kiz = Anzahl der sog. Zahlkinder

bl(2) = Bruttoarbeitslohn des Ehegatten

vfb(2) = Versorgungsfreibetrag des Ehegatten

bl1(2) = bl(2)-vfb(2)-600, 600=Weihnachtsfreibetrag

lst(2) = gezahlte Lohnsteuer des Ehegatten

kst(2) = gezahlte Kirchensteuer des Ehegatten

Werbungskosten:

i = Index (1=Antragsteller, 2=Ehegatte)

j = Laufvariable

a1-a7 = Menüantworten

wk(1) = Werbungskosten des Antragstellers

wk(2) = Werbungskosten des Ehegatten

bl2(1) = bl1(1)-480-wk(1), 480=Arbeitnehmerfreibetrag

bl2(2) = bl1(2)-480-wk(2), 480=Arbeitnehmerfreibetrag

bl = Gesamtbetrag der Einkünfte = bl2(1)+bl2(2)

ajw = Arbeitstage je Woche

ukd = Urlaubs- und Krankheitstage

ekm = einfache Entfernung in Kilometer

fk = Fahrkosten

erf = steuerfreie Fahrkostenerstattung durch den Arbeitgeber

bbv = Beiträge für Berufsverbände

awa = Aufwendungen für Arbeitsmittel

fbk = Fortbildungskosten

uzk = Umzugskosten

mfv = Mehraufwendungen für Verpflegung

mvt = Anzahl der Tage bezüglich mfv

dh = Mehraufwendungen für doppelte Haushaltsführung

gkm = gefahrene Kilometer

anz = Anzahl der Heimfahrten

pfk = Preis einer Fahrkarte

kdu = Kosten der Unterkunft

vpk = Verpflegungskosten

vpf = Verpflegungskosten für die Folgezeit

abe = Arbeitgeberersatzleistungen

wwk = weitere Werbungskosten

Sonderausgaben:

sa = Sonderausgaben

wsa = weitere Sonderausgaben

pb = Pauschbetrag

bl1 = bl-sa

rdl = Renten und dauernde Lasten

ule = Unterhaltsleistungen "Ehegatte"

a8 = Menüantwort

ker = Kirchensteuererstattung im letzten Jahr

kae = gezahlte Kirchensteuer abzüglich der Erstattung

sks = Steuerberatungskosten als Sonderausgabe

aeb = Aufwendungen für die eigene Berufsausbildung

spb = Spenden und Beiträge für wissenschaftliche, kulturelle, mildtätige, kirchliche, religiöse und gemeinnützige Zwecke

Vorsorgeaufwendungen:

ags = Arbeitnehmeranteil zur gesetzlichen Sozialversicherung bzw. andere gleichgestellte Leistungen

agr = Arbeitgeberanteil zur gesetzlichen Rentenversicherung bzw. steuerfreie Zuschüsse des Arbeitgebers zu gleichgestellten Aufwendungen

arh = freiwillige Angestellten-, Renten- und Höherversicherung

fkv = freiwillige Krankenversicherung

fuv = freiwillige Unfallversicherung

lvs = Lebensversicherung und Sterbekasse

hpv = Haftpflichtversicherung

bbs = Beiträge an Bausparkassen

vb = Versicherungsbeiträge (insgesamt)

z29

·
· = Hilfsvariablen (siehe Schema zur Selbstberechnung)

z42

Berechnung der Vorsorgepauschale:

z46

·
· = Hilfsvariablen (siehe Schema zur Selbstberechnung)

z77

vp = Vorsorgepauschale

abv = abziehbarer Betrag für Vorsorgeaufwendungen

Außergewöhnliche Belastungen:

agb = außergewöhnliche Belastungen

kub = Aufwendungen, wenn Sie Körperbehinderter oder Hinterbliebener sind

bhh = Beschäftigung einer Haushaltshilfe oder Heimunterbringung

ff = Freibetrag für besondere Fälle

fuk = Freibetrag bei Unterhaltsverpflichtungen gegenüber Kindern

ubp = Unterstützung bedürftiger Personen

afb = Ausbildungsfreibetrag

kba = Kinderbetreuungskosten Alleinstehender

aab = andere außergewöhnliche Belastungen

pge = Prozent des Gesamtbetrags der Einkünfte

Berechnung des zu versteuernden Einkommens:

vek = zu versteuerndes Einkommen = Gesamtbetrag der Einkünfte - Sonderausgaben - abv - agb - Altersfreibetrag - Haushaltsfreibetrag - Kinderfreibetrag

Berechnung des Erstattungsbetrages:

stt\$
= Lohnsteuer laut Einkommensteuertabelle
stt

u = Einkommensuntergrenze für den Näherungslauf zur
Lohnsteuerbestimmung

o = Einkommensobergrenze für den Näherungslauf zur
Lohnsteuerbestimmung

a = Lohnsteuer zur Einkommensuntergrenze

b = Lohnsteuer zur Einkommensobergrenze

d = b - a

zek = zu erstattende Kirchensteuer

jks = Jahreskirchensteuer

Programmbeschreibung:

Satz 10- 80 : Überschrift

Satz 90- 690 : Angaben zur Person

Satz 700-2240 : Werbungskosten

Satz 2250-3350 : Sonderausgaben, die nicht Vorsorgeauf-
wendungen sind

Satz 3360-3940 : Vorsorgeaufwendungen
3360-3780 : Eingaben
3790-3940 : Berechnung der abziehbaren
Vorsorgeaufwendungen

Satz 3950-4510 : Berechnung der Vorsorgepauschale

Satz 4520-5560 : Außergewöhnliche Belastungen
5460-5520 : Berechnung er Zumutbarkeit

Satz 5570-5630 : Berechnung des zu versteuernden Ein-
kommens

Satz 5640-6130 : Berechnung des Erstattungsbetrages
5780-5930 : Schätzung der Steuer
5940-6130 : Schlußausdruck

Satz 6140 : Programmende

Satz 6150-6190 : Unterprogramm "Warten"

6.4. Ideenfundgrube

- Arbeitslosengeld und Arbeitslosenhilfe - Anspruchsvoraussetzungen prüfen und Berechnungen vornehmen
- der Warenkorb zur Bestimmung des Regelsatzes zur Sozialhilfe - speichern und manipulieren
- Wohngeldtabelle programmiert
- staatliche Förderung für Bauspare (Prämien und Steuervorteile)
- abrufbare Prozeßkostentabelle
- Durchleuchtung des Subventionsdschungels
- der Staatshaushalt auf einem Blick
- und noch vieles mehr



7. Haushalt und Gesundheit

7.1. Vorbemerkung

Um es gleich vorwegzunehmen: in diesem Kapitel wird kein Finanzbuchhaltungsprogramm für den Haushalt vorgestellt. Ein normaler Haushalt ist schließlich auch nicht wie ein mittlerer Handwerksbetrieb zu führen. Sofern Sie also in die Geheimnisse der doppelten Buchführung eingeweiht sind, können Sie diese hier getrost vergessen.

Vielmehr wird Ihnen hier eine programmtechnische Hilfe für die Erhaltung (oder Erlangung) des Überblicks bezüglich der im Haushalt anfallenden Kosten geboten.

Weiterhin geht es in diesem Kapitel um das Thema Gesundheit. Es gibt vielfältige Möglichkeiten Ihren CPC zum Thema Gesundheit einzusetzen. Man denke da zum Beispiel an Jogging- oder Diätpläne. Wir wollen uns hier hingegen mit einem Kalorienzählprogramm begnügen. Dieses Programm ist nicht zuletzt deshalb sehr praktisch, da es ohne größere Probleme für Diabetiker von Kalorien auf Broteinheiten umgestellt werden kann. Die Umstellung auf andere Nährwerte ist natürlich ebenfalls möglich.

7.2. Der rationale Haushalt

Je vernünftiger Sie mit Ihrem Geld umgehen, desto mehr können Sie sich leisten. Diese weitverbreitete Ansicht ist sicherlich richtig. Aber was heißt denn hier vernünftig? Für den Verbraucher (Verbraucher sind wir alle) ist es häufig kaum möglich die Qualität eines Produktes zu beurteilen. Außerdem sind die Preise in unserer Wirtschaft ständig in Bewegung. Es ist also gar nicht so einfach, sich vernünftig zu verhalten.

Daher ist es vielleicht sinnvoll zu fragen, wofür denn das Geld im Haushalt ausgegeben wird. Mancher, der sich diese Frage noch nie gestellt hat, wird sich wundern, wenn er sieht, wie sich seine Ausgaben auf die verschiedenen Kategorien wie Kleidung, Ernährung, Reisen etc. verteilen. Eine derartige Kontrolle Ihrer Geldausgaben umfasst natürlich eine Unmenge an Einzelinformationen. Daher bietet sich der Einsatz Ihres CPC zur Verwaltung dieser Daten an.

Das folgende Programm übernimmt diese Aufgabe. Es hat die folgenden Funktionen:

- Geldausgaben eingeben (mit den Variablen "Warenbezeichnung", "DM-Betrag", "Kategorie" und "Datum")

- Kategorien bilden (bis zu 10 Kategorien können gebildet werden)

Geldausgaben und Kategorien

- einlesen
 - abspeichern
 - ansehen (in Form einer Tabelle)
 - berechnen (für beliebige Zeiträume werden die DM-Beträge pro Kategorie ausgegeben. Die Darstellung der prozentualen Anteile der einzelnen Kategorien erfolgt in Form eines Balkendiagramms.)
- Arbeit beenden

Das Programm verwaltet also nicht nur die einzelnen Informationen, sondern es übernimmt auch die wichtige Aufgabe Berechnungen durchzuführen. So können Sie am Monats- oder Jahresende leicht feststellen, für welche Bereiche (Kategorien) Sie wieviel Geld ausgegeben haben.

Da nur 10 Kategorien gebildet werden können (mehr würden die Übersichtlichkeit einschränken), ist es sinnvoll, eine Kategorie "sonstiges" zu bilden, um Zuordnungsschwierigkeiten zu vermeiden. Das Einlesen müssen Sie vor der Eingabe neuer Daten erledigen, da ansonsten die schon eingegebenen Informationen überschrieben werden.

Für den Diskettenbetrieb ist die Einleseroutine zu ändern, da ein konkreter Dateiname angegeben werden muß. Dies kann folgendermaßen aussehen:

```
795 INPUT "Wie heisst die Datei ";n$  
800 OPENIN n$
```

Insgesamt ist im Programm Platz für jeweils 1000 Eingaben reserviert worden. Sie können diese Zahl zwar auch noch erhöhen (Satz 80-90 des Programms, aber: Speicherkapazität Ihres Rechners beachten!), jedoch wäre eine Monats- oder Jahres-"Endabrechnung" sicherlich sinnvoller.

Ansonsten ist das Programm weitestgehendst selbsterklärend. Nur nach der Ausgabe des Balkendiagramms wird die Aufforderung "Bitte eine Taste drücken" nicht ausgedruckt, da der zur Verfügung stehende Platz auf dem Bildschirm für das Diagramm verwendet wird und ein "Hochrollen" des Bildschirms vermieden werden sollte.

Programm:

```
10 REM I17
20 CLS
30 PRINT "I17 - Programm zur Kontrolle"
40 PRINT TAB(7)"Ihrer Geldausgaben"
50 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
60 PRINT TAB(9) "Bernd Kowal, 1985"
70 GOSUB 2060
80 DIM wb$(1000),dm(1000),k(1000)
90 DIM tg(1000),mo(1000),ja(1000),ja$(1000)
100 REM
110 REM Menue
120 REM
130 PRINT TAB(17)"Menue":PRINT:PRINT
140 PRINT TAB(33)"Eingabe":PRINT
150 PRINT"Geldausgaben eingeben";TAB(36)"1":PRINT
160 PRINT"Kategorien bilden";TAB(36)"2":PRINT
170 PRINT"Geldausgaben und Katgorien":PRINT
180 PRINT" einlesen";TAB(36)"3":PRINT
190 PRINT" abspeichern";TAB(36)"4":PRINT
200 PRINT" ansehen";TAB(36)"5":PRINT
210 PRINT" berechnen";TAB(36)"6"
220 PRINT:PRINT"Arbeit beenden";TAB(36)"7"
230 PRINT:PRINT:INPUT"Ihre Wahl ";a:CLS
240 IF a<1 OR a>7 THEN CLS:GOTO 130
250 ON a GOTO 270,620,780,920,1080,1240,1930
260 REM
```

```

270 REM Geldausgaben eingeben
280 REM
290 IF ka>0 THEN 320
300 PRINT"Sie muessen erst Kategorien bilden."
310 GOSUB 2050:GOTO 620
320 WINDOW #2,20,40,1,25
330 PRINT #2,"Kategorien":PRINT #2
340 FOR i=1 TO ka
350 PRINT #2,i;". ";k$(i)
360 NEXT i
370 WINDOW #1,1,19,1,25
380 ga=ga+1
390 PRINT #1,"Die Eingabe wird"
400 PRINT #1,"beendet,wenn Sie"
410 PRINT #1,"fuer die Waren-"
420 PRINT #1,"bezeichnung eine"
430 PRINT #1,"0 eingeben.":PRINT #1
440 PRINT #1,"Bitte machen Sie"
450 PRINT #1,"folgende Eingaben.":PRINT #1:PRINT #1
460 PRINT #1,ga;".te Eingabe.":PRINT #1
470 PRINT #1,"Warenbezeichnung"
480 INPUT #1,wb$(ga):PRINT #1
490 IF wb$(ga)="0" THEN ga=ga-1:CLS:GOTO 110
500 PRINT #1,"DM-Betrag"
510 INPUT #1,dm(ga):PRINT #1
520 PRINT #1,"Kategorie"
530 INPUT #1,"Nr. ";k(ga):PRINT #1
540 IF k(ga)<1 OR k(ga)>ka THEN 520
550 PRINT #1,"Datum"
560 INPUT #1,"Tag ";tg(ga)
570 INPUT #1,"Monat ";mo(ga)
580 INPUT #1,"Jahr ";ja$(ga)
590 ja$(ga)=RIGHT$(ja$(ga),2):ja(ga)=VAL(ja$(ga))
600 CLS #1:GOTO 380
610 REM
620 REM Kategorien bilden
630 REM
640 IF ka=0 THEN 690
650 PRINT"Folgende Kategorien bestehen bereits : "
660 PRINT:FOR i=1 TO ka
670 PRINT i;".te Kategorie. ";k$(i)
680 NEXT i:PRINT:PRINT
690 PRINT"Sie koennen max. 10 Kategorien bilden."
700 PRINT"Geben Sie eine 0 ein, wenn Sie keine"
710 PRINT"weitere Kategorie mehr bilden wollen.":PRINT
720 ka=ka+1
730 IF ka=11 THEN ka=ka-1:GOSUB 2050:GOTO 110

```

```

740 PRINT ka;"te Kategorie ";:INPUT k$(ka)
750 IF k$(ka)="0" THEN ka=ka-1:CLS:GOTO 110
760 GOTO 720
770 REM
780 REM Geldausgaben und Kategorien einlesen
790 REM
800 OPENIN ""
810 INPUT #9,ga,ka
820 FOR i=1 TO ga
830 INPUT #9,wb$(i)
840 INPUT #9,dm(i),k(i),tg(i),mo(i),ja(i)
850 NEXT i
860 FOR i=1 TO ka
870 INPUT #9,k$(i)
880 NEXT i
890 CLOSEIN
900 CLS:GOTO 110
910 REM
920 REM Geldausgaben und Kategorien abspeichern.
930 REM
940 PRINT"Unter welchem Namen sollen die Infor-"
950 INPUT"mationen abgespeichert werden ";n$
960 OPENOUT n$
970 PRINT #9,ga,ka
980 FOR i=1 TO ga
990 PRINT #9,wb$(i)
1000 PRINT #9,dm(i),k(i),tg(i),mo(i),ja(i)
1010 NEXT i
1020 FOR i=1 TO ka
1030 PRINT #9,k$(i)
1040 NEXT i
1050 CLOSEOUT
1060 CLS:GOTO 110
1070 REM
1080 REM Geldausgaben und Kategorien ansehen
1090 REM
1100 MODE 2:af=1:h=20
1110 PRINT"Eingabe-Nr.   Warenbezeichnung   ";
1120 PRINT"DM-Betrag   Kategorie   Datum"
1130 PRINT:FOR i=af TO ga
1140 PRINT USING "   #####   ";i;
1150 PRINT USING "   ö   ö";wb$(i);
1160 PRINT USING "   #####.##";dm(i);
1170 PRINT USING "   ö   ö   ";k$(k(i));
1180 PRINT USING "##.";tg(i);mo(i);
1190 PRINT USING "##";ja(i)
1200 IF i=h THEN h=h+20:af=af+20:GOSUB 2060:GOTO 1110

```

```

1210 NEXT i
1220 GOSUB 2050:MODE 1:GOTO 110
1230 REM
1240 REM Berechnungen
1250 REM
1260 PRINT"Geben Sie bitte den Zeitraum an, fuer"
1270 PRINT"den Berechnungen erfolgen sollen.":PRINT
1280 PRINT"von :";TAB(18)"bis :":PRINT
1290 PRINT:PRINT"Tag";TAB(18)"Tag":PRINT
1300 PRINT"Monat";TAB(18)"Monat":PRINT
1310 PRINT"Jahr";TAB(18)"Jahr"
1320 LOCATE 7,7:INPUT at
1330 LOCATE 7,9:INPUT am
1340 LOCATE 7,11:INPUT aja$
1350 LOCATE 24,7:INPUT et
1360 LOCATE 24,9:INPUT em
1370 LOCATE 24,11:INPUT eja$
1380 aja$=RIGHT$(aja$,2):aja=VAL(aja$)
1390 eja$=RIGHT$(eja$,2):eja=VAL(eja$)
1400 PRINT:PRINT:TAB(8)"Ich rechne !"
1410 su=0:FOR i=1 TO ka
1420 ks(i)=0:NEXT i
1430 FOR i=1 TO ga
1440 IF ja(i)<aja OR ja(i)>eja THEN 1510
1450 IF ja(i)>aja AND ja(i)<eja THEN 1490
1460 IF mo(i)<am OR mo(i)>em THEN 1510
1470 IF mo(i)>am AND mo(i)<em THEN 1490
1480 IF tg(i)<at OR tg(i)>et THEN 1510
1490 ks(k(i))=ks(k(i))+dm(i)
1500 su=su+dm(i)
1510 NEXT i
1520 CLS:PRINT"In diesem Zeitraum haben Sie insge-"
1530 PRINT"samt";INT(su*100+0.5)/100;"DM ausgegeben."
1540 PRINT:PRINT
1550 PRINT"Fuer die einzelnen Kategorien"
1560 PRINT"ergibt sich folgendes Bild :":PRINT
1570 PRINT"Nr. der      Name der";TAB(35)"DM"
1580 PRINT"Kategorie  Kategorie":PRINT
1590 FOR i=1 TO ka
1600 PRINT USING "#####";i;
1610 PRINT USING "      ö                ö";k$(i);
1620 z=INT(ks(i)*100+0.5)/100
1630 PRINT TAB(30) USING "#####.##";z
1640 NEXT i
1650 GOSUB 2050
1660 PRINT"Graphische Darstellung der gerundeten"
1670 PRINT"prozentualen Anteile der einzelnen"

```

```

1680 PRINT"Kategorien an den Gesamtausgaben : "
1690 pm=0
1700 FOR i=1 TO ka
1710 p(i)=ks(i)*100/su
1720 IF p(i)>pm THEN pm=p(i)
1730 NEXT i
1740 FOR i=1 TO ka
1750 pp(i)=p(i)*100/pm
1760 NEXT i
1770 PLOT 17,17:DRAW 17,320
1780 PLOT 17,17:DRAW 520,17
1790 LOCATE 1,5:PRINT""
1800 LOCATE 34,25:PRINT"Kateg."
1810 j=0
1820 FOR i=3 TO ka*3 STEP 3
1830 j=j+1
1840 LOCATE i,25:PRINT j
1850 x=ROUND(18*pp(j)/100)
1860 FOR l=24 TO 24-x STEP -1
1870 LOCATE i+1,l:PRINT CHR$(143)
1880 NEXT l
1890 LOCATE i,23-x:PRINT USING "##"; p(j)
1900 NEXT i
1910 GOSUB 2100:GOTO 110
1920 REM
1930 REM Programm beenden
1940 REM
1950 PRINT"Haben Sie auch alle Informationen"
1960 PRINT"ordnungsgemaess abgespeichert ?"
1970 PRINT"Falls nicht, geben Sie einfach"
1980 PRINT"ein 'oh' ein (ansonsten reicht die"
1990 PRINT"Leertaste). "
2000 INPUT a$
2010 IF a$="oh" THEN CLS:GOTO 920
2020 PRINT:PRINT:PRINT"      Auf Wiedersehen"
2030 PRINT:PRINT"Sie koennen mich jetzt ausschalten"
2040 END
2050 REM
2060 REM UP Warten
2070 REM
2080 LOCATE 5,25
2090 PRINT"Bitte eine Taste druecken !"
2100 x$=INKEY$
2110 IF x$="" THEN 2100
2120 CLS:RETURN

```

Variablenliste:

a = Antwort "Menüauswahl"

a\$ = Antwortstring im Programmteil "Arbeit beenden"

af = Schleifenanfangswert

aja = Jahresangabe (Anfang)

aja\$ = Eingabefeld für aja

am = Monatsangabe (Anfang)

at = Tagesangabe (Anfang)

dm(i) = DM-Betrag

eja = Jahresangabe (Ende)

ea\$ = Eingabefeld für eja

em = Monatsangabe (Ende)

et = Tagesangabe (Ende)

ga = Anzahl der Geldausgaben

h = Hilfsvariable

i = Laufindex

j = Index

ja(i) = Jahr der Geldausgabe

ja\$(i) = Eingabefeld für ja(i)

k(i) = Kategoriennummer bezüglich einer Geldausgabe

$k(i)$ = Kategorienbezeichnung

ka = Anzahl der definierten Kategorien

$ks(i)$ = Summe der Geldausgaben für die jeweilige Kategorie bezüglich eines Zeitraumes

l = Schleifenanfangswert für das Zeichnen der Balken

$mo(i)$ = Monat der Geldausgabe

$n\$$ = Name der abzuspeichernden Informationen

$p(i)$ = Prozentsatz der Geldausgaben pro Kategorie bezogen auf die Gesamtausgaben

pm = maximaler Wert bezüglich der Variablen $p(i)$

$pp(i)$ = Prozentsatz der Geldausgaben pro Kategorie bezogen auf den maximalen Wert pm

su = Summe der Geldausgaben bezüglich eines Zeitraumes

$tg(i)$ = Tag der Geldausgabe

$wb(i)$ = Warenbezeichnung

x = Hilfsvariable zur Bestimmung des Schleifenendwertes in der Programmschleife zum Zeichnen der Balken

z = auf zwei Dezimalstellen gerundeter Wert bezüglich der Variablen ks

Programmbeschreibung:

Satz 10- 70 : Überschrift

Satz 80- 90 : Speicherplatzreservierung

Satz 100-250 : Menüauswahl und entsprechende Programmverzweigungen

Satz 260-600 : Eingabe der Geldausgaben

In diesem Programmteil werden zwei Bildschirmbereiche definiert. Während im ersten Fenster (linker Bildschirmbereich, definiert in Satz 370) die Eingaben erfolgen, werden im zweiten Fenster(rechter Bildschirmbereich, definiert in Satz 320) die vorhandenen Kategorien ausgegeben. Somit hat der Programmbenutzer die einzelnen Kategorien immer vor Augen. Nach erfolgter Eingabe wird dann in Satz 600 nur das erste Fenster geräumt. Sind noch keine Kategorien festgelegt worden (siehe Satz 290-310), wird ein Kommentar ausgegeben und zum Programmteil "Kategorien bilden" verzweigt. Die Eingabe einer 0 für die Variable "Warenbezeichnung" bewirkt den Rücksprung zur Menüauswahl (Satz 490). Das Eingabefeld für die Variable ja ist ein String, um in Satz 590 aus der möglichen Eingabe mit Jahrhundertangabe (z. B. 1985) nur die letzten beiden Ziffern (also 85) im Feld ja(i) abzulegen.

Satz 610- 760 : Bildung der Kategorien

Falls schon Kategorien vorhanden sind, werden diese ausgegeben (Satz 640-680). Anschließend können Kategorien gebildet

werden. Wenn 10 Kategorien gebildet sind oder eine 0 eingegeben wird, erfolgt ein Rücksprung zur Menüauswahl (siehe Satz 730 und 750).

Satz 770- 900 : Geldausgaben und Kategorien werden eingelesen

Satz 910-1060 : Geldausgaben und Kategorien werden abgespeichert
Beim Einlesen und Abspeichern von String-Variablen ist zu beachten, daß dies in jeweils gesonderten INPUT- bzw. PRINT-Anweisungen geschieht (siehe Satz 830-840 bzw. Satz 990-1000).

Satz 1070-1220 : Geldausgaben und Kategorien ansehen
Dazu wird in Satz 1100 der Bildschirmmodus auf 80 Zeichen pro Zeile gesetzt. Mit Hilfe der PRINT USING-Anweisung erfolgt dann eine formatierte tabellarische Ausgabe. Die Variablen af und h dienen der Ausgabebeschränkung auf jeweils 20 Tabellenzeilen. Vor dem Rücksprung zur Menüauswahl wird wieder auf den 40-Zeichen-Modus umgeschaltet (Satz 1220).

Satz 1230-1910 : Berechnungen

1260-1390 : Eingabe eines Zeitraumes für den die Berechnungen erfolgen sollen. Dabei wird zuerst der Eingabetext ausgegeben und anschließend werden die INPUT-Statements entsprechend positioniert. Aus den gleichen Gründen wie in Satz 590 werden die Jahresangaben als Strings eingegeben.

- 1400 : Kommentar
- 1410-1420 : Die zu berechnenden Variablen s_u und $k_s(i)$ werden mit dem Anfangswert 0 belegt.
- 1430-1510 : Liegt eine Geldausgabe innerhalb des angegebenen Zeitraumes, so werden die Variablen s_u und k_s um den entsprechenden DM-Betrag erhöht. Der Vergleich der Jahres-, Monats- und Tagesangabe zu einer Geldausgabe mit dem angegebenen Zeitraum ist sozusagen nur für dieses Jahrhundert gültig, da in die Jahresangabe nicht das zugehörige Jahrhundert eingeht. Beispielsweise wird die Jahresangabe 2005 genauso behandelt wie die Angabe 1905.
- 1520-1650 : Formatierte Ausgabe der auf zwei Dezimalstellen gerundeten DM-Beträge (pro Kategorie und insgesamt).
- 1660-1680 : Überschrift zur graphischen Darstellung der prozentualen Anteile der einzelnen Kategorien an den Gesamtausgaben.
- 1690 : Die Variable p_m wird gleich 0 gesetzt.
- 1700-1730 : Berechnung der Prozentätze der Geldausgaben pro Kategorie bezogen auf die Gesamtausgaben und Bestimmung des maximalen Prozentsatzes p_m .
- 1740-1760 : Die Prozentsätze der Ausgaben pro Kategorie bezogen auf den maximalen Prozentsatzwert p_m

werden berechnet. Dies dient dazu, den auf dem Bildschirm zur Verfügung stehenden Platz optimal auszunutzen (siehe Satz 1850-1880).

- 1770-1780 : Ein Koordinatenkreuz wird gezeichnet. Bevor die Linien mit der DRAW-Anweisung gezeichnet werden, wird der Graphikcursor jeweils mit der PLOT-Anweisung auf den Koordinatenursprung gesetzt.
- 1790-1800 : Die Koordinatenachsen werden beschriftet
- 1810 : Die Indexvariable j wird gleich 0 gesetzt. .
- 1820-1900 : In dieser Programmschleife wird die x -Achse beschriftet und die Balken werden gezeichnet. Die Variable i dient dabei der waagerechten Positionierung des Cursors, während die Variable j die Prozentsatzwerte $pp(i)$ der einzelnen Kategorien identifiziert. Bei dem größten Prozentsatzwert ist auch die Variable x am größten, so daß dann die 18 zur Verfügung stehenden Zeilen für die Balkeneinzeichnung (Satz 1860-1880) genutzt werden. Bei kleineren Werten $pp(i)$ vergrößert sich der Schleifenendwert in Satz 1860 entsprechend, so daß dann kürzere Balken gezeichnet werden. In Satz 1890 werden direkt über


den jeweiligen Balken die zugehörigen prozentualen Anteile der jeweiligen Kategorie an den Gesamtausgaben ausgegeben. Da dies gerundete Werte sind, kann es passieren, daß die Summe der ausgegebenen Prozentsatzwerte nicht gleich 100 ist.

1910 : Ende dieses Programmteils

Satz 1920-2040 : Programmende

Es besteht in diesem Programmteil noch die Möglichkeit - falls dies vergessen wurde - eingegebene Daten abzuspeichern.

Satz 2050-2120 : Unterprogramm "Warten"



7.3. Ein elektronischer Kalorienzähler

Jeder der schon einmal eine Diät gemacht hat weiß, daß es gar nicht so einfach ist, seine tägliche Kalorienzufuhr zu berechnen. Die Angaben in den Kalorientabellen beziehen sich nämlich in der Regel auf eine bestimmte Gramm- oder Literzahl.

Sie stehen dabei beispielsweise vor folgendem Dreisatzproblem:

eine Flasche Altbier (0.5 Liter) hat 210 Kilokalorien,
wieviel Kilokalorien x hat nun ein Glas (0.2 Liter) ?

Sie müssen eine einfache Gleichung aufstellen, um diese Rechenaufgabe zu lösen:

$$\frac{210}{0.5} = \frac{x}{0.2}$$

$$x = 0.2 * 210 / 0.5 = 84$$

Wollten Sie diese Rechnung für jedes Nahrungsmittel das Sie täglich zu sich nehmen durchführen, so hätten Sie eine Menge zu tun. Der Einsatz eines Taschenrechners wäre da schon ratsam. Mit Ihrem CPC und einem entsprechenden Programm läßt sich dieser Zeitaufwand aber noch reduzieren.

Sie müssen sich dann nur einmal die Arbeit machen und alle für Sie wichtigen Nahrungsmittel in das Programm eingeben.

Die Eingabe erfolgt mit Hilfe des DATA-Statements. Jede DATA-Anweisung ist dabei wie folgt aufgebaut:

DATA < Nahrungsmittel, Maßeinheit, Anzahl der Maßeinheiten auf die sich die Kalorienangabe bezieht, Kalorienangabe >

zum Beispiel:

DATA < Broetchen, Stueck (gleich 40 Gramm), 1, 111 >

mit anderen Worten: 1 Brötchen (gleich 40 Gramm) hat 111 Kilokalorien.

Im vorliegenden Programm sind nur 15 DATA-Anweisungen exemplarisch aufgeführt. Diese Anweisungskette können Sie beliebig erweitern oder verändern. Eventuell müssen Sie dann auch den Wert der Variablen k in Satz 100 ändern, da nur k DATA-Statements gelesen werden.

Im Programmablauf ist bei der Kalorienberechnung nur das Nahrungsmittel und eine Zahl bezüglich der Menge einzugeben, also beispielsweise:

Nahrungsmittel ? Broetchen
Wieviel Stueck (gleich 40 Gramm) ? 2.5

Als Ergebnis erhalten Sie dann:

Das sind 277.5 Kilokalorien

Die einzelnen Ergebnisse werden natürlich aufsummiert, so daß sich beispielsweise für die Nahrungsmittelzufuhr eines Tages ergeben könnte:

Insgesamt ergeben sich somit 2428.5 Kilokalorien

Bei der Eingabe des Nahrungsmittels achten Sie bitte auf eine exakte Schreibweise, da ansonsten die Meldung

Zu dem angegebenen Nahrungsmittel
sind keine Informationen vorhanden

auf dem Bildschirm erscheint.

Programm:

```
10 REM I18
20 CLS
30 PRINT "I18 - Programm zur Kalorien-"
40 PRINT TAB(7)"berechnung"
50 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
60 PRINT TAB(9) "Bernd Kowal, 1985"
70 GOSUB 1480
80 REM k ist gleich der Anzahl der eingegebenen
90 REM Nahrungsmittel
100 k=15
110 DIM n$(k),m$(k),am(k),ka(k)
120 DATA Butter,Gramm,100,775
130 DATA Ei,Stueck (Klasse 4),1,92
140 DATA Banane,Stueck (klein),1,99
150 DATA Broetchen,Stueck (gleich 40 Gramm),1,111
160 DATA Aal,Gramm,100,299
170 DATA Pommes frites,Gramm,100,538
180 DATA Edamer Kaese (40% Fett),Gramm,100,331
190 DATA Altbier,Liter,0.2,84
200 DATA Milch (3.5% Fett),Liter,1,660
210 DATA Nudeln (gekocht),Essloeffel
    (gleich 40 Gramm),1,37
220 DATA Doppelbock,Liter,0.2,132
230 DATA Rehkeule,Gramm,100,106
240 DATA Knoblauch,Zehe (5 Gramm),1,7
250 DATA Marmelade,Teeloeffel (gleich 5 Gramm),1,38
260 DATA Rosenkohl,Gramm,100,52
1000 FOR i=1 TO k
1010 READ n$(i),m$(i),am(i),ka(i)
1020 NEXT i
1030 PRINT TAB(16)"Menue":PRINT:PRINT:PRINT
```

```

1040 PRINT TAB(33)"Eingabe":PRINT:PRINT
1050 PRINT"alle Daten ansehen";TAB(36)"1":PRINT
1060 PRINT"Kalorien berechnen";TAB(36)"2":PRINT
1070 PRINT"Arbeit beenden";TAB(36)"3":PRINT
1080 PRINT:PRINT:INPUT"Ihre Wahl ";a
1090 IF a<1 OR a>3 THEN CLS:GOTO 1030
1100 CLS:ON a GOTO 1110,1260,1470
1110 FOR i=1 TO k-1
1120 FOR j=i+1 TO k
1130 IF n$(i)<=n$(j) THEN 1180
1140 h$=n$(j):n$(j)=n$(i):n$(i)=h$
1150 m$=m$(j):m$(j)=m$(i):m$(i)=h$
1160 h=am(j):am(j)=am(i):am(i)=h
1170 h=ka(j):ka(j)=ka(i):ka(i)=h
1180 NEXT j
1190 NEXT i
1200 l=20
1210 FOR i=1 TO k
1220 PRINT n$(i)
1230 IF i=1 THEN l=l+20:GOSUB 1480
1240 NEXT i
1250 GOSUB 1480:GOTO 1030
1260 ksu=0
1270 PRINT"Betaetigen Sie bei der Anforderung"
1280 PRINT"eines Nahrungsmittels einfach die"
1290 PRINT"Leertaste, so wird die Kalorien="
1300 PRINT"berechnung abgebrochen.":PRINT:PRINT
1310 INPUT"Nahrungsmittel ";en$:PRINT
1320 IF en$="" THEN 1390
1330 FOR i=1 TO k
1340 IF n$(i)<>en$ THEN 1430
1350 PRINT"Wieviel ";m$(i);" ";INPUT em
1360 ks=em*ka(i)/am(i):PRINT:PRINT
1370 PRINT"Das sind";ks;"Kilokalorien.":PRINT
1380 ksu=ksu+ks
1390 PRINT:PRINT"Insgesamt ergeben sich somit";
1400 PRINT ksu;"Kilokalorien."
1410 GOSUB 1480
1420 IF en$="" THEN 1030 ELSE 1270
1430 NEXT i
1440 PRINT:PRINT"Zu dem angegebenen Nahrungsmittel"
1450 PRINT"sind keine Informationen vorhanden."
1460 GOSUB 1480:GOTO 1270
1470 END
1480 REM UP Warten
1490 LOCATE 5,25
1500 PRINT"Bitte eine Taste druecken !"
1510 x$=INKEY$
1520 IF x$="" THEN 1510
1530 CLS:RETURN

```

Variablenliste:

a = Antwort "Menüauswahl"

am(i) = Anzahl der Maßeinheiten auf die sich die Kalorienangabe bezieht

em = Anzahl der eingegebenen Maßeinheiten

en\$ = eingegebenes Nahrungsmittel

i = Laufindex

j = Laufindex

k = Anzahl der DATA-Statements

ka(i) = Kalorienangabe

k = berechnete Kalorienanzahl

ksu = Summe der berechneten Kalorien

l = Hilfsvariable

m\$(i) = Maßeinheit

n\$(i) = Nahrungsmittel

Programmbeschreibung:

Satz 10- 70 : Überschrift

Satz 80- 90 : Kommentar

Satz 100 : Belegung der Variablen k

Satz 110 : Speicherplatzreservierung

Satz 120- 260 : Dem Programm werden Daten zur Verfügung gestellt

Satz 1000-1020 : Die Daten werden eingelesen

Satz 1030-1100 : Auswahlmenü mit den entsprechenden Programmverzweigungen

Satz 1110-1250 : Programmteil "Daten ansehen"

1110-1190 : Alphabetisches Sortieren der Daten (siehe dazu auch Kapitel 2.4)

1200-1240 : Jeweils 20 (siehe Hilfsvariable l) Nahrungsmittel werden ausgegeben

1250 : "Warten" und Rücksprung zur Menüauswahl

Satz 1260-1460 : Programmteil "Kalorien berechnen"

1260 : Die Variable ksu wird gleich 0 gesetzt

1270-1300 : Erläuterungen zum Programmablauf werden ausgegeben

1310 : Eingabe eines Nahrungsmittels

1320 : Falls die Leertaste gedrückt wird, erfolgt eine Verzweigung zur Ausgabe des Gesamtergebnisses

1330-1430 : Die durch Eingabe (Satz 1310) belegte Variable en\$ wird mit den gespeicherten Variablen n\$(i) verglichen. Bei Gleichheit wird ein Wert bezüglich der Menge angefordert (1350), die Kalorienanzahl wird berechnet (1360), das Ergebnis wird ausgegeben (1370) und im Feld ksu wird die Summe aller Berechnungen gebildet (1380). In Satz 1390 und 1400 wird dann das Zwischenergebnis (bzw. Gesamtergebnis, siehe Satz 1320) ausgegeben. Nach einem Sprung ins Unterprogramm "Warten" (1410), erfolgt in Satz 1420 eine Verzweigung zur Menüauswahl (falls die Berechnungen abgebrochen werden sollen) oder zur erneuten Eingabe (1270). Bei Ungleichheit wird der Vergleich fortgesetzt (Sprung zum Satz 1430).

1440-1460 : Ausgabe eines Kommentars, "Warten" und Verzweigung zur erneuten Eingabe, falls sich der Name des eingegebenen Nahrungsmittels en\$ nicht unter den gespeicherten Variablen n\$(i) befindet.

Satz 1470 : Programmende

Satz 1480-1530 : Unterprogramm "Warten"

7.4. Ideenfundgrube

- abrufbare Preislisten z. B. für Obst und Gemüse
- Organisation der Haushaltsarbeit
- Ordnung in der Tiefkühltruhe
- Rezeptkartei für Hobbyköche
- persönliches Gesundheitsarchiv (Blutzucker-Werte, Kreislauf, Puls usw. erfassen, berechnen und bewerten)
- Diätpläne
- Jogging- und Fitnesspläne
- Risikoberechnungen (z. B. die Berechnung der Lebenserwartung in Abhängigkeit vom Rauchen)
- und noch vieles mehr

8. Ihr Rechner und Ihre Hobbys

8.1. Vorbemerkung

Dieses Kapitel ist vielleicht eines der interessantesten in diesem Buch, da Sie den CPC für die Pflege vieler Hobbys einsetzen können. Egal ob Sie nun Briefmarken sammeln oder Anhänger einer bestimmten Sportart sind, mit einem Computer und den entsprechenden Programmen läßt sich vieles einfacher gestalten. Das reicht von der Übernahme organisatorischer Arbeiten (wo finde ich diese oder jene Briefmarke) bis hin zur direkten Ausübung eines Hobbys (damit ist nicht das Programmieren selbst gemeint, sondern z. B. das Musizieren mit dem Computer für den Musikbegeisterten).

Natürlich ist es hier nicht möglich, auf alle nur denkbaren Einsatzmöglichkeiten Ihres Rechners näher einzugehen. Vielmehr nehmen zwei Programmbeispiele einen breiten Raum ein, die Ihnen die technischen Möglichkeiten Ihres CPC verdeutlichen sollen. Das sind einmal die exzellenten Graphikeigenschaften und zum anderen die hervorragenden Möglichkeiten der Musikprogrammierung. Diese beiden Anwendungsbereiche sind bisher auch ein wenig vernachlässigt worden. Ein drittes Beispiel beschäftigt sich mit König Fußball, da der Fußball immer noch die beliebteste Sportart und somit Hobby vieler Leute in unserem Land ist.

Viele weitere Anwendungsbereiche sind jedoch denkbar. In der Ideenfundgrube zu diesem Kapitel finden Sie dazu einige Anregungen. Sie bleiben also wiederum aufgefordert, ausgehend von den hier gemachten Vorschlägen, auf Ihre persönlichen Bedürfnisse zugeschnittene Programme zu entwickeln. Vielleicht hilft Ihnen Ihre Liebe zu Ihrem Hobby über die unter Umständen auftretenden Schwierigkeiten bei der Verwirklichung Ihrer Software-Wünsche hinweg. Vielleicht wird jedoch auch das Programmieren zu Ihrem neuen Hobby (falls es das noch nicht ist).

8.2. Malen, zeichnen und die Erstellung von Graphiken

Im heutigen EDV-Markt zeichnet sich ein unverkennbarer Trend zu grafikfähigen Systemen ab. Da die Tastatur der Rechner bei der Erstellung von Bildern häufig überfordert ist, sind in den letzten Jahren neue Schnittstellen zwischen dem Computer und dem Menschen entwickelt worden. Bei diesen Schnittstellen handelt es sich quasi um neue Eingabegeräte. Neben dem schon länger bekannten Joystick sind insbesondere die Instrumente "Maus" und Graphik-Tablett zu nennen. Vor allem das Graphik-Tablett ähnelt dabei dem konventionellen Gedankenträger Papier. Diese zusätzlichen Eingabegeräte sollen jedoch im folgenden unberücksichtigt bleiben, da die Darstellung der professionellen Graphik-Verarbeitung sicherlich den Rahmen des vorliegenden Buches sprengen würde. Sollten Sie hingegen über das vergleichsweise einfache Instrument Joystick verfügen, so ist eine Änderung des hier vorgestellten Graphik-Programms mit wenigen Handgriffen (siehe Benutzerhandbuch) zu erledigen. Mit Hilfe des vorliegenden Programms können Sie beliebige Bilder (Graphiken oder Zeichnungen) erstellen, die Sie abspeichern und wieder einlesen können. Ob das nun direkt eine künstlerische Arbeit wird, oder ob Sie den Grundriß eines Hauses aufzeichnen, bleibt Ihnen überlassen. Auf jeden Fall werden die jeweils aktuellen x- und y-Koordinaten auf dem Bildschirm angezeigt (rechts oben die y-Koordinate und rechts unten die x-Koordinate), um die Darstellung exakter technischer Zeichnungen zu ermöglichen.

Zur Steuerung des "Stiftes" auf dem Bildschirm werden im Programmablauf einige Tasten mit speziellen Funktionen belegt. Sofern es sich dabei um INPUT-Anweisungen handelt, vollzieht sich die Eingabe farblos, d. h. der "Stift" und das "Papier" haben die gleiche Farbe, um das Bild der Graphik nicht zu beeinträchtigen. In der Regel wird jedoch der Wert der gedrückten Taste mit Hilfe der Anweisung INKEY\$ zurückgegeben. Die ENTER-Taste muß dann bekanntlich nicht betätigt werden.

Im Gegensatz zu den vorangegangenen Programmen werden hier also keine Erläuterungen zur Programm Benutzung ausgegeben. Entweder haben Sie die Bedeutung der Steuertasten im Kopf oder Sie benutzen die folgende Aufstellung:

Steuertaste	Bedeutung
↑	Linie nach oben
→	Linie nach rechts
↓	Linie nach unten
←	Linie nach links
a farblose Linie bzw.	"radieren" Anfang
b farblose Linie bzw.	"radieren" Ende
k Kreis	
bel. Zahl (gr. ENTER)	Radius in Pixel
bel. Zahl (gr. ENTER)	Winkel in Grad

Steuertaste	Bedeutung
d	Diagonale bzw. durchgezogene Linie
für die Steigung:	
bel. Zahl (gr. ENTER)	x-Koordinate (waagrecht)
bel. Zahl (gr. ENTER)	y-Koordinate (senkrecht)
COPY-Taste	Linie ziehen
kl. ENTER	Beendigung dieser Funktion
g	ausgefüllter Kreis
bel. Zahl (gr. ENTER)	Radius in Pixel
bel. Zahl (gr. ENTER)	Winkel in Grad
c	Farbänderung
b (gr. Enter)	Farbstift wird blau
r (gr. Enter)	Farbstift wird rot
s (gr. Enter)	Farbstift wird schwarz
s	Programmende

Nachdem Sie die Taste "s" gedrückt haben, geben Sie PEN 1 ein, um den Cursor wieder sichtbar zu machen.

Die ausgegebenen Erläuterungen sind auf die Anwendung eines Cassettengerätes ausgerichtet. Für den Diskettenbetrieb sind daher die Sätze 150, 190, 260, 310 und 320 entsprechend zu ändern. Weiterhin ist dann in den Sätzen 200 und 300 ein kürzerer Dateiname zu wählen.

Da der Umgang mit diesem Programm Schwierigkeiten bereiten könnte, ist im Anschluß an die Programmbeschreibung ein kleines Beispiel aufgeführt, in dem alle Steuertasten mindestens einmal berücksichtigt werden. Auch die Programmbeschreibung selbst ist etwas ausführlicher, da der eine oder andere Leser vielleicht Probleme mit den Graphik-Anweisungen hat.

Programm:

```
10 REM I19
20 CLS
30 PRINT"I19 - Programm zur Erstellung einer"
40 PRINT TAB(7)"Zeichnung bzw. Graphik, welche"
50 PRINT TAB(7)"extern gespeichert und"
60 PRINT TAB(7)"eingelesen werden kann"
70 PRINT:PRINT:PRINT
80 PRINT TAB(11)"Bernd Kowal, 1985"
90 GOSUB 1100
100 BORDER 26
110 INK 0,26
120 INK 1,0
130 INK 2,6
140 INK 3,11
150 PRINT"Soll eine Zeichnung von Cassette"
160 INPUT"eingelesen werden (j/n) ";nj#
170 IF nj#="n" THEN PRINT:PRINT:GOTO 220
180 IF nj#<>"j" THEN 150
190 PRINT:PRINT:PRINT "press PLAY then any key : "
200 OPENIN "!Zeichnung"
```

```

210 INPUT #9,x,y:GOTO 340
220 PRINT"Geben Sie bitte die Ausgangskoordinaten"
230 PRINT"fuer den Graphikcursor an :":PRINT
240 INPUT"x-Koordinate ";x
250 INPUT"y-Koordinate ";y:PRINT:PRINT
260 PRINT"Soll die Zeichnung auf Cassette"
270 INPUT"gespeichert werden (j/n) ";jn#
280 IF jn#="n" THEN 340
290 IF jn#<>"j" THEN 260
300 OPENOUT "!Zeichnung":PRINT:PRINT
310 PRINT "Fuer die Cassettenspeicherung"
320 PRINT "press REC and PLAY then any key : "
330 PRINT #9,x,y
340 GOSUB 1100
350 KEY DEF 0,1,79
360 KEY DEF 1,1,82
370 KEY DEF 2,1,85
380 KEY DEF 8,1,76
390 KEY DEF 9,1,68
400 KEY DEF 6,1,83
410 PLOT x,y
420 f=1
430 IF EOF=-1 AND nj#="j" THEN CLOSEIN:PEN 0:END
440 IF nj#="j" THEN INPUT #9,a#:GOTO 460
450 a#=INKEY#:IF a#="" THEN 450
460 IF a#="0" THEN y=y+1:DRAW x,y,f
470 IF a#="R" THEN x=x+1:DRAW x,y,f
480 IF a#="U" THEN y=y-1:DRAW x,y,f
490 IF a#="L" THEN x=x-1:DRAW x,y,f
500 IF a#="a" THEN f=0
510 IF a#="e" THEN f=1
520 IF a#="k" THEN 610
530 IF a#="d" THEN 730
540 IF a#="g" THEN 880
550 IF a#="c" THEN 1010
560 IF a#="s" THEN PEN 0:CLOSEOUT:END
570 LOCATE 35,25:PEN 1:PRINT x
580 LOCATE 35,1:PEN 1:PRINT y
590 IF jn#="j" THEN PRINT #9,a#
600 GOTO 430
610 REM Kreisumfang
620 IF nj#="n" THEN 640
630 INPUT #9,r,w:GOTO 670
640 PEN 0:LOCATE 1,1:INPUT r
650 LOCATE 1,1:INPUT w:PEN f
660 IF jn#="j" THEN PRINT #9,a#:PRINT #9,r,w

```

```

670 DEG
680 FOR i=1 TO w
690 PLOT x+r*COS(i),y+r*SIN(i),f
700 NEXT i
710 PLOT x,y,f
720 GOTO 430
730 REM Diagonale
740 IF nj#="n" THEN 760
750 INPUT #9,x1,y1:GOTO 790
760 PEN 0:LOCATE 1,1:INPUT x1
770 LOCATE 1,1:INPUT y1:PEN f
780 IF jn#="j" THEN PRINT #9,a#:PRINT #9,x1,y1
790 IF EOF=-1 AND nj#="j" THEN CLOSEIN:PEN 0:END
800 IF nj#="j" THEN INPUT #9,b#:GOTO 820
810 b#=INKEY#:IF b#="" THEN 810
820 IF b#="D" THEN x=x+x1:y=y+y1:DRAW x,y,f
830 LOCATE 35,25:PEN 1:PRINT x
840 LOCATE 35,1:PEN 1:PRINT y
850 IF jn#="j" THEN PRINT #9,b#
860 IF b#="S" THEN 430
870 GOTO 790
880 REM ausgefuellter Kreis
890 IF nj#="n" THEN 910
900 INPUT #9,r,w:GOTO 940
910 PEN 0:LOCATE 1,1:INPUT r
920 LOCATE 1,1:INPUT w:PEN f
930 IF jn#="j" THEN PRINT #9,a#:PRINT #9,r,w
940 DEG
950 FOR i=1 TO w
960 PLOT x,y,f
970 DRAW x+r*COS(i),y+r*SIN(i),f
980 NEXT i
990 PLOT x,y,f
1000 GOTO 430
1010 REM Farbaenderung
1020 IF nj#="n" THEN 1040
1030 INPUT #9,fa#:GOTO 1050
1040 PEN 0:LOCATE 1,1:INPUT fa#:PEN 1
1050 IF fa#="b" THEN f=3
1060 IF fa#="r" THEN f=2
1070 IF fa#="s" THEN f=1
1080 IF jn#="j" THEN PRINT #9,a#:PRINT #9,fa#
1090 GOTO 430
1100 REM UP Warten
1110 LOCATE 7,25
1120 PRINT"Bitte eine Taste druecken"
1130 x#=INKEY#:IF x#="" THEN 1130
1140 CLS:RETURN

```

Variablenliste:

a\$ = String zur Steuerung der Zeichenbewegungen

b\$ = String zur Diagonalen- bzw. Linienzeichnung

f = Farbe

fa\$ = String zur Änderung der Farbe

i = Laufindex

jn\$ = Antwortstring bezüglich der Frage, ob eine
Zeichnung abgespeichert werden soll

nj\$ = Antwortstring bezüglich der Frage, ob eine
Zeichnung eingelesen werden soll

r = Kreisradius

w = Winkelgröße

x,y = Koordinaten des "Zeichenbrettes"

x1,y1 = Koordinaten zum Zeichnen einer Diagonalen bzw.
Linie

Programmbeschreibung:

Satz 10- 90 : Überschrift

Satz 100 : Darstellung der Randfläche des Bildschirms
in leuchtendweiß

Satz 110-140 : Die PEN- bzw. PAPER-Nummern 0 bis 3 werden
mit den Farben leuchtendweiß (Satz 110),
schwarz (Satz 120), hellrot (Satz 130) und
himmelblau (Satz 140) belegt

Satz 150-160 : Abfrage, ob eine Zeichnung eingelesen wer-
den soll

Satz 170 : Ausgabe zweier Leerzeilen und Sprung zum
Satz 220, wenn mit der Eingabe von "n" ge-
antwortet wurde

Stz 180 : Ist nicht mit "j" geantwortet worden, wird
zum Satz 150 verzweigt

Satz 190 : Dieser Satz wird erreicht, wenn mit "j" ge-
antwortet wurde. Es werden zwei Leerzeilen
ausgegeben und es erfolgt eine Anweisung
zur Benutzung des Datacorders (wie bei den
vorangegangenen Programmen ist auch hier
prinzipiell die Benutzung eines Disketten-
laufwerks möglich).

Satz 200 : Die Eingabedatei "!Zeichnung" wird eröffnet

Satz 210 : Die Ausgangskordinaten x und y für den
Graphikcursor werden vom Band gelesen und
es wird zum Satz 340 verzweigt

Satz 220-250 : Die Werte für die Koordinaten x und y werden angefordert

Satz 260-290 : Es wird gefragt, ob die noch zu erstellende Zeichnung abgespeichert werden soll. Bei der Eingabe von "n" wird zum Satz 340 verzweigt und bei der Eingabe von "j" wird das Programm ab Satz 300 fortgesetzt.

Satz 300-330 : Die Ausgabedatei "!Zeichnung" wird eröffnet, es wird eine entsprechende Meldung ausgegeben und die Werte der Variablen x und y werden auf Cassette gespeichert.

Satz 340 : Verzweigung zum Unterprogramm "Warten"

Satz 350-400 : Zur leichteren Handhabung des Programms erhalten folgende Tasten andere Zuordnungen: die ↑ -Taste (Satz 350) den Wert O (wie oben), die → -Taste (Satz 360) den Wert R (wie rechts), die ↓ -Taste (Satz 370) den Wert U (wie unten), die ← -Taste (Satz 380) den Wert L (wie links), die COPY-Taste (Satz 390) den Wert D (wie draw) und die kleine ENTER-Taste (Satz 400) den Wert S (wie stop).

Satz 410 : Der Ausgangspunkt des Graphikcursors wird gezeichnet

Satz 420 : Die Variable f wird auf 1 gesetzt, d. h. die Farbe des Stiftes (genauer Graphikcursors) ist bis auf weiteres schwarz

Satz 430 : Ist nj\$ gleich "j" (d. h. es wird eine Zeichnung eingelesen) und ist EOF=-1 (EOF ist nur dann gleich -1, wenn das Ende der

Eingabedatei erreicht ist), wird die Eingabedatei geschlossen und das Programmende ist erreicht. Die Anweisung PEN 0 führt dazu, daß der sonst nach Programmende auf dem Bildschirm erscheinende Ausdruck READY sozusagen farblos (für PEN wird die gleiche Farbe wie für PAPER benutzt) erfolgt, um das Bild der fertigen Zeichnung nicht zu beeinträchtigen.

- Satz 440 : Ist nj\$ gleich "j", aber EOF<>-1, wird für die Variable a\$ ein neuer Wert eingelesen und es wird zum Satz 460 verzeigt.
- Satz 450 : Wird keine Zeichnung eingelesen, so müssen die Werte für die Steuervariable a\$ durch Tastendruck eingegeben werden.
- Satz 460-490 : In Abhängigkeit von dem Zeichen für a\$ wird in der aktuellen Farbe (Variable f) eine Linie von der augenblicklichen Cursorposition zur angegebenen x,y Position gezeichnet.
- Satz 500-510 : Ist a\$ gleich a (a wie Anfang), so erhält der Graphikcursor die gleiche Farbe wie der Hintergrund, d. h. es können bestehende Linien "ausradiert" werden. Dies geschieht solange, bis a\$ gleich e (e wie Ende) ist.
- Satz 520 : Ist a\$ gleich k, erfolgt ein Sprung zum Programmteil "Kreisumfang" (Satz 610). Dort wird dann ein nicht ausgefüllter Kreis gezeichnet.
- Satz 530-550 : Entsprechende Programmverzweigungen erfolgen, wenn a\$ gleich g ("ausgefüllter Kreis"), gleich d ("Diagonale") oder gleich c ("Farbänderung") ist.

Satz 560 : Das Programmende (für den Fall, daß keine Daten eingelesen werden) ist erreicht, wenn für a\$ ein s eingegeben wird. Die Meldung wird farblos ausgegeben (vgl. Satz 430). Die Anweisung CLOSEOUT schließt die Ausgabedatei. Falls die Zeichnung nicht abgespeichert werden sollte, ist diese Anweisung für den Programmablauf nicht störend.

Satz 570-580 : Während des Zeichenvorgangs wird der aktuelle x-Wert des Graphikcursors in der rechten unteren Ecke - und der aktuelle y-Wert in der linken oberen Ecke angezeigt.

Satz 590 : Das Zeichen für a\$ wird auf Band geschrieben, wenn die Zeichnung gespeichert werden soll

Satz 600 : Zum nochmaligen Durchlauf des Programmteils Satz 430-590 erfolgt eine entsprechende Verzweigung

Satz 610-720 : Ein Kreis wird gezeichnet

610 : Kommentar

620-630 : Ist nj\$ gleich j, so werden die Werte für r (Radius) und w (Winkel) vom Band gelesen und es wird zum Satz 670 verzweigt

640-650 : Farblose (Pen 0) Eingabe der Werte für r und w in der linken oberen Ecke des Bildschirms

660 : Falls dies gewünscht wurde, werden die Werte für r und w gespeichert. Das Zeichen für a\$ wird ebenfalls an dieser Stelle gespeichert, da in Satz 720 ein Sprung zum Satz 430 erfolgt (der

Satz 590 ist im Programmablauf aufgrund der Verzweigung in Satz 520 nicht erreicht worden).

670 : Umschaltung auf Winkelgradmaß

680-700 : In Abhängigkeit von r und w wird ein Kreis gezeichnet, genauer: die x,y-Koordinaten eines jeden Punktes auf dem Kreis werden geplottet

710 : Der Graphikcursor wird wieder auf den Kreismittelpunkt gesetzt

720 : Sprung zum Satz 430 zur Eingabe eines weiteren Steuerzeichens

Satz 730-870 : Eine Diagonale bzw. eine beliebige Linie wird gezeichnet

730-780 : Kommentar, Abfragen und Eingaben analog zu den Sätzen 610-660

790 : EOF Abfrage (vgl. Satz 430)

800-810 : Die Zeichen für b\$ werden eingelesen oder eingegeben (durch Betätigen der COPY-Taste)

820 : In Abhängigkeit von den Werten für x1 und y1 wird eine Linie gezeichnet, solange b\$ gleich D ist

830-840 : Die Koordinaten des Graphikcursors werden angezeigt

850 : Die Zeichen für b\$ werden bei Bedarf gesichert

860-870 : Ist die kleine ENTER-Taste gedrückt worden (bzw. ein "S" für b\$ eingelesen worden), erfolgt ein Sprung zum Satz 430, ansonsten zur EOF Abfrage in Satz 790

Satz 880-1000 : Jeweils vom Kreismittelpunkt aus (Satz 960), wird hier eine Linie zu jedem Punkt auf dem Kreisumfang gezeichnet (Satz 970), um einen ausgefüllten Kreis darzustellen. Ansonsten ist dieser Programmteil identisch mit den Sätzen 610-720.

Satz 1010-1090 : Programmteil "Farbänderung"

1010-1040 : Eingabe bzw. Einlesen des Zeichens für fa\$

1050-1070 : Bei der Eingabe (bzw. dem Einlesen) von "b" wird der Zeichenstift blau, bei "r" rot und bei "s" schwarz

1080-1090 : Schreiben (falls jn\$ gleich j) des Zeichens für fa\$ auf Band und Sprung zum Satz 430

Satz 1100-1140 : Unterprogramm "Warten"


Beispiel zur Programmbenutzung:

Nach dem Programmstart werden zuerst die Ausgangskordinaten mit den Werten 120 (für x) und 200 (für y) belegt. Daran anschließend kann nun eine Zeichnung erstellt werden:

Tastatureingabe	Erläuterung
→ (bis x=170)	Linie nach rechts
a	farbloser Stift "Anfang"
d	Diagonale
150 (ENTER)	150 Einheiten nach rechts
200 (ENTER)	200 Einheiten nach oben
1 mal COPY	zeichnen
kl. ENTER	Diagonale "Ende"
e	farbloser Stift "Ende"
↓ (bis y=350)	Linie nach unten
a	farbloser Stift "Anfang"
d	Diagonale
200 (ENTER)	200 Einheiten nach rechts
-150 (ENTER)	150 Einheiten nach unten
1 mal COPY	zeichnen
kl. ENTER	Diagonale "Ende"
e	farbloser Stift "Ende"
← (bis x=470)	Linie nach links
a	farbloser Stift "Anfang"

Tastatureingabe	Erläuterung
d -150 (ENTER) -200 (ENTER) 1 mal COPY kl. ENTER	Diagonale 150 Einheiten nach links 200 Einheiten nach unten zeichnen Diagonale "Ende"
e	farbloser Stift "Ende"
↑ (bis y=50)	Linie nach oben
a	farbloser Stift "Anfang"
↑ (bis y=200)	Linie nach oben
e	farbloser Stift "Ende"
c r (ENTER)	Farbänderung rot
k 150 (ENTER) 360 (ENTER)	Kreis Radius = 150 Einheiten Winkel = 360 Grad
c b (ENTER)	Farbänderung blau
g 80 (ENTER) 360 (ENTER)	ausgefüllter Kreis Radius = 80 Einheiten Winkel = 360 Grad
s	Programmende

Die entstandene Zeichnung ist sicherlich kein Kunstwerk.
Zudem ist nicht eindeutig feststellbar, was denn nun eigentlich dargestellt werden soll. Lassen Sie also Ihren Interpretationen freien Lauf ...



8.3. Wir machen Musik

"Endlich" werden die einen, "um Gottes willen" werden die anderen sagen, denn in diesem Kapitel geht es um programmierte Musik oder sagen wir besser um programmierte Klänge.

Die Musikprogrammierung ist unter Computer-Anwendern eine umstrittene Angelegenheit. Sollte man das Musizieren nicht dem Menschen überlassen, der darin seine Gefühle und seine Vitalität zum Ausdruck bringen kann? Diese Frage soll hier bewußt unbeantwortet bleiben. Es sei nur daran erinnert, daß viele Pop-Gruppen heutzutage ausgiebigst mit Computern arbeiten.

Auf jeden Fall ist die Musikprogrammierung ein Kapitel für sich. Und Sie müssen natürlich erst einmal Musikkenntnisse haben, um Musik programmieren zu können, sonst bleibt es bei einzelnen Klängen oder gar Geräuschen.

Bei Geräuschen muß es für Sie als CPC-Besitzer in der Tat nicht bleiben, da die Musik- und Klangeigenschaften des CPC ausgezeichnet sind. So können Sie - mit etwas Übung - einzelne Instrumente nachbilden und die programmierte Melodie über Ihre Hifi-Lautsprecher wiedergeben.

Der Umgang mit den einzelnen Musik-Anweisungen ist jedoch nicht ganz so einfach. Daher ist das Programm in diesem Kapitel auch mehr ein Übungsprogramm, um Sie mit den Musik-Anweisungen vertrauter zu machen. Sie können das Programm aber auch nutzen, um mit wenig Aufwand (im Vergleich

zur direkten Erstellung eines Musik-Programms) eine Melodie zu erzeugen. Falls Sie noch keine Erfahrung auf dem Gebiet der Musik-Programmierung haben, sollten Sie Ihre Ansprüche ein wenig herunterschrauben und sich mit einzelnen Klängen und Geräuschen zufrieden geben. Die Programmierung einer Melodie folgt dann anschließend.

Zu dem Programm ist noch zu sagen, daß es nur auf den drei grundlegenden Musik-Anweisungen SOUND (zur Erzeugung eines Tons), ENV (zur Definition einer Lautstärken-Hüllkurve) und ENT (zur Definition einer Ton-Hüllkurve) beruht. Wollen Sie die weiteren Musik-Funktionen und -Kommandos benutzen, so müssen Sie das vorliegende Programm entweder erweitern oder ein eigenständiges Musik-Programm entwickeln.

Ansonsten ist das Programm weitestgehendst selbsterklärend. Die folgenden Funktionen bietet es dem Benutzer:

Musik-Anweisungen

- eingeben (nur die einzelnen Parameter)
- einlesen
- abspeichern
- verbessern (mit der Möglichkeit des Ansehens der vorhandenen Anweisungen und deren Löschung)
- abspielen

Die Eingaben zu den fünf Abschnitten der Hüllkurven-Anweisungen sind durch Kommata voneinander zu trennen, also zum Beispiel:

	Schrittzahl, -größe, Pausenlänge
Ab- schnitt 1	5,5,2 (ENTER)

Erwähnt sei, daß Sie dieses Programm noch vereinfachen können. Und zwar können Sie einmal immer dann, wenn sich Programmteile wiederholen auf die Unterprogrammtechnik zurückgreifen und zum anderen können Sie die Variablen p1(i) ... p5(i) usw. usf. auch durch doppelt indizierte Variablen nach dem Muster p(j,i) für j = 1 ... 5 darstellen. Darauf ist hier aus Gründen der Übersichtlichkeit verzichtet worden.

Es ist ebenfalls darauf verzichtet worden, nicht erlaubte Eingabewerte abzufangen. In Ihrem Bedienungshandbuch sind die Eingabemöglichkeiten jedoch im einzelnen aufgeführt.

Zum Schluß noch ein Tip: beim ersten Abspielen stellen Sie den Lautstärkeregler am besten etwas leiser ...

```
10 REM I20
20 CLS
30 PRINT TAB(7)"I20 - Musikprogramm"
40 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
50 PRINT TAB(9) "Bernd Kowal, 1985"
60 GOSUB 3230
70 DIM ks(500),tp(500),d(500),l(500),lh(500)
80 DIM th(500),gp(500),n1(15),p1(15),q1(15),r1(15)
90 DIM p2(15),q2(15),r2(15),p3(15),q3(15),r3(15)
100 DIM p4(15),q4(15),r4(15),p5(15),q5(15),r5(15)
110 DIM n2(15),t1(15),v1(15),w1(15),t2(15),v2(15)
120 DIM w2(15),t3(15),v3(15),w3(15),t4(15)
130 DIM v4(15),w4(15),t5(15),v5(15),w5(15)
140 REM
150 REM Menue
160 REM
170 PRINT TAB(17)"Menue":PRINT:PRINT
180 PRINT TAB(33)"Eingabe":PRINT
190 PRINT"Musik-Anweisungen":PRINT
200 PRINT" eingeben";TAB(36)"1":PRINT
210 PRINT" einlesen";TAB(36)"2":PRINT
220 PRINT" verbessern";TAB(36)"3":PRINT
230 PRINT" abspielen";TAB(36)"4":PRINT
240 PRINT" abspeichern";TAB(36)"5":PRINT
```

```

250 PRINT"Arbeit beenden";TAB(36)"6":PRINT:PRINT:PRINT
260 INPUT"Ihre Wahl ";a:CLS
270 IF a<1 OR a>6 THEN CLS:GOTO 170
280 ON a GOTO 300,990,1300,2610,2820,3100
290 REM
300 REM Musikanweisungen eingeben
310 REM
320 PRINT TAB(17)"Menue":PRINT:PRINT
330 PRINT TAB(33)"Eingabe":PRINT
340 PRINT"Eingabe von":PRINT
350 PRINT" SOUND-Anweisungen";TAB(36)"1":PRINT
360 PRINT" ENV-Anweisungen";TAB(36)"2":PRINT
370 PRINT" ENT-Anweisungen";TAB(36)"3":PRINT
380 PRINT"Eingabe beenden";TAB(36)"4":PRINT:PRINT
390 PRINT:INPUT"Ihre Wahl ";a:CLS
400 ON a GOTO 420,570,770,970
410 IF a<1 OR a>4 THEN 320
420 so=so+1
430 PRINT"Geben Sie eine 0 fuer 'Kanal-Status'"
440 PRINT"ein, so wird die Eingabe beendet."
450 PRINT:PRINT
460 PRINT"Die";so;".te SOUND-Anweisung ":PRINT
470 INPUT"Kanal-Status ";ks(so)
480 IF ks(so)=0 THEN so=so-1:CLS:GOTO 320
490 INPUT"Ton-Periode ";tp(so)
500 INPUT"Dauer ";d(so)
510 INPUT"Lautstaerke ";l(so)
520 INPUT"Lautstaerken-Huellkurve ";lh(so)
530 INPUT"Ton-Huellkurve ";th(so)
540 INPUT"Geraeusch-Periode ";gp(so)
550 IF ver=1 THEN 1720
560 CLS:GOTO 420
570 ev=ev+1
580 PRINT"Geben Sie eine 0 fuer 'Nummer der"
590 PRINT"Huellkurve' ein, so wird die Eingabe"
600 PRINT"beendet.":PRINT:PRINT
610 PRINT"Die";ev;".te ENV-Anweisung ":PRINT
620 INPUT"Nummer der Huellkurve ";n1(ev):PRINT
630 IF n1(ev)=0 THEN ev=ev-1:CLS:GOTO 320
640 PRINT"Geben Sie (durch Komma getrennt) fuer"
650 PRINT"die 5 Abschnitte die Schrittzahl,die"
660 PRINT"Schrittgroesse und die Pausenlaenge ein."
670 PRINT" Schrittzahl,-groesse,Pausenlaenge"
680 PRINT"Ab-"
690 PRINT"schnitt"
700 PRINT TAB(4) 1;" ";:INPUT p1(ev),q1(ev),r1(ev)
710 PRINT TAB(4) 2;" ";:INPUT p2(ev),q2(ev),r2(ev)

```

```

720 PRINT TAB(4) 3;"          ";;INPUT p3(ev),q3(ev),r3(ev)
730 PRINT TAB(4) 4;"          ";;INPUT p4(ev),q4(ev),r4(ev)
740 PRINT TAB(4) 5;"          ";;INPUT p5(ev),q5(ev),r5(ev)
750 IF ver=1 THEN 2140
760 CLS:GOTO 570
770 et=et+1
780 PRINT"Geben Sie eine 0 fuer 'Nummer der"
790 PRINT"Huellkurve' ein, so wird die Eingabe"
800 PRINT"beendet.":PRINT:PRINT
810 PRINT"Die";et;"-te ENT-Anweisung "":PRINT
820 INPUT"Huellkurven-Nummer ";n2(et):PRINT
830 IF n2(et)=0 THEN et=et-1:CLS:GOTO 320
840 PRINT"Geben (Sie durch Komma getrennt) fuer"
850 PRINT"die 5 Abschnitte die Schrittzahl,die"
860 PRINT"Schrittgroesse und die Pausenlaenge ein."
870 PRINT"          Schrittzahl,-groesse,Pausenlaenge"
880 PRINT"Ab-"
890 PRINT"schnitt"
900 PRINT TAB(4) 1;"          ";;INPUT t1(et),v1(et),w1(et)
910 PRINT TAB(4) 2;"          ";;INPUT t2(et),v2(et),w2(et)
920 PRINT TAB(4) 3;"          ";;INPUT t3(et),v3(et),w3(et)
930 PRINT TAB(4) 4;"          ";;INPUT t4(et),v4(et),w4(et)
940 PRINT TAB(4) 5;"          ";;INPUT t5(et),v5(et),w5(et)
950 IF ver=1 THEN 2560
960 CLS:GOTO 770
970 CLS:GOTO 170
980 REM
990 REM Musikanweisungen einlesen
1000 REM
1010 k1=so+1:k2=ev+i:k3=et+1
1020 OPENIN ""
1030 INPUT #9,k4,k5,k6
1040 k4=k4+so:k5=k5+ev:k6=k6+et
1050 FOR i=k1 TO k4
1060 INPUT #9,ks(i),tp(i),d(i),l(i),lh(i),th(i),gp(i)
1070 so=so+1
1080 NEXT i
1090 FOR i=k2 TO k5
1100 INPUT #9,n1(i)
1110 INPUT #9,p1(i),q1(i),r1(i)
1120 INPUT #9,p2(i),q2(i),r2(i)
1130 INPUT #9,p3(i),q3(i),r3(i)
1140 INPUT #9,p4(i),q4(i),r4(i)
1150 INPUT #9,p5(i),q5(i),r5(i)
1160 ev=ev+1
1170 NEXT i
1180 FOR i=k3 TO k6

```

```

1190 INPUT #9,n2(i)
1200 INPUT #9,t1(i),v1(i),w1(i)
1210 INPUT #9,t2(i),v2(i),w2(i)
1220 INPUT #9,t3(i),v3(i),w3(i)
1230 INPUT #9,t4(i),v4(i),w4(i)
1240 INPUT #9,t5(i),v5(i),w5(i)
1250 et=et+1
1260 NEXT i
1270 CLOSEIN
1280 CLS:GOTO 170
1290 REM
1300 REM Musikanweisungen verbessern
1310 REM
1320 PRINT TAB(17)"Menue":PRINT:PRINT
1330 PRINT TAB(33)"Eingabe":PRINT
1340 PRINT"Verbesserungen von":PRINT
1350 PRINT" SOUND-Anweisungen";TAB(36)"1":PRINT
1360 PRINT" ENV-Anweisungen";TAB(36)"2":PRINT
1370 PRINT" ENT-Anweisungen";TAB(36)"3":PRINT
1380 PRINT"Verbesserungen beenden";TAB(36)"4"
1390 PRINT:PRINT:PRINT
1400 INPUT"Ihre Wahl ";a:CLS
1410 ON a GOTO 1430,1750,2170,2590
1420 IF a<1 OR a>4 THEN 1320
1430 FOR i=1 TO so
1440 PRINT"Die";i;".te SOUND-Anweisung lautet ":PRINT
1450 PRINT"Kanal-Status           ";ks(i)
1460 PRINT"Ton-Periode           ";tp(i)
1470 PRINT"Dauer                 ";d(i)
1480 PRINT"Lautstaerke           ";l(i)
1490 PRINT"Lautstaerken-Huellkurve ";lh(i)
1500 PRINT"Ton-Huellkurve         ";th(i)
1510 PRINT"Ger aeusch-Periode     ";gp(i)
1520 PRINT:PRINT:PRINT"Solll diese Anweisung"
1530 PRINT:PRINT TAB(33)"Eingabe":PRINT
1540 PRINT"verbessert";TAB(36)"1"
1550 PRINT"oder "
1560 PRINT"geloescht";TAB(36)"2"
1570 PRINT"werden"
1580 PRINT"oder so bleiben ?";TAB(36)"3":PRINT:PRINT
1590 INPUT"Ihre Wahl ";a
1600 IF a<1 OR a>3 THEN 1520
1610 ON a GOTO 1690,1620,1730
1620 PRINT:PRINT:PRINT"Jetzt wird geloescht !"
1630 FOR j=i TO so-1
1640 ks(j)=ks(j+1):tp(j)=tp(j+1):d(j)=d(j+1)
1650 l(j)=l(j+1):lh(j)=lh(j+1)

```

```

1660 th(j)=th(j+1):gp(j)=gp(j+1)
1670 NEXT j
1680 so=so-1:CLS:GOTO 1320
1690 CLS:PRINT"Dann geben Sie jetzt die"
1700 PRINT"neuen Werte ein ":PRINT
1710 ver=i:h=so:so=i:GOTO 470
1720 ver=0:so=h
1730 CLS:NEXT i
1740 CLS:GOTO 1320
1750 FOR i=1 TO ev
1760 PRINT"Die";i;".te ENV-Anweisung lautet ":PRINT
1770 PRINT"Nummer der Hueilkurve ":n1(i):PRINT
1780 PRINT"      Schrittzahl,-groesse,Pausenlaenge"
1790 PRINT"Ab-"
1800 PRINT"schnitt"
1810 PRINT TAB(4);i;TAB(11)p1(i);TAB(21)q1(i);
1820 PRINT TAB(31)r1(i)
1830 PRINT TAB(4);2;TAB(11)p2(i);TAB(21)q2(i);
1840 PRINT TAB(31)r2(i)
1850 PRINT TAB(4);3;TAB(11) p3(i);TAB(21)q3(i);
1860 PRINT TAB(31)r3(i)
1870 PRINT TAB(4);4;TAB(11) p4(i);TAB(21)q4(i);
1880 PRINT TAB(31)r4(i)
1890 PRINT TAB(4);5;TAB(11) p5(i);TAB(21)q5(i);
1900 PRINT TAB(31)r5(i)
1910 PRINT:PRINT:PRINT"Soll diese Anweisung"
1920 PRINT:PRINT TAB(33)"Eingabe":PRINT
1930 PRINT"verbessert";TAB(36)"1"
1940 PRINT"oder "
1950 PRINT"geloescht";TAB(36)"2"
1960 PRINT"werden"
1970 PRINT"oder so bleiben ?";TAB(36)"3":PRINT
1980 INPUT"Ihre Wahl ";a
1990 IF a<1 OR a>3 THEN 1910
2000 ON a GOTO 2110,2010,2150
2010 PRINT:PRINT:PRINT"Jetzt wird geloescht !"
2020 FOR j=i TO ev-1
2030 n1(j)=n1(j+1)
2040 p1(j)=p1(j+1):q1(j)=q1(j+1):r1(j)=r1(j+1)
2050 p2(j)=p2(j+1):q2(j)=q2(j+1):r2(j)=r2(j+1)
2060 p3(j)=p3(j+1):q3(j)=q3(j+1):r3(j)=r3(j+1)
2070 p4(j)=p4(j+1):q4(j)=q4(j+1):r4(j)=r4(j+1)
2080 p5(j)=p5(j+1):q5(j)=q5(j+1):r5(j)=r5(j+1)
2090 NEXT j
2100 ev=ev-1:CLS:GOTO 1320
2110 CLS:PRINT"Dann geben Sie jetzt die"
2120 PRINT"neuen Werte ein ":PRINT

```

```

2130 ver=1:h=ev:ev=i:GOTO 620
2140 ver=0:ev=h
2150 CLS:NEXT i
2160 CLS:GOTO 1320
2170 FOR i=1 TO et
2180 PRINT"Die";i;".te ENT-Anweisung lautet :":PRINT
2190 PRINT"Huellkurvennummer :";n2(i):PRINT
2200 PRINT"          Schrittzahl,-groesse,Pausenlaenge"
2210 PRINT"Ab-"
2220 PRINT"schnitt"
2230 PRINT TAB(4)1;TAB(11)t1(i);TAB(21)v1(i);
2240 PRINT TAB(31)w1(i)
2250 PRINT TAB(4)2;TAB(11)t2(i);TAB(21)v2(i);
2260 PRINT TAB(31)w2(i)
2270 PRINT TAB(4) 3;TAB(11) t3(i);TAB(21)v3(i);
2280 PRINT TAB(31)w3(i)
2290 PRINT TAB(4) 4;TAB(11) t4(i);TAB(21)v4(i);
2300 PRINT TAB(31)w4(i)
2310 PRINT TAB(4) 5;TAB(11) t5(i);TAB(21)v5(i);
2320 PRINT TAB(31)w5(i)
2330 PRINT:PRINT:PRINT"Soll diese Anweisung"
2340 PRINT:PRINT TAB(33)"Eingabe":PRINT
2350 PRINT"verbessert";TAB(36)"1"
2360 PRINT"oder"
2370 PRINT"geloescht";TAB(36)"2"
2380 PRINT"werden"
2390 PRINT"oder so bleiben?";TAB(36)"3":PRINT
2400 INPUT"Ihre Wahl ";a
2410 IF a<1 OR a>3 THEN 2330
2420 ON a GOTO 2530,2430,2570
2430 PRINT:PRINT:PRINT"Jetzt wird geloescht !"
2440 FOR j=i TO et-1
2450 n2(j)=n2(j+1)
2460 t1(j)=t1(j+1):v1(j)=v1(j+1):w1(j)=w1(j+1)
2470 t2(j)=t2(j+1):v2(j)=v2(j+1):w2(j)=w2(j+1)
2480 t3(j)=t3(j+1):v3(j)=v3(j+1):w3(j)=w3(j+1)
2490 t4(j)=t4(j+1):v4(j)=v4(j+1):w4(j)=w4(j+1)
2500 t5(j)=t5(j+1):v5(j)=v5(j+1):w5(j)=w5(j+1)
2510 NEXT j
2520 et=et-1:CLS:GOTO 1320
2530 CLS:PRINT"Dann geben Sie jetzt die"
2540 PRINT"neuen Werte ein :":PRINT
2550 ver=1:h=et:et=i:GOTO 820
2560 ver=0:et=h
2570 CLS:NEXT i
2580 CLS:GOTO 1320
2590 CLS:GOTO 170

```

```

2600 REM
2610 REM Musikstueck abspielen
2620 REM
2630 FOR i=1 TO ev
2640 p1=p1(i):q1=q1(i):r1=r1(i):p2=p2(i):q2=q2(i)
2650 r2=r2(i):p3=p3(i):q3=q3(i):r3=r3(i):p4=p4(i)
2660 q4=q4(i):r4=r4(i):p5=p5(i):q5=q5(i):r5=r5(i)
2670 n=n1(i)
2680 ENV n,p1,q1,r1,p2,q2,r2,p3,q3,r3,p4,q4,r4,p5,q5,r5
2690 NEXT i
2700 FOR i=1 TO et
2710 t1=t1(i):v1=v1(i):w1=w1(i):t2=t2(i):v2=v2(i)
2720 w2=w2(i):t3=t3(i):v3=v3(i):w3=w3(i):t4=t4(i)
2730 v4=v4(i):w4=w4(i):t5=t5(i):v5=v5(i):w5=w5(i)
2740 n=n2(i)
2750 ENT n,t1,v1,w1,t2,v2,w2,t3,v3,w3,t4,v4,w4,t5,v5,w5
2760 NEXT i
2770 FOR i=1 TO so
2780 SOUND ks(i),tp(i),d(i),l(i),lh(i),th(i),gp(i)
2790 NEXT i
2800 GOTO 170
2810 REM
2820 REM Musikanweisungen abspeichern
2830 REM
2840 PRINT"Wie soll das abzuspeichernde"
2850 INPUT"Musikstueck heissen ";n#:FRINT
2860 OPENOUT n#
2870 PRINT #9,so,ev,et
2880 FOR i=1 TO so
2890 PRINT #9,ks(i),tp(i),d(i),l(i),lh(i),th(i),gp(i)
2900 NEXT i
2910 FOR i=1 TO ev
2920 PRINT #9,n1(i)
2930 PRINT #9,p1(i),q1(i),r1(i)
2940 PRINT #9,p2(i),q2(i),r2(i)
2950 PRINT #9,p3(i),q3(i),r3(i)
2960 PRINT #9,p4(i),q4(i),r4(i)
2970 PRINT #9,p5(i),q5(i),r5(i)
2980 NEXT i
2990 FOR i=1 TO et
3000 PRINT #9,n2(i)
3010 PRINT #9,t1(i),v1(i),w1(i)
3020 PRINT #9,t2(i),v2(i),w2(i)
3030 PRINT #9,t3(i),v3(i),w3(i)
3040 PRINT #9,t4(i),v4(i),w4(i)
3050 PRINT #9,t5(i),v5(i),w5(i)
3060 NEXT i

```

```
3070 CLOSEOUT
3080 CLS:GOTO 170
3090 REM
3100 REM Programm beenden
3110 REM
3120 PRINT"Haben Sie auch alle Musikanweisungen"
3130 PRINT"ordnungsgemaess abgespeichert ?"
3140 PRINT"Falls nicht, geben Sie einfach"
3150 PRINT"ein 'oh' ein (ansonsten reicht die"
3160 PRINT"Leertaste)."
```

Variablenliste:

Die Parameter der SOUND-Anweisung:

ks(i) = Kanal-Status

tp(i) = Ton-Periode

d(i) = Dauer

l(i) = Lautstärke

lh(i) = Lautstärken-Hüllkurve

th(i) = Ton-Hüllkurve

gp(i) = Geräusch-Periode

Die Parameter der ENV-Anweisung:

n1(i) = Nummer der Hüllkurve

p1(i) ... p5(i) = Schrittzahl (Abschnitte 1 ... 5)

q1(i) ... q5(i) = Schrittgröße (Abschnitte 1 ... 5)

r1(i) ... r5(i) = Pausenlänge (Abschnitte 1 ... 5)

Die Parameter der ENT-Anweisung:

n2(i) = Nummer der Hüllkurve

t1(i) ... t5(i) = Schrittzahl (Abschnitte 1 ... 5)

v1(i) ... v5(i) = Schrittgröße (Abschnitte 1 ... 5)

w1(i) ... w5(i) = Pausenlänge (Abschnitte 1 ... 5)

Die übrigen Variablen:

a = Antwort "Menüauswahl"

a\$ = Antwortstring

et = Anzahl der ENT-Anweisungen

ev = Anzahl der ENV-Anweisungen

h = Hilfsgröße

i = Laufindex

j = Laufindex

k1 }
k2 } = Schleifenanfangswerte
k3 }

k4 }
k5 } = Schleifenendwerte
k6 }

n\$ = Name des Musikstücks

so = Anzahl der SOUND-Anweisungen

ver = Hilfsvariable (gleich 1, wenn Verbesserungen durchgeführt werden, ansonsten gleich 0)

Programmbeschreibung:

Satz 10- 60 : Überschrift

Satz 70- 130 : Speicherplatzreservierung für 500 SOUND-
Anweisungen und jeweils 15 ENV- bzw. ENT-
Anweisungen

Satz 140- 280 : Auswahlmenü

Satz 290- 970 : Musik-Anweisungen werden eingegeben
320- 410 : Menü
420- 560 : Die Parameter der Sound-An-
weisung
570- 760 : Die Parameter der ENV-Anwei-
sung
770- 960 : Die Parameter der ENT-Anwei-
sung
970 : Beendigung der Eingabe

Satz 980-1280 : Musik-Anweisungen werden eingelesen
Die Schleifenanfangs- und endwerte werden
so gewählt, daß auch ein Einlesen zusätz-
licher Anweisungen möglich ist

Satz 1290-2590 : Musik-Anweisungen werden verbessert
1320-1420 : Menü
1430-1740 : Die SOUND-Anweisungen werden
auf dem Bildschirm ausgege-
ben. Sie können verbessert
oder gelöscht werden (Menü
Satz 1520-1600). Zur Löschung
einer SOUND-Anweisung dient
die Programmschleife Satz
1630-1670. Im Falle einer
Verbesserung wird die Varia-

ble ver gleich 1 gesetzt und es wird zum Programmteil "eingeben" (Satz 470) verzweigt (Rücksprung Satz 550). Im Feld h wird der aktuelle Wert für so gespeichert, da die Variable so im Programmteil "eingeben" als Index dient.

1750-2580 : Das gleiche geschieht für die ENV- und ENT-Anweisungen


2590 : Ende dieses Programmteils

Satz 2600-2800 : Abspielen der Musik-Anweisungen
Zuerst werden die Hüllkurven-Anweisungen und dann die SOUND-Anweisungen ausgeführt. Die indizierten Parameter werden dazu jeweils ohne Index dargestellt, da sonst die Parametervariablen aufgrund zuvieler Zeichen die Fehlermeldung "improper argument" zur Folge hätten.

Satz 2810-3080 : Musik-Anweisungen werden abgespeichert

Satz 3090-3210 : Das Programm wird beendet
Es besteht hier jedoch noch die Möglichkeit Musik-Anweisungen abzuspeichern, falls dies vorher vergessen wurde.

Satz 3220-3290 : Unterprogramm "Warten"



8.4. Die Fußball-Bundesliga

Das Programm zu diesem Kapitel verwaltet die Ergebnisse der Fußball-Bundesliga und erstellt die entsprechende Tabelle. Mit wenigen Handgriffen können Sie dieses Programm auch für andere Sportarten umschreiben. Sie müssen dann lediglich die DATA-Anweisungen, in denen die einzelnen Mannschaften aufgeführt sind -, und evtl. die Rangfolgeberechnungen für die Tabellenerstellung ändern. Die DATA-Anweisungen sind natürlich auch dann zu ändern, wenn am Ende einer Saison Vereine in die Bundesliga auf- oder absteigen.

Für die Benutzung des Programms ist die folgende Schrittfolge empfehlenswert:

1. Schritt: Einlesen abgespeicherter Ergebnisse
2. Schritt: Eingabe neuer Ergebnisse
3. Schritt: Abspeichern der alten und der neuen Ergebnisse
4. Schritt: Erstellung der Tabelle

Zusätzlich können Sie sich jederzeit die entweder eingelesenen oder eingegebenen Ergebnisse ansehen. Dabei besteht dann die Möglichkeit, alle Ergebnisse einer einzigen Mannschaft auf dem Bildschirm auszugeben.

Erfolgt die Eingabe vor dem Einlesen, werden die eingegebenen Ergebnisse überschrieben. Dies läßt sich jedoch - falls dies gewünscht wird - durch eine Änderung des Programmteils "Ergebnisse einlesen" verhindern:

```
95 DIM esp(18),eg(18),eu(18),ev(18),et1(18),et2(18)
1050 REM Ergebnisse einlesen
1060 REM
1070 OPENIN "Ergebnisse"
1080 INPUT #9,eas
1090 FOR i=1 TO 18
1100 INPUT #9,esp(i),eg(i),eu(i),ev(i),et1(i),et2(i)
1105 sp(i)=sp(i)+esp(i):g(i)=g(i)+eg(i):u(i)=u(i)+eu(i)
1106 v(i)=v(i)+ev(i):t1(i)=t1(i)+et1(i):t2(i)
    =t2(i)+et2(i)
1110 NEXT i
1120 FOR i=as+1 TO as+eas
1130 INPUT #9,s1#(i)
1140 INPUT #9,s2#(i)
1150 INPUT #9,st#(i)
1160 NEXT i:CLOSEIN
1165 as=as+eas
1170 CLS:GOTO 220
```

In den Sätzen 1080 und 1100 werden neue Variablen gebildet (für die natürlich auch dimensioniert werden muß), deren Werte dann den für den weiteren Programmablauf relevanten absoluten Variablen (also plus vorheriger Eingabe) hinzugefügt werden. Weiterhin werden Anfangs- und Endwert der Schleife in Satz 1120 geändert, um ein Überschreiben schon vorhandener Werte zu vermeiden.

Ein vielleicht entscheidender Nachteil dieser Programmversion ist jedoch, daß die Programmfunktion "ansetzen" nun nicht mehr die Ergebnisse in der Reihenfolge ihrer Eingabe

wiedergibt. Insbesondere wenn Sie daran interessiert sind, die Ergebnisse einer speziellen Mannschaft in ihrer zeitlichen Folge zu beobachten bzw. Trends bezüglich der Leistungsstärke in den verschiedenen Saisonphasen zu erkennen, werden Sie vermutlich auf diese Programmversion verzichten.

Programm:

```
10 REM I21
20 CLS
30 PRINT "I21 - Programm zur Fussball-Bundesliga"
40 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
50 PRINT TAB(9) "Bernd Kowal, 1985"
60 GOSUB 1750
70 DIM bm$(18),k$(18),sp(18),g(18),u(18),v(18)
80 DIM t1(18),t2(18),p1(18),p2(18)
90 DIM s1$(306),s2$(306),st$(306)
100 DATA Bayern Muenchen,BM,Werder Bremen,WB
110 DATA Bor. M'gladbach,MB,Hamburger SV,HSV
120 DATA 1. FC Koeln,K,VfL Bochum,Bo
130 DATA Waldhof Mannheim,WM,Bayer Uerdingen,BU
140 DATA VfB Stuttgart,St,Eintr. Frankfurt,EF
150 DATA FC Schalke 04,Sch,Bayer Leverkusen,BL
160 DATA 1. FC Kaiserslautern,Kl,Borussia Dortmund,BD
170 DATA Fort. Duesseldorf,FD,Hannover 96,Ha
180 DATA 1. FC Nuernberg,N,1. FC Saarbruecken,Sa
190 FOR i=1 TO 18
200 READ bm$(i),k$(i)
210 NEXT i
220 PRINT TAB(10)"Menue":PRINT:PRINT
230 PRINT TAB(33)"Eingabe":PRINT:PRINT
240 PRINT"Ergebnisse":PRINT
250 PRINT" eingeben";TAB(36)"1":PRINT
260 PRINT" ansehen";TAB(36)"2":PRINT
270 PRINT" einlesen";TAB(36)"3":PRINT
280 PRINT" abspeichern (vor der"
290 PRINT" Tabellenerstellung)";TAB(36)"4":PRINT
300 PRINT" Tabelle erstellen und"
```

```

310 PRINT"Programm beenden";TAB(36)"5":PRINT
320 PRINT:INPUT"Ihre Wahl ";a:CLS
330 IF a<1 OR a>5 THEN 220
340 ON a GOTO 360,700,1040,1180,1320
350 REM
360 REM Ergebnisse eingeben
370 REM
380 WINDOW #1,17,40,1,25
390 PRINT #1,"Benutzen Sie bei der"
400 PRINT #1,"Eingabe die folgenden"
410 PRINT #1,"Abkuerzungen :":PRINT #1
420 FOR i=1 TO 18
430 PRINT #1,bm$(i);"=";k$(i)
440 NEXT i
450 WINDOW #2,1,16,1,25
460 PRINT #2,"Wieviele Spiel-"
470 PRINT #2,"ergebnisse sol-"
480 PRINT #2,"len eingegeben"
490 INPUT #2,"werden ";es:PRINT #2
500 FOR i=as+1 TO as+es
510 INPUT #2,"Heimteam ";ht$
520 INPUT #2,"Tore ";th
530 INPUT #2,"Gastteam ";gt$
540 INPUT #2,"Tore ";tg
550 FOR j=1 TO 18
560 IF ht$=k$(j) THEN j1=j
570 IF gt$=k$(j) THEN j2=j
580 NEXT j
590 sp(j1)=sp(j1)+1:sp(j2)=sp(j2)+1
600 IF th>tg THEN g(j1)=g(j1)+1:v(j2)=v(j2)+1
610 IF th=tg THEN u(j1)=u(j1)+1:u(j2)=u(j2)+1
620 IF th<tg THEN v(j1)=v(j1)+1:g(j2)=g(j2)+1
630 t1(j1)=t1(j1)+th:t2(j1)=t2(j1)+tg
640 t1(j2)=t1(j2)+tg:t2(j2)=t2(j2)+th
650 th$=STR$(th):tg$=STR$(tg)
660 s1$(i)=bm$(j1):s2$(i)=bm$(j2):st$(i)=th$+" "+tg$
670 PRINT #2,:NEXT i:as=as+es
680 CLS:GOTO 220
690 REM
700 REM Ergebnisse ansehen
710 REM
720 PRINT TAB(10)"Menue":PRINT:PRINT
730 PRINT TAB(33)"Eingabe":PRINT:PRINT
740 PRINT"Alle Ergebnisse ansehen";TAB(36)"1":PRINT
750 PRINT"Ergebnisse einer Mann-"
760 PRINT"schaft ansehen";TAB(36)"2":PRINT
770 PRINT"genug gesehen ";TAB(36)"3"
780 PRINT:PRINT:INPUT"Ihre Wahl ";a:MODE 2

```

```

790 IF a<1 OR a>3 THEN MODE 1:GOTO 720
800 l=0:ON a GOTO 810,860,1020
810 FOR i=1 TO as
820 PRINT s1$(i);"-";s2$(i);TAB(45)st$(i)
830 l=l+1:IF l=18 THEN l=0:GOSUB 1750
840 NEXT i
850 GOSUB 1750:MODE 1:GOTO 720
860 PRINT"Benutzen Sie fuer die Eingabe die"
870 PRINT"folgenden Abkuerzungen :":PRINT
880 FOR i=1 TO 18
890 PRINT bm$(i);"=";k$(i)
900 NEXT i:PRINT
910 INPUT"Welche Mannschaft ";m$:CLS
920 FOR i=1 TO 18
930 IF m#=k$(i) THEN m#=bm$(i)
940 NEXT i
950 FOR i=1 TO as
960 IF m#=s1$(i) OR m#=s2$(i) THEN 980
970 GOTO 1000
980 PRINT s1$(i);"-";s2$(i);TAB(45)st$(i)
990 l=l+1:IF l=18 THEN l=0:GOSUB 1750
1000 NEXT i
1010 GOSUB 1750:MODE 1:GOTO 720
1020 MODE 1:GOTO 220
1030 REM
1040 REM Ergebnisse einlesen
1050 REM
1060 OPENIN "Ergebn."
1070 INPUT #9,as
1080 FOR i=1 TO 18
1090 INPUT #9,sp(i),g(i),u(i),v(i),t1(i),t2(i)
1100 NEXT i
1110 FOR i=1 TO as
1120 INPUT #9,s1$(i)
1130 INPUT #9,s2$(i)
1140 INPUT #9,st$(i)
1150 NEXT i:CLOSEIN
1160 CLS:GOTO 220
1170 REM
1180 REM Ergebnisse abspeichern
1190 REM
1200 OPENOUT "Ergebn."
1210 PRINT #9,as
1220 FOR i=1 TO 18
1230 PRINT #9,sp(i),g(i),u(i),v(i),t1(i),t2(i)
1240 NEXT i
1250 FOR i=1 TO as

```

```

1260 PRINT #9,s1#(i)
1270 PRINT #9,s2#(i)
1280 PRINT #9,st#(i)
1290 NEXT i:CLOSEOUT
1300 CLS:GOTO 220
1310 REM
1320 REM Tabelle erstellen
1330 REM
1340 FOR i=1 TO 18
1350 p1(i)=g(i)*2+u(i)
1360 p2(i)=v(i)*2+u(i)
1370 NEXT i
1380 FOR i=1 TO 17
1390 FOR j=i+1 TO 18
1400 IF p1(i)<p1(j) THEN GOSUB 1650
1410 IF p1(i)>p1(j) THEN 1470
1420 IF p2(i)>p2(j) THEN GOSUB 1650
1430 IF p2(i)<p2(j) THEN 1470
1440 IF t1(i)-t2(i)<t1(j)-t2(j) THEN GOSUB 1650
1450 IF t1(i)-t2(i)>t1(j)-t2(j) THEN 1470
1460 IF t1(i)<t1(j) THEN GOSUB 1650
1470 NEXT j
1480 NEXT i
1490 MODE 2
1500 PRINT TAB(20)"Die Tabelle der Fussball-Bundesliga"
1510 PRINT:PRINT:PRINT TAB(30)
      "Sp.  g.  u.  v.      Tore      Punkte"
1520 FOR i=1 TO 18
1530 PRINT USING "##";i;:PRINT ". ";bm#(i);
1540 PRINT TAB(30) USING "##  ";sp(i);
1550 PRINT USING "##  ";g(i);u(i);v(i);
1560 PRINT USING "  ###  ";t1(i);
1570 PRINT":";:PRINT USING "###  ";t2(i);
1580 PRINT USING "  ##  ";p1(i);
1590 PRINT":";:PRINT USING "  ##";p2(i)
1600 NEXT i
1610 GOSUB 1750:MODE 1
1620 PRINT"Ende des Programms":END
1630 REM
1640 REM Unterprogramm Vertauschen
1650 h#=bm#(j):bm#(j)=bm#(i):bm#(i)=h#

```

```
1660 h=sp(j):sp(j)=sp(i):sp(i)=h
1670 h=g(j):g(j)=g(i):g(i)=h
1680 h=u(j):u(j)=u(i):u(i)=h
1690 h=v(j):v(j)=v(i):v(i)=h
1700 h=t1(j):t1(j)=t1(i):t1(i)=h
1710 h=t2(j):t2(j)=t2(i):t2(i)=h
1720 h=p1(j):p1(j)=p1(i):p1(i)=h
1730 h=p2(j):p2(j)=p2(i):p2(i)=h
1740 RETURN
1750 REM Unterprogramm Warten
1760 LOCATE 5,25
1770 PRINT"Bitte eine Taste druecken !"
1780 x#=INKEY#
1790 IF x#="" THEN 1780
1800 CLS:RETURN
```

Variablenliste:

a = Antwort "Menüauswahl"

as = Anzahl der Spielergebnisse

bm\$(i) = Bundesligamannschaften

es = Anzahl der eingegebenen Ergebnisse

g(i) = Zahl der gewonnenen Spiele einer Mannschaft

gt\$ = Gastteam

h = Hilfsvariable im Unterprogramm "Vertauschen"

h\$ = Hilfsvariable im Unterprogramm "Vertauschen"

ht\$ = Heimteam

i = Laufindex

j = Laufindex

j1 = Indexzahl der Heim-Mannschaft

j2 = Indexzahl der Gast-Mannschaft

k\$(i) = Bundesligamannschaften in Kurzform

l = Zählindex

m\$ = Mannschaft im Programmteil "Ergebnisse ansehen"

p1(i) = Pluspunkte einer Mannschaft

p2(i) = Minuspunkte einer Mannschaft

$s1(i)$
= die beteiligten Mannschaften eines Fußballspiels
 $s2(i)$

$sp(i)$ = Anzahl der Spiele einer Mannschaft

$st(i)$ = Ergebnis eines Fußballspiels

$t1(i)$ = geschossene Tore

$t2(i)$ = "kassierte" Tore

tg = Tore des Gastteams

$tg\$$ = tg als Stringvariable

th = Tore des Heimteams

$th\$$ = th als Stringvariable

$u(i)$ = Zahl der unentschieden ausgegangenen Spiele
einer Mannschaft

$v(i)$ = Zahl der verlorenen Spiele einer Mannschaft

Programmbeschreibung:

Satz 10- 60 : Überschrift

Satz 70- 90 : Speicherplatzreservierung für 18 Mannschaften und insgesamt 306 Spiele

Satz 100- 180 : Die Mannschaften der Fußball-Bundesliga werden dem Programm mit einer dazugehörigen Kurzschreibweise als Daten zur Verfügung gestellt

Satz 190- 210 : Einlesen der Daten

Satz 220- 340 : Menü und Sprunganweisungen

Satz 350- 680 : Eingabe der Fußballergebnisse

380- 440 : Ausgabe einer Erläuterung in einem definierten Textfenster

450- 670 : Eingabe der Ergebnisse im restlichen Bildschirmbereich.
In den Sätzen 550-580 werden die eingegebenen Mannschaftsnamen mit den Daten verglichen, um die Indizes der beiden am Spiel beteiligten Mannschaften zu ermitteln. Entsprechend dieser Indizes werden die für die Tabellerstellung relevanten Variablen aktualisiert (Satz 590-640) und das Fußballergebnis als solches wird in den dafür vorgesehenen Feldern abgelegt (Satz 650-660). Dabei zeigt sich in Satz 660 nochmals eine Möglichkeit der String-

bearbeitung.

680 : Rücksprung zur Menüauswahl

Satz 690-1020 : Ansehen der Ergebnisse

720- 800 : Menüauswahl und Programmver-
zweigungen

810- 850 : Ausgabe aller Ergebnisse

860-1010 : Ein Mannschaftsname wird an-
gefordert (Satz 910), die
Kurzschreibweise wird identi-
fiziert (Satz 920-940) und es
werden die Spielergebnisse
mit Beteiligung der angege-
benen Mannschaft ausgegeben
(Satz 950-1000)

1020 : Rücksprung zur Menüauswahl

Satz 1030-1160 : Ergebnisse werden eingelesen

Satz 1170-1300 : Ergebnisse werden abgespeichert

Satz 1310-1610 : Die Tabelle wird erstellt

1340-1370 : Die Plus- und Minuspunkte
werden für jede Mannschaft
berechnet

1380-1480 : Jede Mannschaft wird mit je-
der anderen Mannschaft ver-
glichen und die Rangfolge
wird gegebenenfalls im Unter-
programm (Satz 1650-1740)
vertauscht (siehe dazu auch
Kapitel 2.4 dieses Buches).
Als Vergleichswerte dienen
die Pluspunkte (Satz 1400-
1410), die Minuspunkte (Satz
1420-1430), die Tordifferenz
(Satz 1440-1450) und die Zahl
der geschossenen Tore (Satz
1460). Nur bei Gleichheit

eines überprüften Kriteriums
wird das jeweils folgende
Kriterium herangezogen.
1490-1610 : Die formatierte Tabellenaus-
gabe

Satz 1620 : Programmende

Satz 1630-1740 : Unterprogramm "Vertauschen"

Satz 1750-1800 : Unterprogramm "Warten"

8.5. Ideenfundgrube

- Gedichte vom Computer (in den meisten Fällen haben nämlich sich reimende Wörter in ihren hinteren Teilen gleiche Zeichenfolgen)
- abrufbare Blumen- oder Gartenpflege
- Rezeptkartei für Hobbyköche
- Sammlungen jeglicher Art ordnen und verwalten
- programmierte Lehrgänge (Segeln, Morsen usw.)
- Ergebnisauswertung von großen Sportveranstaltungen
- Ihr Rechner als Trainingsplaner
- Belichtungszeiten im Fotolabor (dazu eignet sich Ihr CPC mit seiner Echtzeituhr besonders gut)
- 2D- und 3D-Graphik
- Computer als Filmmacher
- CAD für Strickarbeiten
- Schnittmusterbögen programmiert

- Planung der Streckenführung Ihrer Modelleisen (oder -renn) bahn
- Computersteuerung Ihrer Modelleisenbahn (nur mit zusätzlichen Hardware-Kenntnissen)
- und noch vieles mehr


und so könnte ein einfaches Programm zur Blumenpflege aussehen.

```
10 REM Blumenpflege
20 k=4
30 REM k ist gleich der Anzahl der Blumen bzw. Pflanzen
40 DIM b$(k), w$(k), ls$(k)
50 DATA Ceropogia woodii (Leuchterblume)
51 DATA weing giessen
52 DATA hell und warm halten
60 DATA Helxine soleirolii (Bubikopf)
61 DATA hauptsaechlich von unten giessen
62 DATA Helxine hell und etwas warm halten
70 DATA Hedera (Efeu)
71 DATA wenig giessen
72 DATA kuehl und hell bis halbschaetig stellen
80 DATA Blattbegonie (Koenigsbegonie)
81 DATA nicht zu feucht halten
82 DATA halbschaetig stellen
480 REM Aufbau der DATA-Anweisungen
490 REM DATA Blume, Wasser, Licht bzw. Standort
500 FOR i=1 TO k
510 READ b$(i), w$(i), ls$(i)
520 NEXT i
530 CLS
540 PRINT "Zu den folgenden Blumen bzw. Pflanzen"
550 PRINT "koennen Sie Pflegeinformationen abrufen:"
560 FOR i=1 TO k
570 PRINT i;". ";b$(i)
580 NEXT i
590 PRINT:PRINT:INPUT"Geben Sie eine Nummer an ";n
600 IF n>k OR n<1 THEN 590
610 CLS
620 PRINT b$(n):PRINT:PRINT w$(n):
PRINT:PRINT ls$(n)
630 PRINT:PRINT:PRINT:PRINT:PRINT "Euer weitere
Informationen bitte das"
640 PRINT"Programm erneunt starten"
```

Dieses Programm ist ähnlich aufgebaut wie das Programm zur Kalorienberechnung (I18). Ohne großen Aufwand läßt es sich um weitere Pflanzenarten erweitern. Zudem ist natürlich auch eine andere Aufteilung der Pflegekriterien möglich.

Weiterhin sei noch ein kurzes Progrämmchen zum Thema "Stricken" vorgestellt. Neben der programmtechnisch etwas aufwendigen Möglichkeit, Pullover auf dem Bildschirm zu entwerfen und mittels einer Hardcopy-Routine Schnittmusterbögen auszudrucken, läßt sich eine "Maschenprobe" sehr leicht programmieren, da es sich dabei um ein einfaches Dreisatz-Problem handelt:

```
10 REM Maschenprobe
20 CLS
30 INPUT "Anzahl der Maschen auf 10cm Breite ";m
40 PRINT:PRINT:INPUT "Breite des Pollovers in cm ";b
50 PRINT:PRINT:PRINT:PRINT
60 PRINT "Sie benoetigen somit ";m*b/10;" Maschen."
```



9. Die lieben Kleinen

9.1. Vorbemerkung

In diesem Kapitel soll gezeigt werden, wie man als Schüler oder Lernender den Schneider CPC einsetzen und benutzen kann. Vielleicht haben Sie schulpflichtige Kinder, denen Sie mit den Programmen in diesem Kapitel eine Alternative zu den in Schülerkreisen weit verbreitete Spielprogrammen bieten können.

Natürlich ist es möglich zu den verschiedensten Problemen des Schulalltags hilfreiche Programme zu entwickeln. Hier stehen jedoch mathematische Probleme im Vordergrund, da das Fach Mathematik immer noch die größte "Nähe" zum Computer aufweist. Hinzu kommt ein etwas umfangreicheres Vorkabelprogramm, welches nicht zuletzt deshalb sehr interessant ist, weil es ohne große Schwierigkeiten für beliebige Fragestellungen umgeschrieben werden kann.

9.2. Das Einmaleins

Das Programm in diesem Kapitel eignet sich in besonderer Weise für Grundschüler. Es ist ohne Probleme auf eine andere Grundrechnungsart umzustellen, so daß ein wesentlicher Teil der elementaren Mathematik mit diesem Programm abgedeckt werden kann.

Ausgehend von einem einzugebenden Maximalwert werden im Programmablauf Zufallszahlen erzeugt, die der Stellung einer Multiplikationsaufgabe dienen. Diese Aufgabe sollte dann möglichst korrekt gelöst werden. Am Ende der Lernarbeit wird dann die Erfolgsquote ausgegeben. Insofern handelt es sich hier - ähnlich wie beim "Vokabeltrainer" - um ein Abfrageprogramm.

Programm:

```
10 REM I22
20 CLS
30 PRINT "I22 - Programm zur Einuebung"
40 PRINT TAB(7)"des Einmaleins"
50 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
60 PRINT TAB(9) "Bernd Kowal, 1985"
70 GOSUB 350
80 PRINT"Durch die Eingabe des Buchstabens 'z'"
90 PRINT"koennen Sie die folgnde Uebung abbrechen"
100 PRINT:PRINT"Zunaechst geben Sie aber bitte an,"
110 PRINT"wie gross die groesste Zahl der Uebung"
120 INPUT"sein soll ";gz:CLS
130 i=i+1
140 PRINT"Die";i;". te Aufgabe ":":PRINT:PRINT:PRINT
150 f1=ROUND(RND*gz):f2=ROUND(RND*gz)
```

```

160 PRINT TAB(10) f1;"mal";f2;:INPUT" = ",ep#
170 IF ep#="z" THEN 250
180 ep=VAL(ep#)
190 p=f1*f2:PRINT:PRINT
200 IF ep<>p THEN 220
210 PRINT TAB(11)"richtig":ra=ra+1:GOTO 240
220 PRINT TAB(11)"falsch":fa=fa+1:PRINT:PRINT
230 PRINT TAB(10) f1;"mal";f2;" = ";p
240 GOSUB 350:GOTO 130
250 CLS:i=i-1
260 PRINT" Von";i;"Aufgaben sind":PRINT
270 PRINT fa;"falsch und":PRINT
280 PRINT ra;"richtig":PRINT
290 PRINT" beantwortet worden.":PRINT:PRINT:PRINT
300 PRINT ROUND(fa/i*100,2);
310 PRINT"% der Antworten sind also falsch"
320 PRINT:PRINT" und";ROUND(ra/i*100,2);
330 PRINT" % sind richtig gewesen."
340 PRINT:PRINT:PRINT"Ende der Berechnungen":END
350 REM Unterprogramm Warten
360 LOCATE 5,25
370 PRINT"Bitte eine Taste druecken !"
380 x#=INKEY#
390 IF x#="" THEN 380
400 CLS:RETURN

```

Variablenliste:

ep = eingegebenes Produkt

ep\$ = Eingabefeld für ep

f1 = erster Faktor

f2 = zweiter Faktor

fa = Anzahl der falschen Antworten

gz = größte Zahl der Übung

i = Anzahl der gestellten Aufgaben

p = berechnetes Produkt

ra = Anzahl der richtigen Antworten

Programmbeschreibung:

Satz 10- 70 : Überschrift

Satz 80-120 : Ausgabe einer Erläuterung zur Programmbe-
nutzung und Anforderung eines Wertes für
die größte Zahl der Übung

Satz 130-160 : Eine Aufgabe wird gestellt und ein Ergebnis
wird angefordert

Satz 170-180 : Wenn die Übung abgebrochen werden soll, er-
folgt ein Sprung zur Erfolgsausgabe. Anson-
sten wird die gemachte Eingabe in einen nu-
merischen Wert umgewandelt

Satz 190-240 : Die Eingabe wird auf ihre Richtigkeit hin
überprüft, ein entsprechender Kommentar
wird ausgegeben und das Ergebnis der Über-
prüfung wird registriert. Anschließend er-
folgt eine Verzweigung zum Satz 130, damit
eine weitere Aufgabe gestellt werden kann.

Satz 250-330 : Ausgabe der absoluten und relativen Er-
folgsquote

Satz 340 : Programmende

Satz 350-400 : Unterprogramm "Warten"

9.3. Primzahlenprüfung

Eine ganze Zahl z heißt Primzahl, wenn sie ohne Rest nur durch 1 und sich selbst teilbar ist. Nur die Zahl 1 stellt eine Ausnahme dar. Obwohl 1 nur durch 1 teilbar ist, zählt die 1 nicht zu den Primzahlen.

Sieht man von der Zahl 2 ab, die man üblicherweise nicht zu den Primzahlen zählt, so läßt sich weiterhin feststellen, daß eine gerade Zahl keine Primzahl sein kann, da sie auf jeden Fall durch 2 teilbar ist.

Die Überprüfung ungerader Zahlen nimmt hingegen etwas Zeit in Anspruch. Sie müssen nämlich durch alle Teiler geteilt werden, die größer als 1, aber kleiner als sie selbst sind. Sofern bei diesen Divisionen ein ganzzahliges Ergebnis auftritt (Division ohne Rest), ist die überprüfte Zahl offenbar keine Primzahl. Erst wenn bei allen Divisionen ein Rest auftritt, liegt eine Primzahl vor.

Es ist offenkundig, daß insbesondere die Überprüfung größerer Zahlen ohne Computereinsatz recht aufwendig ist. Zudem ist das entsprechende Programm nicht sehr schwierig. Der Programmablauf ergibt sich eigentlich direkt aus den obigen Ausführungen.

Programm:

```
10 REM I23
20 CLS
30 PRINT "I23 - Programm zur Primzahlen-"
40 PRINT TAB(7)"ueberpruefung"
50 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
60 PRINT TAB(9) "Bernd Kowal, 1985"
70 GOSUB 210
80 PRINT"Geben Sie eine ganze, positive"
90 INPUT"Zahl ein ";z
100 PRINT:PRINT:PRINT
110 IF z<>1 THEN 140
120 PRINT"Die Zahl 1 zaehlt nicht"
130 PRINT"zu den Primzahlen.":GOTO 200
140 FOR i=2 TO z/2
150 IF z/i=INT(z/i) THEN 180
160 NEXT i
170 PRINT z;"ist eine Primzahl.":GOTO 200
180 PRINT z;"ist keine Primzahl, da";z;"durch"
190 PRINT i;"teilbar ist."
200 PRINT:PRINT:PRINT"Ende der Berechnungen":END
210 REM Unterprogramm Warten
220 LOCATE 5,25
230 PRINT"Bitte eine Taste druecken !"
240 x#=INKEY#
250 IF x#="" THEN 240
260 CLS:RETURN
```

Variablenliste:

i = Teiler

z = zu prüfende Zahl

Programmbeschreibung:

Satz 10- 70 : Überschrift

Satz 80- 90 : Eingabe einer Zahl

Satz 100 : Druck dreier Leerzeilen

Satz 110-130 : Ist die Zahl 1 eingegeben worden, wird ein Kommentar ausgegeben und zum Programmende verzweigt

Satz 140-160 : In dieser Programmschleife wird der Teiler i jeweils um 1 erhöht, bis maximal der Wert $z/2$ erreicht ist. Ist eine Division ohne Rest gelungen, wird zur entsprechenden Ergebnisausgabe verzweigt. Die Schleife läuft nur bis $z/2$, da mit größeren Teilern auch kein ganzzahliges Ergebnis mehr erzielt werden kann.
Für $z=2$ und $z=3$ wird diese Schleife nicht durchlaufen, da $z/2 < 2$. Es wird jedoch der richtige Teil der Ergebnisausgabe in Satz 170 erreicht.

Satz 170-190 : Ergebnisausgabe

Satz 200 : Programmende

Satz 210-260 : Unterprogramm "Warten"

9.4. Die Ermittlung einer Nullstelle

Die Berechnung einer Nullstelle für beliebige Funktionen ist ein wichtiger Bestandteil der Mathematik. Viele mathematische Probleme stehen mit ihr im Zusammenhang. Das Verfahren, das dem vorliegenden Programm zugrunde liegt, ist das der Intervallschachtelung. Innerhalb eines Intervalls mit den Intervallgrenzen a und b liegt mit Sicherheit eine Nullstelle, wenn die Vorzeichen der Funktionswerte $f(a)$ und $f(b)$ verschieden sind. Ist nun das Vorzeichen von $f(\frac{a+b}{2})$ gleich dem Vorzeichen von $f(a)$, dann liegt die Nullstelle im rechten Teilintervall. In diesem Fall wird a gleich $\frac{a+b}{2}$ gesetzt und das Verfahren wird wiederholt. Analog zu dieser Vorgehensweise wird b gleich $\frac{a+b}{2}$ gesetzt, wenn die Nullstelle im linken Teilintervall liegt, das Vorzeichen von $f(\frac{a+b}{2})$ also gleich dem Vorzeichen von $f(b)$ ist.

Dieses Verfahren ließe sich quasi endlos wiederholen. Daher ist als Abbruchkriterium ein Wert vorzugeben, welcher die Anzahl der Intervallschachtelungen begrenzt. Im Programm ist das der Wert 500 (Satz 270). Zwar reichten bei kleinen Intervallen auch weniger Schachtelungen aus um ein genaues Ergebnis zu erzielen, jedoch kann dieser Wert bei sehr großen Intervallen schon zu klein sein.

Das vorliegende Programm ist an die Voraussetzung ungleicher Vorzeichen für die Funktionswerte der Intervallgrenzen gebunden, obwohl natürlich auch dann Nullstellen im Intervall vorhanden sein können, wenn diese Voraussetzung nicht erfüllt ist. Es ist dann lediglich ein zu großes Intervall gewählt worden. Weiterhin bleibt der Sonderfall

mehrerer Nullstellen unberücksichtigt. Daher ist es ratsam, mit relativ kleinen Intervallen zu arbeiten. Das Programm können Sie auf beliebige Funktionen anwenden. Es ist dann jeweils nur die Programmzeile 340 entsprechend zu ändern. Achten Sie aber auf die Stetigkeit der eingegebenen Funktion, da ansonsten Fehlermeldungen auftreten. Im vorliegenden Programm ist die einfache Funktion

$$y = x^3 - 5$$

vorgegeben. Diese Funktion hat eine Nullstelle bei

$$x = 1.7099 \dots$$

Programm:

```
10 REM I24
20 CLS
30 PRINT "I24 - Programm zur Nullstellen-"
40 PRINT TAB(7)"bestimmung"
50 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
60 PRINT TAB(9) "Bernd Kowal, 1985"
70 GOSUB 360
80 PRINT"In Satz 340 ist eine stetige Funktion"
90 PRINT"vorgegeben. Durch eine Aenderung dieser"
100 PRINT"Programmzeile, koennen Sie das Programm"
110 PRINT"auch fuer eine andere Funktion benutzen."
120 GOSUB 360
130 PRINT"Welches Intervall soll untersucht"
140 PRINT"werden?":PRINT
150 INPUT"Intervallanfang a ";a
160 INPUT"Intervallende b ";b
170 x=a:GOSUB 340:fa=y
180 x=b:GOSUB 340:fb=y
190 PRINT:PRINT:PRINT"f(a) =";fa
200 PRINT"f(b) =";fb:PRINT:PRINT
```

```

210 IF fa=0 THEN PRINT"Nullstelle bei x =";a:GOTO 330
220 IF fb=0 THEN PRINT"Nullstelle bei x =";b:GOTO 330
230 IF SGN(fa)<>SGN(fb) THEN 270
240 PRINT"Die Voraussetzung, Vorzeichen f(a)<>"
250 PRINT"Vorzeichen f(b) ist nicht erfuehlt."
260 GOTO 330
270 FOR i=1 TO 500
280 x=(a+b)/2:GOSUB 340
290 IF SGN(y)=SGN(fa) THEN a=x:fa=y ELSE b=x:fb=y
300 NEXT i
310 PRINT:PRINT:PRINT"Nullstelle bei x =";x
320 PRINT"f(x) =";y
330 PRINT:PRINT:PRINT"Ende der Berechnungen":END
340 y=x^3-5
350 RETURN
360 REM Unterprogramm Warten
370 LOCATE 5,25
380 PRINT"Bitte eine Taste druecken !"
390 x#=INKEY#
400 IF x#="" THEN 390
410 CLS:RETURN

```

Variablenliste:

a = Intervallanfang

b = Intervallende

fa = Funktionswert des Intervallanfangs

fb = Funktionswert des Intervallendes

i = Laufindex

x = x-Wert der Funktion

y = y-Wert der Funktion

Programmbeschreibung:

Satz 10- 70 : Überschrift

Satz 80-120 : Ausgabe einer Erläuterung und "Warten"

Satz 130-160 : Eingabe der Intervallgrenzen

Satz 170-200 : Die Funktionswerte der Intervallgrenzen
werden durch Sprünge ins Unterprogramm
(Satz 340) berechnet und ausgegeben

Satz 210-220 : Ausgabe eines Kommentars und Verzweigung
zum Programmende, wenn der Funktionswert
einer Intervallgrenze gleich 0 ist

Satz 230-260 : Ist die Voraussetzung ungleicher Vorzeichen nicht erfüllt, wird ein Kommentar ausgegeben und zum Programmende verzweigt

Satz 270 : Eröffnung einer Programmschleife für 500 Intervallschachtelungen

Satz 280 : Das Intervall wird halbiert und der entsprechende Funktionswert wird berechnet

Satz 290 : Wenn das Vorzeichen des Funktionswertes der Intervallmitte $SGN(y)$ gleich dem Vorzeichen des Funktionswertes der linken Intervallgrenze $SGN(fa)$ ist, dann wird die Intervallmitte zur linken Intervallbegrenzung für das darauffolgende halbierte Intervall und der Funktionswert der Intervallmitte wird zum Funktionswert der linken Intervallgrenze des folgenden Intervalls. Ist $SGN(y) \neq SGN(fa)$, wird die Intervallmitte bzw. der Funktionswert der Intervallmitte zur rechten Intervallgrenze bzw. zum Funktionswert der rechten Intervallgrenze des darauffolgenden Intervalls.

Satz 300 : Ende der Programmschleife

Satz 310-320 : Ergebnisausgabe

Satz 330 : Programmende

Satz 340-350 : Unterprogramm für die Funktionsvorgabe

Satz 360-410 : Unterprogramm "Warten"

9.5. Der Vokabeltrainer

Auch beim Erlernen von Fremdsprachen kann Ihnen Ihr CPC wirkungsvolle Hilfe leisten. So wäre es zum Beispiel möglich, ein komplettes Sprachunterrichtsprogramm zu entwickeln. Dies würde jedoch den Rahmen dieses Buches übersteigen. Mit guten bis sehr guten BASIC-Kennnissen könnten Sie es ja einmal selbst versuchen. Natürlich müssen Sie dann schon entsprechende Sprachkenntnisse - und vor allen Dingen Interesse an dieser Thematik haben.

In diesem Kapitel geht es um das "Pauken" von Vokabeln. Wahrscheinlich haben Sie auch Vokabeln mit zugehaltener Spalte im Vokabelheft gelernt. Diese Möglichkeit Vokabeln zu lernen hat den Nachteil, daß die Reihenfolge der Vokabeln immer die gleiche ist. Wird dann im Vokabeltest eine andere Reihenfolge gewählt - und dies ist mit Sicherheit der Fall -, werden Sie Schwierigkeiten haben, die "heimischen" Erfolge zu wiederholen. Sie brauchen also jemanden, der Zeit für Sie hat und mit Ihnen übt. Dieser jemand kann ab heute Ihr CPC sein. Er hat immer Zeit für Sie. Das Programm, das Sie dazu benötigen, finden Sie auf den nächsten Seiten. Es ist so komfortabel gestaltet, daß es die gesamte Organisation des Vokabellernens übernehmen kann.

Folgende Funktionen hat das Programm:

Vokabeln

- eingeben
- einlesen
- verbessern (mit der Möglichkeit des Ansehens aller im Arbeitsspeicher befindlichen Vokabeln, sowie des Löschsens einzelner Vokabeln)
- ergänzen (über die Tastatur oder durch zusätzliches Einlesen)
- lernen (deutsches Wort - Fremdwort oder Fremdwort - deutsches Wort)
- abspeichern

Befinden sich beispielsweise 50 Vokabeln im Arbeitsspeicher des Rechners, dann geht das Lernen so vor sich, daß zunächst 50 Vokabeln zufällig ausgewählt werden. Dabei finden schon richtig beantwortete Vokabeln keine Berücksichtigung. Falsch beantwortete Vokabeln können hingegen mehrmals auftreten. Anschließend kommen dann die Vokabeln an die Reihe, die noch nicht vom Computer ausgewählt worden sind. Zum Schluß werden dann die falsch beantworteten Vokabeln solange abgefragt, bis jede Vokabel mindestens einmal richtig beantwortet ist.

Natürlich ist es auch möglich, die Lernarbeit vorzeitig zu beenden. Auf jeden Fall wird nach Abschluß der Lernarbeit eine "Erfolgsquote" ausgegeben.

Für die Lernarbeit ist es sinnvoll, einzelne Lektionen getrennt abzuspeichern. Nach Belieben können Sie diese dann mit der Ergänzungsfunktion zusammenfassen. Insgesamt ist im Programm Platz für 1000 Vokabeln reserviert worden.

Für den Diskettenbetrieb ist eine unwesentliche Programmänderung notwendig:

```
565 INPUT "Wie heisst die Lektion ";n$
570 OPENIN n$
```

Die Vokabeleingabe vollzieht sich so, daß zuerst das deutsche Wort, dann ein Komma und dann das Fremdwort eingetippt werden, also zum Beispiel

? er,he (ENTER)

Zu beachten ist, daß Ergänzungen nur mit der Ergänzungsfunktion möglich sind. Ein wiederholtes Eingeben oder Einlesen hat zur Folge, daß Vokabeln, die sich im Programmspeicher befinden, überschrieben werden.

Ansonsten ist das Programm weitgehendst selbsterklärend.

Das beste ist, Sie probieren es gleich einmal aus!

Programm:

```
10 REM I25
20 CLS
30 PRINT TAB(7)"I25 - Vokabelprogramm"
40 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
50 PRINT TAB(9) "Bernd Kowal, 1985"
60 GOSUB 2280
70 DIM d$(1000),f$(1000),w(1000),q(1000)
80 k1=1:h=15
90 REM
100 REM Menue
110 REM
120 PRINT TAB(17)"Menue":PRINT:PRINT
130 PRINT TAB(33)"Eingabe":PRINT
140 PRINT"Vokabeln":PRINT
```

```

150 PRINT " eingeben";TAB(36)"1":PRINT
160 PRINT " einlesen";TAB(36)"2":PRINT
170 PRINT " verbessern";TAB(36)"3":PRINT
180 PRINT " ergaenzen";TAB(36)"4":PRINT
190 PRINT " lernen";TAB(36)"5":PRINT
200 PRINT " abspeichern";TAB(36)"6":PRINT
210 PRINT"Arbeit beenden";TAB(36)"7":PRINT:PRINT:PRINT
220 INPUT"Ihre Wahl ";a:CLS
230 IF a<1 OR a>7 THEN CLS:GOTO 120
240 ON a GOTO 370,540,670,260,1390,1880,1980
250 REM
260 REM Vokabeln ergaenzen
270 REM
280 PRINT"Sollen zusaetzliche Vokabeln einge="
290 INPUT"lesen werden (j/n) ";a#
300 IF a#="j" THEN e=k1+1:GOTO 570
310 IF a#<>"n" THEN 280
320 PRINT:PRINT
330 PRINT"Dann koennen Sie jetzt zusaetzliche"
340 PRINT"Vokabeln eingeben.":GOSUB 2280
350 k1=k1+1:h=k1+15
360 REM
370 REM Vokabeln eingeben
380 REM
390 PRINT"Das deutsche - und das Fremdwort sind"
400 PRINT"durch Komma voneinander zu trennen."
410 PRINT"Geben Sie *,* ein, wenn alle Vokabeln"
420 PRINT"eingetippt sind.":PRINT
430 PRINT"Deutsches Wort, Fremdwort":PRINT
440 FOR i=k1 TO 1000
450 INPUT d$(i),f$(i)
460 IF d$(i)<>"*" THEN 500
470 k1=i-1:PRINT:PRINT"Es befinden sich jetzt";k1
480 PRINT"Vokabeln im Arbeitsspeicher."
490 GOSUB 2280:GOTO 120
500 IF i=h THEN k1=h+1:h=h+15:GOSUB 2280:GOTO 390
510 NEXT i
520 GOSUB 2280
530 REM
540 REM Vokabeln einlesen
550 REM
560 e=1
570 OPENIN ""
580 FOR i=e TO 1000
590 INPUT #9,d$(i):INPUT #9,f$(i)
600 IF EOF=-1 THEN k1=i:GOTO 620
610 NEXT i

```

```

620 CLOSEIN
630 PRINT:PRINT"Es befinden sich jetzt";k1;
640 PRINT"Vokabeln im Arbeitsspeicher."
650 GOSUB 2280:GOTO 120
660 REM
670 REM Vokabeln verbessern
680 REM
690 PRINT"Koennen Sie das falsche Wort benennen,"
700 PRINT"oder wollen Sie sich lieber alle"
710 PRINT"Vokabeln einmal ansehen?":PRINT
720 PRINT TAB(20)"Eingabe":PRINT
730 PRINT"benennen";TAB(23)"1":PRINT
740 PRINT"ansehen";TAB(23)"2"
750 PRINT:INPUT"Ihre Wahl ";a:CLS
760 IF a=2 THEN 920
770 IF a<>1 THEN 690
780 PRINT"Geben Sie bitte das falsche Wort"
790 INPUT"ein ";fw#
800 INPUT"Und wie heisst es richtig ";rw#
810 FOR i=1 TO k1
820 IF d$(i)=fw# THEN d$(i)=rw#:GOTO 870
830 IF f$(i)=fw# THEN f$(i)=rw#:GOTO 870
840 NEXT i
850 PRINT:PRINT"Das angegebene Wort ist nicht zu finden"
860 GOTO 880
870 PRINT:PRINT"Die Verbesserung ist erfolgt."
880 PRINT:INPUT"Noch eine Verbesserung (j/n) ";a#
890 IF a#="n" THEN CLS:GOTO 120
900 IF a#<>"j" THEN 880
910 CLS:GOTO 690
920 WINDOW #1,20,40,1,25
930 k2=1:k3=10
940 IF k2<=k1 THEN 1040
950 FOR i=1 TO k1
960 IF d$(i)<>" " THEN 1020
970 k1=k1-1
980 FOR j=1 TO k1
990 d$(j)=d$(j+1)
1000 f$(j)=f$(j+1)
1010 NEXT j
1020 NEXT i
1030 CLS:GOTO 120
1040 FOR i=k2 TO k3
1050 IF i>k1 THEN 1090
1060 PRINT #1,i;". ";d$(i);"- "
1070 PRINT #1,TAB(6)f$(i)
1080 NEXT i

```

```

1090 PRINT"Wo befindet sich"
1100 PRINT"der Fehler ?"
1110 PRINT"Geben Sie eine 0"
1120 PRINT"ein, wenn alles"
1130 PRINT"richtig ist."
1140 PRINT"Die Eingabe eines"
1150 PRINT"Leerzeichens fuer"
1160 PRINT"das deutsche Wort"
1170 PRINT"bewirkt die Loe-"
1180 PRINT"schung der ent-"
1190 PRINT"sprechenden"
1200 PRINT"Nummer. ";PRINT
1210 INPUT"Nummer ";b
1220 IF b=0 THEN 1370
1230 PRINT:PRINT"Ist das deutsche"
1240 PRINT"Wort falsch"
1250 INPUT"(j/n) ";a#
1260 IF a#="n" THEN 1300
1270 IF a#("<"j" THEN 1230
1280 PRINT"Wie heisst es rich-"
1290 INPUT"tig ";d#(b)
1300 PRINT:PRINT"Ist das Fremdwort"
1310 INPUT"falsch (j/n) ";a#
1320 IF a#="n" THEN 1360
1330 IF a#("<"j" THEN 1300
1340 PRINT"Wie heisst es rich-"
1350 INPUT"tig ";f#(b)
1360 CLS:CLS #1:GOTO 1040
1370 CLS:CLS #1:k2=k3+1:k3=k3+10:GOTO 940
1380 REM
1390 REM Vokabeln lernen
1400 REM
1410 l=0:rr=0:ff=0
1420 FOR i=1 TO k1:q(i)=0:w(i)=0:NEXT i
1430 PRINT k1;"Vokabeln gehen in die Uebung ein."
1440 PRINT:PRINT"Geben Sie ein * fuer die abgefragte"
1450 PRINT"Vokabel ein, wenn Sie die Lernarbeit"
1460 PRINT"beenden wollen.":PRINT
1470 PRINT:PRINT"Wie sollen die Vokabeln abgefragt"
1480 PRINT"werden ?"
1490 PRINT:PRINT TAB(33)"Eingabe":PRINT
1500 PRINT"Deutsches Wort - Fremdwort";TAB(36)"1"
1510 PRINT
1520 PRINT"Fremdwort - Deutsches Wort";TAB(36)"2"
1530 PRINT:INPUT"Ihre Wahl ";c
1540 IF c<1 OR c>2 THEN 1470 ELSE CLS
1550 y=INT(RND*k1+1)

```

```

1560 IF w(y)=1 THEN 1550
1570 GOSUB 2110
1580 l=l+1;q(y)=1
1590 IF l<k1 THEN 1550
1600 FOR i=1 TO k1
1610 IF q(i)=1 THEN 1630
1620 y=i:GOSUB 2110:l=l+1
1630 NEXT i
1640 hi=0
1650 FOR i=1 TO k1
1660 IF w(i)=1 THEN 1690
1670 y=i:GOSUB 2110:l=l+1
1680 IF w(y)=0 THEN hi=1
1690 NEXT i
1700 IF hi=1 THEN 1640
1710 PRINT"Es sind jetzt alle Vokabeln mindestens"
1720 PRINT"einmal richtig beantwortet worden.":PRINT
1730 PRINT:PRINT"Bei der Abfrage von insgesamt";l;
1740 PRINT"Vokabeln, waren";rr;"Antworten"
1750 PRINT"richtig und";ff;"falsch.":PRINT:PRINT
1760 PRINT"In Prozenten ausgedrueckt :":PRINT
1770 PRINT; USING "###.##";rr*100/l;
1780 PRINT"% der Antworten waren richtig"
1790 PRINT"und"
1800 PRINT; USING "###.##";ff*100/l;
1810 PRINT"% der Antworten waren falsch"
1820 PRINT:PRINT: IF ff<=rr THEN 1840
1830 PRINT"Sie muessen also noch etwas ueben."
1840 IF ff*10>rr THEN 1860
1850 PRINT"Eine ordentliche Leistung."
1860 GOSUB 2280:GOTO 120
1870 REM
1880 REM Vokabeln abspeichern
1890 REM
1900 PRINT"Wie soll die abzuspeichernde"
1910 INPUT"Lektion heissen ";n#:PRINT
1920 OPENOUT n#
1930 FOR i=1 TO k1
1940 PRINT #9,d#(i):PRINT #9,f#(i)
1950 NEXT i:CLOSEOUT
1960 CLS:GOTO 120
1970 REM
1980 REM Programm beenden
1990 REM
2000 PRINT"Haben Sie auch alle Vokabeln"
2010 PRINT"ordnungsgemaess abgespeichert ?"
2020 PRINT"Falls nicht, geben Sie einfach"

```

```

2030 PRINT"ein 'oh' ein (ansonsten reicht die"
2040 PRINT"Leertaste)."
```

2050 INPUT a#

```

2060 IF a#="oh" THEN CLS:GOTO 1900
2070 PRINT:PRINT:PRINT"      Auf Wiedersehen"
2080 PRINT:PRINT"Sie koennen mich jetzt ausschalten"
2090 END
2100 REM
2110 REM UP Abfragen
2120 REM
2130 LOCATE 8,4
2140 IF c=2 THEN 2170
2150 PRINT d#(y);"  ";:INPUT p#
2160 IF p#=f#(y) THEN 2190 ELSE 2210
2170 PRINT f#(y);"  ";:INPUT p#
2180 IF p#<>d#(y) THEN 2210
2190 PRINT:PRINT:PRINT TAB(9)"richtig"
2200 rr=rr+1:w(y)=1:GOTO 2260
2210 PRINT:IF p#="*" THEN CLS:GOTO 1730
2220 PRINT:PRINT TAB(9)"falsch":ff=ff+1:w(y)=0
2230 PRINT:PRINT:PRINT TAB(9)"richtig heisst es :"
```

2240 PRINT:IF c=1 THEN PRINT TAB(9)f#(y)

```

2250 IF c=2 THEN PRINT TAB(9)d#(y)
2260 GOSUB 2280:RETURN
2270 REM
2280 REM UP Warten
2290 REM
2300 LOCATE 7,25
2310 PRINT"Bitte eine Taste druecken !"
```

2320 x#=INKEY#

```

2330 IF x#="" THEN 2320
2340 CLS:RETURN
```

Variablenliste:

a = Antwort "Menüauswahl"

a\$ = Antwortstring (j/n)

b = Hilfsvariable und Index (Programmteil "Verbesserungen")

c = Antwort "Menüauswahl" im Programmteil "Lernen"

d\$(i) = deutsches Wort

e = Schleifenanfangswert (größer als 1, wenn Ergänzungen vorgenommen werden)

f\$(i) = Fremdwort

ff = Anzahl der falschen Antworten

fw\$ = falsches Wort (Programmteil "Verbesserungen")

h = Hilfsvariable, welche anzeigt, ob der Bildschirm voll ist

hi = Hilfsvariable (gleich 1, wenn noch richtige Antworten fehlen)

i = Laufindex

j = Laufindex

k1 = Anzahl der Vokabeln im Arbeitsspeicher

k2 = Schleifenanfang

k3 = Schleifenende

l = Anzahl der gesamten Antworten

n\$ = Name der abzuspeichernden Lektion

p\$ = Eingabe im Unterprogramm "Abfragen"

q(i) = Hilfsvariable (gleich 0, wenn eine Vokabel noch nicht abgefragt worden ist, ansonsten gleich 1)

r = Anzahl der richtigen Antworten

rw\$ = richtiges Wort (Programmteil "Verbesserungen")

w(i) = Hilfsvariable (gleich 0, wenn eine Vokabel noch nicht richtig beantwortet worden ist, ansonsten gleich 1)

y = Index der aktuell abgefragten Vokabel

Programmbeschreibung:

Satz 10- 60 : Überschrift

Satz 70 : Speicherplatzreservierung

Satz 80 : Variablen werden mit Ausgangswerten belegt

Satz 90- 240 : Menüauswahl und Programmverzweigungen

Satz 250- 350 : Programmteil "Vokabeln ergänzen"

Es besteht die Möglichkeit entweder zusätzliche Vokabeln einzulesen (Programmverzweigung und Belegung der Variablen e) oder einzugeben (Veränderung der Variablen k1 und h).

Satz 360- 520 : Programmteil "Vokabeln eingeben"

Bis zu 1000 Vokabeln können eingegeben werden. Nach der Eingabe von jeweils 15 Vokabeln erfolgt ein Sprung ins Unterprogramm "Warten".

Satz 530- 650 : Programmteil "Vokabeln einlesen"

Werden Vokabeln ergänzend eingelesen, dann ist e gleich k1+1.

Satz 660-1370 : Programmteil "Vokabeln verbessern"

690- 770 : Menüauswahl

780- 910 : Verbesserung durch Eingabe des falsch - und des richtig geschriebenen Wortes. Der Vergleich der Eingabe "falsches Wort" mit allen intern gespeicherten Vokabeln erfolgt in der Programmschleife Satz 810-840.

920-1370 : Ansehen und verbessern oder löschen der Vokabeln
Die Ausgabe der Vokabeln (Satz 1040-1080) erfolgt in einem gesondert definierten Textfenster (siehe Satz 920). Um ein "Wegrollen" des Ausgabefensters zu vermeiden, werden nur immer 10 Vokabeln angezeigt (siehe Schleifenanfangs- und endwert Satz 1040 (Erstbelegung Satz 930) und deren Neubelegung in Satz 1370).

Die Sätze 1090-1360 dienen der Benutzeranleitung und der Verbesserung von Vokabeln. Erst wenn die ausgegebenen Vokabeln in Ordnung sind (Eingabe 0), werden weitere Vokabeln zum Ansehen ausgegeben (siehe Satz 1370). Ansonsten wiederholt sich die gleiche Ausgabe (natürlich mit evtl. Verbesserungen, siehe Satz 1360).

In den Sätzen 950-1020, die erst nach der Ausgabe aller Vokabeln erreicht werden (siehe Satz 940), wird überprüft, ob Vokabeln gelöscht werden sollen (dann ist $d\$(i) = "$). Ist das der Fall, so verändern sich die Indizes der folgenden Vokabeln entsprechend (siehe Programmschleife Satz 980-1010).

In Satz 1030 wird dieser Programmteil durch Rücksprung zur Menüauswahl beendet.

Satz 1380-1860 : Programmteil "Vokabeln lernen"

1410-1420 : Variablen werden mit Ausgangswerten belegt

1430-1540 : Erläuterungen und Menüauswahl

1550-1590 : k_1 -mal dient die in Satz 1550 erzeugte Zufallszahl (zwischen 1 und k_1) der Indexbestimmung für die Vokabelabfrage (Unterprogramm Satz 2100-2260). Falls eine entsprechende Abfrage schon durchgeführt worden ist, wird die nächste Zufallszahl genommen (siehe Satz 1560).

1600-1630 : Es werden die Indizes der Vokabeln bestimmt, die noch nicht abgefragt wurden

1640-1700 : Solange noch Vokabelabfragen falsch beantwortet werden (siehe Hilfsgröße h_i), dienen die entsprechenden Indizes als Ausgangsgrößen für das Unterprogramm "Abfragen"

1710-1860 : Die Ergebnisse der Lernarbeit werden ausgegeben

Satz 1870-1960 : Programmteil "Vokabeln abspeichern"

Alle im Arbeitsspeicher vorhandenen Vokabeln werden abgespeichert (wie in den vorangegangenen Programmen, so ist auch hier prinzipiell der Einsatz eines Diskettenlaufwerks möglich)

Satz 1970-2090 : Programmteil "Programm beenden"

Es besteht hier noch die Möglichkeit Vokabeln abzuspeichern, da dies - nach erfolgter Lernarbeit - häufig vergessen wird


Satz 2100-2260 : Unterprogramm "Abfragen"

Nachdem im Programmteil "Vokabeln lernen" der Index y bestimmt worden ist, wird hier eine entsprechende Eingabe angefordert. Richtige und falsche Antworten werden registriert.

Satz 2270-2340 : Unterprogramm "Warten"

9.6. Ideenfundgrube

- mathematische Probleme jeglicher Art
- Reaktionsgleichungen und stöchiometrisches Rechnen
- physikalische Gesetze
- optische Abbildungen
- Wörterbücher programmiert
- Rechtschreibtraining
- Literaturdatenbank
- Gedichte analysieren (siehe dazu auch Kapitel 8.5)
- historische Ereignisse speichern und abrufen
- Darstellung von biologischen Wachstumsprozessen
- Musik nach Noten
- und noch vieles mehr



10. Spielen mit dem CPC

10.1. Vorbemerkung

Es gibt Leute, die ausschließlich mit ihrem Computer spielen. Für diesen Personenkreis ist das vorliegende Buch ideal, da doch einige "sinnvolle" Anwendungen dargestellt werden (obgleich natürlich jeder seine eigene Philosophie bezüglich des Wörtchens "sinnvoll" haben wird). Nichtsdestoweniger kann das eine oder andere Spielchen recht interessant sein, zumal wenn man das zugehörige Programm selbst geschrieben hat und somit zusätzliche Variationen ohne Probleme in den Spielablauf einfügen kann. Schade ist es nur, wenn der Computer zum Spielobjekt degradiert wird und seine sonstigen Fähigkeiten ungenutzt bleiben. Aber letztlich muß das selbstverständlich jeder für sich alleine entscheiden.

In diesem Kapitel werden zwei vergleichsweise einfache Spielprogramme vorgestellt, die jedoch zwei wesentliche

Grundelemente der Spieleprogrammierung beinhalten, nämlich die Darstellung von Bewegungsabläufen und die Simulation von "Denkschritten". Allerdings bietet der CPC noch viel mehr Möglichkeiten zur Programmierung von Computerspielen. Vielleicht schauen Sie sich dazu einmal den gesamten Zeichensatz Ihres CPC mit der Funktion CHR\$(i), für i=32 bis i=255, an. Grundsätzlich ist es aber das gleiche, ob nun ein kleines Männchen oder ein Punkt bzw. Ball über den Bildschirm "läuft". Sie bleiben also auch im letzten Kapitel aufgefordert, die vorhandenen Programme zu verändern und zu erweitern oder ganz einfach Anregungen zu übernehmen. Und wenn alle Stricke reißen ... der Software-Markt ist groß.

10.2. Bildschirmtennis in Abwandlung

Vor ca. 10 Jahren waren die Telespiele sehr groß in Mode. Man konnte und kann mit ihnen unter anderem "Tennis" spielen. Mit recht einfachen Mitteln läßt sich ein derartiges Spiel programmieren.

Grundsätzlich werden im Programmablauf Koordinaten eines Punktes berechnet, der Punkt wird gezeichnet und der vorangegangene Punkt wird gelöscht. So entsteht der Eindruck der Bewegung. Erreicht der Punkt die rechte Bildschirmbegrenzung, wird er zurückgespielt, wenn die vom Spieler zu bewegende Linie bzw. der "Schläger" mit dem Punkt bzw. "Ball" zusammentrifft. Ansonsten wird ein

neuer "Ball" ins Spiel gebracht.

Der "Schläger" wird mit den Cursorsteuertasten bewegt. Treffer und Nicht-Treffer werden aufsummiert und angezeigt. Die Geschwindigkeit des "Balles" und des "Schlägers", sowie die Länge des "Schlägers" kann der Benutzer frei wählen, d. h. er kann die Schwierigkeitsstufe selbst bestimmen. Weiterhin kann er festlegen, ob die Ballgeschwindigkeit variieren soll.

Mit Hilfe kleiner Programmänderungen lassen sich für die Steuerung des "Schlägers" auch Joysticks einsetzen. Sollten Sie zwei Joysticks zur Verfügung haben, so könnten Sie auch einen zweiten "Schläger" ins Spiel bringen. Dann müßten Sie aber auch die Spielregeln bzw. die Trefferanzeige ändern. Sie sehen, es gibt viele Möglichkeiten ein derartiges Programm zu ändern, um keine Langeweile aufkommen zu lassen. Viel Spaß!

Programm:

```
10 REM I26
20 CLS
30 PRINT "I26 - Ballspiel (Tennis in Abwandlung)"
40 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
50 PRINT TAB(9) "Bernd Kowal, 1985"
60 GOSUB 570
70 PRINT"Parametereingabe :":PRINT:PRINT
80 PRINT"Ballgeschwindigkeit"
90 INPUT"(Wert zwischen 10 und 30)      ";bg
100 IF bg<10 OR bg>30 THEN 90
110 PRINT:PRINT"Soll diese Geschwindig-"
120 PRINT"keit waehrend des Spiels"
130 INPUT"schwanken (j/n)              ";a$
140 IF a$="n" THEN z=0:GOTO 170
150 IF a$="j" THEN z=bg/2:bg=bg*3/4:GOTO 170
160 GOTO 110
170 PRINT:PRINT"Laenge des 'Schlaegers'"
```

```

180 INPUT"(Wert zwischen 10 und 100)      ";s1
190 IF s1<10 OR s1>100 THEN 180
200 PRINT:PRINT"Geschwindigkeit des Schlaegers"
210 INPUT"(Wert zwischen 5 und 50)      ";sg
220 IF sg<5 OR sg>50 THEN 210
230 PRINT:PRINT:PRINT"Das Spiel wird durch zweimaliges"
240 PRINT"Druecken der ESC-Taste beendet."
250 GOSUB 570
260 KEY DEF 0,1,111
270 KEY DEF 2,1,117
280 SPEED KEY 1,1
290 PLOT 620,0,1:DRAW 0,0,1:DRAW 0,399,1:DRAW 620,399,1
300 a=INT(200-s1/2):b=INT(200+s1/2)
310 PLOT 620,a,1:DRAW 620,b,1
320 GOSUB 510
330 nx=2:ny=200:GOTO 490
340 ax=nx:ay=ny
350 nx=nx+xv:ny=ny+yv
360 IF ny>397 THEN ny=397:yv=-yv
370 IF ny<2 THEN ny=2:yv=-yv
380 IF nx<2 THEN nx=2:xv=-xv
390 ON BREAK GOSUB 630
400 s#=INKEY#
410 IF s#="e" THEN PLOT 620,a,0:a=a+sg:DRAW 620,a,0:
    b=b+sg:DRAW 620,b,1
420 IF s#="u" THEN PLOT 620,b,0:b=b-sg:DRAW 620,b,
    0:a=a-sg:DRAW 620,a,1
430 IF SGN(xv)=-1 THEN 480
440 IF nx<620 THEN 480
450 nx=620:ny=ay+yv/xv*(620-ax)
460 IF ny>=a AND ny<=b THEN PRINT CHR$(7):t=t+i:
    GOSUB 510:xv=-xv:GOTO 480
470 d=d+1:GOSUB 510:PLOT ax,ay,0:GOTO 330
480 PLOT ax,ay,0
490 PLOT nx,ny,1
500 GOTO 340
510 LOCATE 2,2:PRINT"Treffer ";t
520 LOCATE 2,3:PRINT"daneben ";d
530 xv=ROUND(RND*z+bg)
540 yv=ROUND(RND*20-10)
550 IF yv<5 AND yv>=-5 THEN 540
560 RETURN
570 REM Unterprogramm Warten
580 LOCATE 5,25
590 PRINT"Bitte eine Taste druecken !"
600 x#=INKEY#
610 IF x#="" THEN 600
620 CLS:RETURN
630 SPEED KEY 10,4:END

```

Variablenliste:

a = untere y-Koordinatenbegrenzung für den Schläger

a\$ = Antwortstring (j/n)

ax = "alte" x-Koordinate des Balles

ay = "alte" y-Koordinate des Balles

b = obere y-Koordinatenbegrenzung für den Schläger

bg = Ballgeschwindigkeit

d = Anzahl der Nicht-Treffer (d wie daneben)

n = "neue" x-Koordinate des Balles

ny = "neue" y-Koordinate des Balles

s\$ = Wert der gedrückten Taste zur Schlägersteuerung

sg = Schlägergeschwindigkeit

sl = Schlägerlänge

t = Anzahl der Treffer

xv = Bewegungsgeschwindigkeit des Balles in x-Richtung

y = Bewegungsgeschwindigkeit des Balles in y-Richtung

z = Hilfsvariable (gleich $bg/2$, wenn die Ballgeschwindigkeit variieren soll, ansonsten gleich 0)

Programmbeschreibung:

Satz 10- 60 : Überschrift

Satz 70-250 : Eingabe der Parameter und Ausgabe einer Erläuterung zur Programmbenutzung. Die Größenbegrenzungen für die angeforderten Werte sind dabei ausschließlich nach dem Kriterium eines "sinnvollen" Spielverlaufs ausgewählt worden. Die Berechnungen in Satz 150 dienen der Bestimmung der zufällig variierenden Bewegungsgeschwindigkeit in Satz 530.

Satz 260-270 : Den Cursorsteuertasten werden Werte zugeordnet. Beim Drücken der↑-Taste wird dann ein "o" (o wie oben) und beim Drücken der ↓-Taste ein "u" (u wie unten) zurückgegeben.

Satz 280 : Es werden neue Werte bezüglich der Wiederholungsgeschwindigkeit einer dauernd gedrückten Taste gesetzt, um die Beweglichkeit des Schlägers zu vergrößern

Satz 290 : Die Spielfeldumrandung wird gezeichnet

Satz 300 : Die Begrenzungskordinaten (y-Richtung) für den Schläger werden berechnet

Satz 310 : Der Schläger wird gezeichnet

Satz 320 : Sprung ins Unterprogramm (Satz 510) zur Bestimmung der Bewegungsgeschwindigkeit des Balles

Satz 330 : Die Ausgangskordinaten des Balles werden festgelegt und der Ball wird direkt gezeichnet (Sprung zum Satz 490)

Satz 340-350 : Die neue Ballposition wird bestimmt (alte Position + Bewegungsgeschwindigkeit), nachdem die alte Position in den Feldern ax und ay abgelegt wurde

Satz 360-380 : Wird bei der Positionsveränderung der obere -, untere - oder linke Rand des Spielfeldes erreicht oder überschritten, so wird als Maximalordinate ein Wert direkt neben dem Rand angenommen und die Bewegungsrichtung wird umgedreht ($yv=-yv$ bzw. $xv=-xv$)

Satz 390 : Nach zweimaligem Drücken der ESC-Taste wird ins Unterprogramm (Satz 630) zur Beendigung des Spiels und des Programms verzweigt

Satz 400-420 : Der Wert der gedrückten Taste wird zurückgegeben. Die Betätigung der \uparrow -Taste führt dazu, daß in Anhängigkeit von der angegebenen Schlägergeschwindigkeit sg eine "farblose" Linie von der unteren Schlägerposition zur neuen unteren Schlägerposition ($a=a+sg$) - und von dieser Position aus eine Linie zur neuen oberen Schlägerposition ($b=b+sg$) gezeichnet wird. Bei der Betätigung der \downarrow -Taste wird entsprechend verfahren.

Satz 430 : Bewegt sich der Punkt von rechts nach links (der Ball wird also zurückgeschlagen), kann der Punkt direkt gezeichnet werden (Sprung zum Satz 480)

Satz 440-470 : Es wird untersucht, ob der Ball mit dem Schläger getroffen wird. Diese Frage stellt sich nicht, wenn die rechte Bildschirmbegrenzung noch nicht erreicht ist (Satz

440). Ist sie erreicht, werden in Satz 450 die Koordinaten n_x und n_y so bestimmt, daß sie auf der Schlägerlinie mit der x -Koordinate 620 liegen (siehe dazu die Skizze im Anschluß an diese Beschreibung). Liegt dann die Koordinate n_y innerhalb des Intervalls mit den Grenzen a und b , ist ein Treffer erzielt worden ("Piepston", $t=t+1$). Es werden dann im Unterprogramm (Satz 510) neue Bewegungsgeschwindigkeiten bestimmt und der Ball wird zurückgespielt ($xv=-xv$). Liegt n_y nicht innerhalb des Intervalls (daneben), wird mit neuen Bewegungsgeschwindigkeiten ein neuer Ball ins Spiel gebracht (ab Satz 330), nachdem an der alten Position ein farbloser Punkt gezeichnet worden ist.

Satz 480 : Der "alte" Punkt wird gelöscht

Satz 490 : Der "neue" Punkt wird eingezeichnet


Satz 500 : Rücksprung zum Satz 340, um das Spiel fortzusetzen

Satz 510-560 : Unterprogramm zur Ausgabe der Erfolgsquote und zur Bestimmung der Bewegungsgeschwindigkeiten des Balles. In Satz 530 können die Werte für xv um 25% über oder unter der angegebenen Ballgeschwindigkeit liegen, falls dies gewünscht wurde (Satz 110-150). Für yv werden die Werte zufällig bestimmt. Sie liegen entweder zwischen -10 und -5 oder zwischen 5 und 10.

Satz 570-620 : Unterprogramm "Warten"

Satz 630 : Die Wiederholungsgeschwindigkeit einer dauernd gedrückten Taste wird "normalisiert" und das Programmende ist erreicht

Skizze zur Bestimmung der Koordinate ny in Satz 450:



gesucht: y_k

$$\tan \varphi = \frac{y_v}{x_v} \quad \text{und} \quad \tan \varphi = \frac{y_k}{620 - ax}$$

somit:

$$\frac{y_v}{x_v} = \frac{y_k}{620 - ax}$$

$$y_k = \frac{y_v}{x_v} \cdot (620 - ax)$$

für ny ergibt sich:

$$ny = ay + y_k = ay + y_v / x_v \cdot (20 - ax)$$

10.3. Vier gewinnt

Was machen Sie, wenn Sie ein Spiel für zwei Personen spielen wollen? Die Antwort ist leicht: Sie suchen sich einen Spielpartner. Aber was machen Sie, wenn keiner mit Ihnen spielen will? Nur für den Besitzer eines Computers ist diese Frage leicht zu beantworten: Sie laden das entsprechende Programm und ersetzen den Spielpartner durch Ihren Computer.

Wie bringt man denn nun einen Rechner dazu, daß er, wie es sich für einen ordentlichen Spielpartner gehört, "nachdenkt"? Zuerst müssen Sie ihn mit dem Spielfeld und den Spielregeln vertraut machen. Bei "Vier gewinnt" besteht das Spielfeld aus einem Raster von 8x6 Feldern. Ziel des Spiels ist es, vier Chips in eine Reihe zu bekommen (egal ob horizontal, vertikal oder diagonal). Ihre Chips unterscheiden sich dabei natürlich von den Chips Ihres Spielpartners.

Für eine mittlere Spielstärke ist es dann ausreichend, Ihren Computer folgende Überlegungen durchführen zu lassen:

- 1) Kann ich (der Computer) mit einem Spielzug das Spiel für mich entscheiden?
- 2) Kann ich (der Computer) mit dem folgenden Zug verhindern, daß mein Gegner das Spiel gewinnt?

Zusätzlich besteht dann noch die Möglichkeit, taktische Spielvarianten in den Programmablauf aufzunehmen. Bei "Vier gewinnt" könnte eine Taktik - die auch zum Spielgewinn führt - darin bestehen, möglichst drei Chips in eine Reihe zu bekommen, die nicht von gegnerischen Spielsteinen eingegrenzt werden. In den einzelnen Programmteilen müßten dann zusätzliche Anweisungen, beginnend mit

IF h=2 AND ... THEN ...

aufgenommen werden.

In dem folgenden Programm beschränken wir uns aber auf die ersten beiden Denkschritte. Ein Nicht-Zutreffen führt dann eben nicht zum Griff in die taktische Trickkiste, sondern es wird der Zufallsgenerator bemüht. Auswirkungen des übernächsten Spielzugs werden also nicht berücksichtigt.

Ein Wort noch zur Programmiertechnik:

Der Computer denkt im Programm flächendeckend nach, d. h. es werden u. U. sehr viele Felder untersucht, die noch gar nicht spielrelevant sind. Eine andere Programmiertechnik, die in der Anfangsphase des Spiels auch kürzere Rechenzeiten zur Folge hätte, könnte darin bestehen, die Denkarbeit auf die besetzten Felder zu beschränken. Die einzelnen IF ... THEN-Anweisungen blieben dabei prinzipiell die gleichen.

Ansonsten bleibt Ihnen noch viel Erfolg bei dem Vergleich Mensch - Maschine zu wünschen. Bedenken Sie, drei Chips in einer Reihe übersieht Ihr Rechner auf gar keinen Fall ...

Programm:

```
10 REM I27
20 CLS
30 PRINT TAB(9)"I27 - Vier gewinnt"
40 PRINT:PRINT:PRINT:PRINT:PRINT:PRINT
50 PRINT TAB(10) "Bernd Kowal, 1985"
60 GOSUB 1180
70 PRINT"Spielregeln :":PRINT:PRINT
80 PRINT"Ziel des Spiels ist es, vier Chips in"
90 PRINT"eine Reihe zu bekommen (egal ob hori-"
100 PRINT"zontal, vertikal oder diagonal)."
```

```

410 LOCATE j,i+1:PRINT CHR$(149)
420 LOCATE j,i+2:PRINT CHR$(149)
430 NEXT i:LOCATE j,i:PRINT CHR$(155):NEXT j
440 PRINT "   ";:FOR i=1 TO 8
450 PRINT i;" ";
460 r(i,0)=1:NEXT i
470 IF a#="n" THEN s=ROUND(RND*4+2.5):cd=143:
    GOSUB 1080:co(s,j)=1:GOSUB 1110
480 LOCATE 3,23:INPUT"Ihre Wahl ";s
490 IF s<1 OR s>8 THEN 480
500 cd=207:GOSUB 1080:sb(s,j)=1:GOSUB 1110
510 cd=143:FOR i=1 TO 8
520 FOR j=6 TO 3 STEP -1
530 IF r(i,j)=0 THEN 600
540 IF p=1 THEN 580
550 IF co(i,j)=0 THEN 610
560 IF co(i,j-1)=1 AND co(i,j-2)=1 THEN s=i:
    GOSUB 1080:GOSUB 1110:GOTO 1060
570 GOTO 600
580 IF sb(i,j)=0 THEN 610
590 IF sb(i,j-1)=1 AND sb(i,j-2)=1 THEN r
    (s,j1)=0:co(s,j1)=0:GOTO 1000
600 NEXT j
610 NEXT i
620 FOR j=1 TO 6
630 sb1=1:sb2=4:h=0
640 FOR i=sb1 TO sb2
650 IF p=1 THEN 680
660 IF co(i,j)=0 THEN i1=i ELSE h=h+1
670 GOTO 690
680 IF sb(i,j)=0 THEN i1=i ELSE h=h+1
690 NEXT i
700 IF p=0 AND h=3 AND r(i1,j)=0 AND r(i1,j-1)=1
    THEN s=i1:GOSUB 1080:GOSUB 1110 :GOTO 1060
710 IF p=1 AND h=3 AND r(i1,j)=0 AND r(i1,j-1)=1
    THEN r(s,j1)=0:co(s,i1)=0:GOTO 1000
720 IF sb1<6 THEN h=0:sb1=sb1+1:sb2=sb2+1:GOTO 640
730 NEXT j
740 FOR i=1 TO 5
750 FOR j=1 TO 3
760 ip=i:jp=j:h=0

```

```

770 IF p=1 THEN 800
780 IF co(ip,jp)=0 THEN ip1=ip:jp1=jp ELSE h=h+1
790 GOTO 810
800 IF sb(ip,jp)=0 THEN ip1=ip:jp1=jp ELSE h=h+1
810 IF ip<i+3 THEN ip=ip+1:jp=jp+1:GOTO 770
820 IF p=0 AND h=3 AND r(ip1,jp1)=0 AND r
(ip1,jp1-1)=1 THEN s=ip1:GOSUB 1080:GOS
UB 1110:GOTO 1060
830 IF p=1 AND h=3 AND r(ip1,jp1)=0 AND r
(ip1,jp1-1)=1 THEN r(s,j1)=0:co(s,j1)=0:GOTO 1000
840 NEXT j
850 NEXT i
860 FOR i=4 TO 8
870 FOR j=1 TO 3
880 ip=i:jp=j:h=0
890 IF p=1 THEN 920
900 IF co(ip,jp)=0 THEN ip1=ip:jp1=jp ELSE h=h+1
910 GOTO 930
920 IF sb(ip,jp)=0 THEN ip1=ip:jp1=jp ELSE h=h+1
930 IF ip>i-3 THEN ip=ip-1:jp=jp+1:GOTO 890
940 IF p=0 AND h=3 AND r(ip1,jp1)=0 AND r(ip1,jp1-1)=1
THEN s=ip1:GOSUB 1080:GOSUB 1110:GOTO 1060
950 IF p=1 AND h=3 AND r(ip1,jp1)=0 AND r(ip1,jp1-1)=1
THEN r(s,j1)=0:co(s,j1)=0:GOTO 1000
960 NEXT j
970 NEXT i
980 IF p=1 THEN p=0:GOSUB 1110:GOTO 480
990 FOR i=1 TO 8:s(i)=0:NEXT i:l=0
1000 l=l+1:IF l>8 THEN 1050
1010 IF l=1 THEN s(1)=ROUND(RND*8+0.5):GOTO 1030
1020 s(1)=s(1-1)+1:IF s(1)=9 THEN s(1)=1
1030 s=s(1):GOSUB 1080:co(s,j)=1:j1=j
1040 p=1:GOTO 510
1050 LOCATE 19,23:PRINT"Sie haben gewonnen":GOTO 1070

```

```

1060 LOCATE 19,23:PRINT"Ich habe gewonnen"
1070 END
1080 FOR j=1 TO 6
1090 IF r(s,j)=0 THEN r(s,j)=1:RETURN
1100 NEXT j:IF cd=207 THEN 480 ELSE 1000
1110 z=7:FOR i=2 TO 17 STEP 3
1120 z=z-1:k=0
1130 IF r(s,z)=1 AND r(s,z+1)=1 THEN RETURN
1140 LOCATE s*3+s,i:PRINT CHR$(cd);CHR$(cd);CHR$(cd)
1150 LOCATE s*3+s,i+1:PRINT CHR$(cd);CHR$(cd);CHR$(cd)
1160 IF r(s,z)=0 AND k=0 THEN PEN 0:k=1:GOTO 1140
1170 PEN 1:NEXT i:RETURN
1180 REM UP Warten
1190 LOCATE 7,25
1200 PRINT"Bitte eine Taste druecken !"
1210 x$=INKEY$
1220 IF x$="" THEN 1210
1230 CLS:RETURN

```

Variablenliste:

a\$ = Antwortstring (j/n)

cd = "Chipdesign", numerischer Ausdruck für das
Zeichnen der Chips im Unterprogramm Satz 1110-
1170

co(i,j) = Rasterfeld (i für die Spalte und j für die Zei-
le); gleich 1, wenn der Computer das Feld be-
setzt hat, ansonsten gleich 0

h = Zählvariable für die Anzahl der in einer Reihe
liegenden Chips

i = Laufindex

il = Spaltenindex für ein nicht besetztes Feld

ip = Spaltenindex bei der Betrachtung der Diagonalen

ipl = Spaltenindex bei der Betrachtung der Diagonalen
für ein nicht besetztes Feld

j = Laufindex

jl = Zeilenindex für ein probenhalber besetztes Feld
(s ist der entsprechende Spaltenindex)

jp = Zeilenindex bei der Betrachtung der Diagonalen

jpl = Zeilenindex bei der Betrachtung der Diagonalen
für ein nicht besetztes Feld

k = Hilfsvariable im Unterprogramm Satz 1110-1170

l = Laufindex für den zu testenden Spaltenwert

p = Hilfsvariable, die anzeigt, welcher der beiden
Denkschritte ausgeführt wird

$r(i,j)$ = Rasterfeld (i für die Spalte und j für die Zei-
le); gleich 1, wenn es besetzt ist (egal mit
welchen Chips), ansonsten gleich 0

$s,s(i)$ = ausgewählter oder berechneter bzw. per Zufall
ermittelter Spaltenwert

$sb(i,j)$ = Rasterfeld (i für die Spalte und j für die Zei-
le); gleich 1, wenn der Spielpartner des Com-
puters das Feld besetzt hat, ansonsten gleich 0

$sb1$ = untere Begrenzung einer Programmschleife

$sb2$ = obere Begrenzung einer Programmschleife

z = Zeilenindex im Unterprogramm Satz 1110-1170

Programmbeschreibung:

Satz 10- 60 : Überschrift

Satz 70- 240 : Ausgabe der Spielregeln und Abfrage be-
züglich des Spielbeginns

Satz 250- 460 : Das Spielfeld wird aufgezeichnet. Zusätz-
lich werden in Satz 460 die Rasterfelder
mit der Zeile 0 gleich 1 gesetzt, um dem
Computer mitzuteilen, daß ein Chip nur
bis in die erste Zeile und nicht tiefer
fällt.

Satz 470 : Es wird eine Zufallszahl zwischen 3 und 6 gebildet, wenn der Rechner mit dem Spiel beginnen soll. Weiterhin werden dann die entsprechenden Felder belegt und der Spielstein wird eingezeichnet.

Satz 480- 500 : Spaltenauswahl des Programmbenutzers, Felderbelegung und Einzeichnen des Chips

Satz 510-1040 : Der Computer denkt nach

Dieser Programmteil wird zuerst für $p=0$ durchlaufen, d. h. der Computer untersucht, ob er das Spiel zu seinen Gunsten entscheiden und somit beenden kann. Ist das nicht der Fall, erfolgt ein Durchlauf für $p=1$, d. h. der Computer prüft, ob die zufällig in Satz 1010 gebildete Zahl nicht dazu führt, daß der Spielpartner das Spiel gewinnt. Diese zufällig gebildete Zahl geht nur dann als Auswahl des Computers in das Spiel ein (Satz 980), wenn nicht während des Durchlaufs zum Satz 1000 (Test einer neuen Zahl) verzweigt wurde. Ein Sprung zum Satz 1000 erfolgt immer dann, wenn die Auswahl einer Zahl bzw. Spalte zur Niederlage des Rechners führen würde. Er beinhaltet gleichzeitig, daß die in Satz 1030 probehalber besetzten Felder $r(s,j1)$ und $co(s,j1)$ wieder gleich 0 gesetzt werden. Maximal werden die 8 möglichen Spalten durchprobiert (Satz 1000-1020, nur die erste Zahl wird zufällig gebildet). Nach 8 erfolglosen Durchläufen gibt der Computer auf und verzweigt zum Programmende (mit Kommentarausgabe in Satz 1050), d. h. der Programmbenutzer muß u. U. überlegen, warum er gewonnen hat.

Im einzelnen lassen sich folgende Programmteile unterscheiden:
510- 610 : Untersuchung der Vertikalen
620- 730 : Untersuchung der Horizontalen
740- 850 : Untersuchung der von links unten nach rechts oben ansteigenden Diagonalen
860- 970 : Untersuchung der von rechts unten nach links oben ansteigenden Diagonalen
980-1040 : Einzeichnen eines Chips und Sprung zur Auswahl des Spielpartners, falls ein Durchlauf erfolgreich absolviert wurde, sowie Bildung von Probezahlen, Probebelegungen der Felder $r(s,j)$ und $co(s,j)$ und Rücksprung zum Satz 510

Satz 1050-1060 : Kommentar zum Spielausgang

Satz 1070 : Programmende

Satz 1080-1100 : Unterprogramm zur Belegung des Feldes $r(s,j)$ in Abhängigkeit von der ausgewählten Spalte s . Befinden sich schon 6 Chips in der Spalte, so erfolgt ein Rücksprung zur erneuten Auswahl (die Variable cd zeigt dabei an, ob der Computer oder der Spielbenutzer die Auswahl getroffen hat).

Satz 1110-1170 : Unterprogramm zum Einzeichnen eines fallenden Chips

Satz 1180-1230 : Unterprogramm "Warten"

10.4. Ideenfundgrube

- Brettspiele wie Dame, Mühle oder Schach
- Kartenspiele wie Skat, 17 und 4 oder Doppelkopf
- Würfelspiele und Knobeleien
- Memory oder Scramble
- Monopoly oder Risiko
- Labyrinthspiele
- Schiffe versenken
- Mondlandung
- Invaderspielchen jeglicher Art
- und noch vieles mehr


11. Schlußbemerkung

Nachdem Sie dieses Buch eifrig oder weniger eifrig durchgearbeitet haben, werden Sie hoffentlich um die eine oder andere Idee reicher geworden sein.

Vielleicht ist Ihnen aufgefallen, daß sich einige Programmteile und Routinen in den einzelnen Kapiteln wiederholen. Denken Sie nur an die übliche Menüauswahl mit den Funktionen

- eingeben
- ansehen
- einlesen
- abspeichern
- korrigieren

Zwar lassen sich bei grundsätzlich ähnlichen Problemen sog. Programmierrichtlinien entwickeln, welche den Programmablauf standardisieren, jedoch ist darauf bewußt verzichtet worden, um der Vorgehensweise und Denkweise des Hobby- und Amateur-Programmierers näher zu kommen. Dieser wird nämlich in der Regel von einem Thema bzw. einem Problem ausgehend eine Lösung suchen und nicht weitere Softwareprojekte im Hinterkopf haben. Auch insofern stellt das vorliegende Buch eine Übungs- und Trainingshilfe für den angehenden Programmierer dar.



Anhang A

Wichtige BASIC-Anweisungen

Da alle Schneider-BASIC-Anweisungen ausführlich in Ihrem Bedienungshandbuch beschrieben sind, sollen hier nun einige elementare Regeln zu ihrer Verwendung aufgezeigt werden:

1. Jedes BASIC-Programm wird mit dem Statement END abgeschlossen.
2. Die auszugebenden Größen im PRINT-Statement werden durch Kommata getrennt, wenn der Ausdruck zu Beginn der nächsten Bildschirmzone fortgesetzt werden soll. Bei Benutzung des Semikolons schließt sich der Ausdruck unmittelbar an die schon erfolgte Ausgabe an. Ein Komma oder ein Semikolon am Ende eines PRINT-Statements unterdrückt den sonst erfolgenden Zeilenvorschub.
3. Das INPUT-Statement fordert genau so viele Werte an, wie Variablen in der Variablenliste stehen. Die einzelnen Werte sowie die Variablen werden jeweils durch Kommata getrennt.
4. Ist die Bedingung im IF ... THEN-Statement erfüllt, dann wird die hinter THEN stehende Anweisung ausgeführt.

Auch weitere durch Doppelpunkte voneinander getrennte Anweisungen, die hinter dem IF ... THEN-Statement stehen, werden nur dann ausgeführt, wenn die logische Bedingung erfüllt ist. Ist die Bedingung nicht erfüllt, wird einfach der nächste Satz bearbeitet.

5. In FOR ... TO ... NEXT-Statement (Schleife) dürfen Anfangswert, Endwert und Schrittweite während der Schleifendurchläufe nicht verändert werden. Der Laufvariablenname muß vom reellen Typ - und der gleiche wie im NEXT-Statement sein. Schleifen können geschachtelt -, verlassen - und in Schleifen kann hineingesprungen werden.


6. Werden für eine indizierte Variable mehr als die per Voreinstellung zur Verfügung stehenden Plätze benötigt, muß dimensioniert werden, d.h. Platz reserviert werden (DIM-Statement).

7. Bei der Bildung reeller Variablen muß darauf geachtet werden, daß ihre Namen aus maximal 40 alphanumerischen Zeichen bestehen dürfen. Das erste Zeichen muß jedoch ein Buchstabe sein. Stringvariablen (Zeichenfolgen) werden mit dem \$-Zeichen versehen.

8. Werden dem Rechner mit dem DATA- und READ-Statement Daten bereitgestellt, so müssen Werttypen und Variablentypen übereinstimmen. Dabei wird der ersten Variablen der READ-Statements der erste Wert des DATA-Statements zugeordnet usw.

9. Hinter den Statements ON ... GOTO und ON ... GOSUB werden im gleichen Satz keine weiteren Anweisungen ausgeführt.

10. Folgende WINDOW-Statements überdecken die Ergebnisse vorhergehender WINDOW-Statements.



Anhang B

Wichtige Fehlermeldungen

ARRAY ALREADY DIMENSIONED: Ein Feld wurde zweimal dimensioniert.

CANNOT CONTINUE: Versuch, nach einer Fehlermeldung oder einer Programmänderung, das Programm fortzusetzen.

DATA EXHAUSTED: Es sind mehr READ als DATA-Anweisungen vorhanden.

DIVISION BY ZERO: Eine Division durch Null ist unzulässig.

LINE DOES NOT EXIST: Sprung zu einer nicht vorhandenen Satznummer.

NEXT MISSING: In einer Programmschleife fehlt das NEXT-Statement.

OVERFLOW: Das Ergebnis einer Rechnung ist zu groß.

REDO FROM START: Es ist versucht worden, nichtnumerische Daten bei einer INPUT-Anweisung in eine normale Variable

einzulesen. Durch eine erneute Eingabe kann das Programm fortgesetzt werden.


SUBSCRIPT OUT OF RANGE: Ein außerhalb der Dimensionierung liegendes Matrix-Element wurde angesprochen.

SYNTAX ERROR: Eine Anweisung ist fehlerhaft oder existiert nicht.

TYPE MISMATCH: Numerische Variablen wurden als String oder umgekehrt behandelt.


UNEXPECTED NEXT: In einer Programmschleife fehlt das FOR-Statement.

UNEXPECTED RETURN: Subroutine ohne GOSUB




In diesem erstklassigen Buch wird gezeigt, wie man die außergewöhnlichen Grafik- und Soundmöglichkeiten des CPC 464 nutzt. Natürlich mit vielen interessanten Beispielen und nützlichen Hilfsprogrammen. Aus dem Inhalt: Grundlagen der Grafikprogrammierung, Sprites, Shapes und Strings, mehrfarbige Darstellungen, Koordinationstransformation, Verschiebungen, Drehungen, Rotation, 3-D-Funktionsplotter, CAD, Synthesizer, Miniorgel, Hüllkurven und vieles mehr.

Walkowiak
CPC 464 Graphik & Sound
220 Seiten, DM 39,-
ISBN 2-89011-050-9



Alles über Diskettenprogrammierung vom Einsteiger bis zum Profi. Natürlich mit ausführlichem ROM-Listing (Betriebssystem), einer äußerst komfortablen Dateiverwaltung, einem hilfreichen Disk-Monitor und einem ausgesprochen nützlichen Disk-Manager. Dazu eine Fundgrube verschiedener Programme und Hilfsroutinen, die das Buch für jeden Floppyanwender zur Pflichtlektüre machen.

Brückmann/Schieb
Das Floppy-Buch zum CPC 464
ca. 250 Seiten, DM 49,-
ISBN 3-89011-093-2



Wer die wichtigen Peeks und Pokes zum CPC 464 kennen und anwenden will, der findet hier umfassende Information. Sie reicht vom Adreßbereich des Prozessors über Betriebssystem und Interpreter bis hin zur Einführung in die Maschinensprache. Dazu präzise Programmierhilfen, sinnvolle Routinen sowie reichlich Material zu den Themen Grafikfunktionen, Massenspeicherung und Peripherie, Tricks und Formeln in BASIC, RAM-Pages.

Lisert/Schieb

Peeks & Pokes zum CPC 464

ca. 220 Seiten, DM 29,-

ISBN 3-89011-092-4



Endlich CP/M beherrschen – mit diesem Trainingsbuch kein Problem! Von ganz grundsätzlichen Erklärungen zu beispielsweise Speicherung von Zahlen, Schreibschutz oder ASCII-System über Schnittstellen, andere Betriebssysteme und Anpassung von CP/M bis hin zur ausführlichen Behandlung des Schwerpunktes Dateien. Außerdem CP/M-Hilfsprogramme und ihre Anwendung. Speziell zugeschnitten auf Ihren CPC 464!

Weller/Schieb
CP/M Trainingsbuch zum CPC 464
ca. 300 Seiten, DM 49,-
ISBN 3-89011-089-4



Von den Grundlagen der Maschinenspracheprogrammierung über die Arbeitsweise des Z80-Prozessors und einer genauen Beschreibung seiner Befehle bis zur Benutzung von Systemroutinen ist alles ausführlich und mit vielen Beispielen erklärt. Im Buch enthalten sind Assembler, Disassembler und Monitor als komplette Anwenderprogramme. So wird der Einstieg in die Maschinensprache leichtgemacht!


Dulin/Straßenburg
Das Maschinensprachebuch zum CPC 464
330 Seiten, DM 39,-
ISBN 3-89011-070-3




Wer sich für den CPC 464 entschieden hat, findet mit diesem DATA BECKER Buch gleich den richtigen Start. Wichtige Hinweise zu Handhabung und Anschlußmöglichkeiten, Hilfen für eigene Programme auf dem CPC 464. Viele Abbildungen ergänzen den Text. Das ideale Buch für jeden CPC 464 Computeranfänger.
CPC 464 für Einsteiger, 206 Seiten, DM 29,-




Das sollte Ihr erstes Buch zum CPC 6128 sein. CPC 6128 für Einsteiger ist eine sehr leicht verständliche Einführung in Handhabung und Einsatz des CPC 6128, die keinerlei Vorkenntnisse voraussetzt. Dazu eine Einführung in BASIC, wobei Sie eine komplette Adressenverwaltung erstellen, die Sie anschließend nutzen können. Unentbehrlich für jeden CPC 6128 Einsteiger!
CPC 6128 für Einsteiger, ca. 200 Seiten, DM 29,- (erscheint ca. Oktober)




Anregungen und Hilfen zum CPC. Von Hardwareaufbau, Betriebssystem, BASIC-Tokens, Zeichnen mit dem Joystick, Anwendungen der Windowstechnologie und vielen Programmen wie einer umfangreichen Dateiverwaltung, Soundeditor, komfortablem Zeichengenerator bis zu kompletten Listings spannender Spiele bietet das Buch eine Fülle von Möglichkeiten.
CPC Tips & Tricks, 263 Seiten, DM 39,-




Dieser Band ist für alle CPC Besitzer interessant. Aus dem Inhalt: Menügenerator, Maskengenerator, BASIC-Befehlsweiterungen, Programmierhilfen wie Dump, BASIC-Zeile von BASIC aus erzeugen, wichtige Systemroutinen und deren Nutzung, Beschleunigung von Programmen und vieles mehr.
CPC Tips & Tricks Band II, mehr als 250 Seiten, DM 39,- (erscheint im November)



Das Standardwerk zum CPC! Neben dem ausführlich dokumentierten und kommentierten BASIC-ROM-Listing enthält es umfangreiche Kapitel zu Speicheraufteilung, Prozessor, Besonderheiten des Z 80, Gate Array, Video-Controller und Video-Ram, Soundchip, Schnittstellen, Betriebssystem, Routinenutzung, Character-Generator, BASIC-Interpreter und mehr.
CPC 464 INTERN, 548 Seiten, DM 69,-



Ein Muß für jeden, der sich professionell dem CPC 664 oder dem CPC 6128 beschäftigt. Einführung in das System, den Prozessor, das Gate Array, den Video-Controller, den Schnittstellenbaustein 8255, den Soundchip, die Schnittstellen. Dazu natürlich ein komplettes und ausführlich dokumentiertes Listing des BASIC-Interpreters und des Betriebssystems. Mit Schaltplänen!
CPC 664/6128 INTERN, mehr als 500 S., DM 69,- (ersch. im Nov.)



Nützliche und pfiffige Ideen rund um Freizeit und Alltag! Programme mit Beschreibung und Beispielen zu: Lotto – Benzinverbrauch – Geld und Kredit – Schreiben und Verwalten – Staat und Steuer – Haushaltsorganisation – Fußball – Kinder und Schule – und vieles mehr.
Das Ideenbuch zum CPC, ca. 250 Seiten, DM 39,-

Programme zu SCHNEIDER CPC

BUDGET-MANAGER
DM 148,-*

Der BUDGET-MANAGER ist die universelle Buchführung sowohl für private Zwecke als auch zur Planung, Überwachung und Abwicklung von Budgets jeglicher Art.



Es können 430 Budgetsätze und 335 Kontensätze verwaltet werden. Ein Budgetplan sowie auch ein Kontenplan können per Programm erstellt werden. Die Einsatzmöglichkeiten sind äußerst vielfältig: Einzelanzeige von Konten - Geldfälligkeiten - Amortisation - Zinsen - Tilgung - bis zur richtigen kleinen Privatbilanz. Das Programm ist voll menuegesteuert. Es zeigt jeweils an, auf welcher Seite des Handbuchs man die Erklärungen zum entsprechenden Programmteil findet. Zusätzlich werden sehr komfortable Druckausgabe- und Auswertungsmöglichkeiten geboten: Bildschirm, Drucker, Grafikausgabe auf dem Bildschirm und deren Hardcopy auf dem Drucker. Erforderliche Hardware: CPC mit Diskettenlaufwerk, ein zweites Laufwerk wird unterstützt. Komplett mit ausführlichem Handbuch.

TEXTOMAT
DM 148,-*

TEXTOMAT

Deutschlands meistverkaufte Textverarbeitung gibt es jetzt endlich auch für den Schneider CPC 464!

Das gesamte Programm ist in benutzerfreundlicher Menütechnik aufgebaut. Wahlweise kann der amerikanische oder deutsche Zeichensatz verwendet werden. Diese Anpassung gilt auch für den Drucker und ermöglicht individuelle Anpassung an jedes Problem! Selbstverständlich wurden bei dieser Textomat-Version die Möglichkeiten, speziell des Schneider CPC 464 ausgenutzt; so besitzt diese Version zusätzlich zu dem bewährten 64-Textomat 80-Zeichen/Zeile-Darstellung, Tabulatoren, Word Wrap und Trennvorschläge! Textomat kurz zusammengefaßt: Durch Verknüpfungstechnik können beliebig lange Texte erstellt, gespeichert, geladen, verändert und ergänzt werden. Auch Funktionen wie Suchen/Ersetzen, Blockoperationen oder gar Rechenfunktionen für alle Grundrechnungsarten fehlen im TEXTOMAT nicht! Die Ausgabe auf den Drucker geschieht problemlos durch Druckeranpassung. Das Druckformat kann beliebig gewählt werden. TEXTOMAT ist zukunftsorientiert; in Verbindung mit der Dateiverwaltung DATAMAT ist auch die Erstellung von Serienbriefen kein Problem mehr! Im Lieferumfang eingeschlossen ist ein ausführliches deutsches Handbuch! Übrigens: Viele DATA BECKER Autoren nutzen TEXTOMAT wegen seiner Vielseitigkeit und leichten Bedienung sogar zum Erstellen ganzer Bücher!

DATAMAT
DM 148,-*

DATAMAT

Endlich auch auf dem Schneider CPC 464: Die bewährte DATA BECKER-Dateiverwaltung jetzt in einer speziellen Version für alle Freunde dieses universell einsetzbaren Mikrocomputers!

Das Hauptprogramm liegt vollständig im Speicher - Nachladen und lästige Diskettenwechsel sind also nicht mehr nötig! Durch die Menütechnik ist DATAMAT extrem einfach zu bedienen. Es lassen sich jede Art von Daten verarbeiten, die Eingabemaske ist vom Benutzer völlig frei definierbar (Es stehen bis zu 80 Zeichen/Zeile zur freien Verfügung). So läßt sich DATAMAT als Dateiverwaltung praktisch überall einsetzen. Bis zu 512 Zeichen können jetzt pro Datensatz verarbeitet werden. Durch sämtliche Menüpunkte wird der Benutzer mit Hilfe von Begleitanweisungen bedienerfreundlich geführt. Falsche Bedienung ist praktisch unmöglich! Durch entsprechende Routinen läßt sich jeder Datensatz superschnell in 2-3 Sekunden suchen (Programm komplett in Maschinensprache). Es läßt sich nach einem Hauptkriterium oder beliebigen anderen Daten eines Datensatzes suchen. Auch der Ausdruck ist kein Problem: Druckeranpassung an viele Drucker ist selbstverständlich möglich. Listen lassen sich in völlig freiem Format drucken, auch Etiketten - kein Problem! DATAMAT ist nicht allein: selbstverständlich gibt es eine Schnittstelle zum TEXTOMAT - beide lassen sich ideal kombinieren. Dem Ausbau des Systems steht also nichts im Wege! Das Programm wird mit umfangreichem deutschen Handbuch geliefert.

* unverbindliche Preisempfehlung. Alle Programme auf Diskette.

DAS STEHT DRIN:

Sie haben sich Ihren CPC Schneider nicht nur zum Spielen gekauft? Dann ist dies das richtige Buch für Sie. Fast alles, was man mit den CPC's machen kann, ist in diesem Buch beschrieben. Es ist nicht nur interessant geschrieben, sondern enthält neben nützlichen Programmlistings vor allem viele, viele Anwendungsmöglichkeiten. Dabei wurde besonders Wert darauf gelegt, daß das Buch auch für den Einsteiger leicht verständlich ist.

Eine Auswahl aus der Themenvielfalt :

- Schaufensterwerbung
- Auto und Computer
- Fahrtstreckenoptimierung
- Autokosten fest im Griff
- Geld, Kredit und Computer
- Zinseszinsberechnung
- Texten und Drucken
- Textverarbeitung
- Der Staat und das Geld
- Rentenberechnung
- Lohnsteuerjahresausgleich
- Haushalt und Gesundheit
- Elektronischer Kalorienzähler
- Malen, Zeichnen und die Erstellung von Grafiken
- Musik
- Fußballbundesliga
- Vokabeltrainer

UND GESCHRIEBEN HAT DIESES BUCH:

Bernd Kowal, erfahrener Computerbuchautor, zeigt mit diesem Buch, daß er ein ausgezeichneter Kenner des CPC ist und sich mit den Anwendungsmöglichkeiten sehr intensiv auseinandergesetzt hat.

ISBN 3-89011-101-7

Kowal / Das Ideen-Buch zum CPC

CPC

AMSTRAD

CPC



MÉMOIRE ÉCRITE
MEMORY ENGRAVED
MEMORIA ESCRITA



<https://acpc.me/>

[FRA] Ce document a été préservé numériquement à des fins éducatives et d'études, et non commerciales.

[ENG] This document has been digitally preserved for educational and study purposes, not for commercial purposes.

[ESP] Este documento se ha conservado digitalmente con fines educativos y de estudio, no con fines comerciales.